
この文書について

この文書は琉球大学工学部電気電子工学科の学生向けの講義や実験などの補助教材として作成されています. 作
成にあたって, 読者が UNIXの基本的な操作に習熟していることを想定しています.
本文書は基本的にチュートリアルです. 読者が自分がやりたいことに似た例題を探しながら読み進めてゆくこと

を想定しています. 部分的にはリファレンスとしても使えるように配慮してある箇所もありますが, 全般に説明は
網羅的ではないので注意して下さい.
この文書はコピーフリーとします. 本文書に一切改変を加えないという条件の下で, 自由な再配布を許可します.
著者は本文書の内容に関して正確を期してはいますが, 本文書の内容の正しさに関しては一切保証しません. ま

た, 本文書の内容に関する質問は一切受け付けません. ただし, 誤りの報告については歓迎します.

この文書の使い方

この文書は全 10章から成ります.
第 1章は gnuplotの最低限の操作を覚えるための簡単なチュートリアルです. gnuplotの操作に慣れていない読

者はこのチュートリアルをすべて読んで下さい.
第 3章では gnuplotの基本的なコマンドが説明されています. また, 第 4章では gnuplot のコマンドの基本的な

構文と, gnuplotの組み込み関数, gnuplotで新しく関数を宣言する方法が説明されています. ここで説明されてい
る内容は gnuplot の基礎ですので, gnuplotに不慣れな読者はざっと目を通しておくとよいでしょう. ただし, こ
れらの記述を記憶する努力をする必要はありません.
第 5章ではターミナルの切り換えについて説明します. この章は必要に応じて参照すればよく, 読者があらかじ

め目を通しておく必要はありません.
第 6章には gnuplotのグラフの各部の名称とグラフの各部を調整するためのコマンドの一覧が記載されていま

す. この章は, 読者が第 7章以降のチュートリアルを見ていて用語やコマンドの意味について混乱したときに参照
することを意図して書かれています. 通読するときには, gnuplotのグラフの各部の名称に関する記述の部分のみ
に目を通せば十分です.
第 7章, 第 8章, 第 9章はそれぞれ 2次元グラフ, 3次元グラフおよびデータファイルからグラフを作成する方法

のチュートリアルです. これらの章のうち, 第 7章の関する記述はかなり詳細なのですが, 第 8章, 第 9章では, 第
7章と同じ使い方ができる事項に関する記述はある程度省略されています. これらの章は, 読者が自分の希望する
操作に類似した例を探して使うことをを意図して書かれています. 通読する必要はありません.
第 10章は plusモード に関するリファレンスです. この章は, LATEXの知識を持たない読者にも plus モードが

使えるように書かれています. この章はリファレンスですから, 通読する必要はありません.
本文書では, 原則として, ユーザが直接入力する文字列をタイプ文字で標記しています. なお,コマンド入力後に

は, 行末で Enter キー を押す必要があります. 本文中のコマンドの例では, [Enter]という記号で Enter キー

を押す箇所を標記しています.

このドキュメントに含まれないもの

gnuplotが取り扱える多数のターミナルのうち取り扱われているのは x11, postscript, tgif, latex, png,

pbm のみです. 多くのコンピュータシステムのターミナルについては, 筆者の手元に動作チェックできる環境が

ないので説明を省略しました. また, 複数用意されている TEX系のターミナルの中では, latex のみを簡単に

取り上げました. これは, 機能がすくない LATEXの picture環境で作図をおこなう必然性が筆者にはあまり感じ

られなかったからです.

2次元グラフ作成に関する機能のうち, financebars などの工学系の学生が使うことはまずないと思われる事

項に関する説明は省略しました.

gnuplotの 3次元グラフ作成に関する機能に関する説明は, gnuplotの 3次元グラフ作成機能にやや不安定な

部分があるのと, 3次元グラフそのものの利用頻度が 2次元グラフと比較して高いとは思えないので, ある程度

省略してあります.

gnuplotでアニメーションを作成することもできるのですが, これに関する記述も省略しました.

第1章 gnuplotとは

gnuplotは, 2次元および 3次元のグラフを描画するためのフリーウェアです.

gnuplotのユーザインターフェースは利用者がコマンドを打ち込んでゆく形態 (CUI)です. ですから, gnuplot

を使うためには, 最低限のコマンドをいくつか覚えておく必要があります. 一方, コマンド help で閲覧する

ことができるオンラインマニュアルが充実しているため, 最低限のコマンドさえ覚えてしまえば, より高級な使

い方をするときに困ることはほとんどありません.

gnuplotの 2次元グラフ描画機能は極めて強力であり, 各種の関数やデータのグラフが自由自在に作成できま

す. 非常に多様な画像の形式をサポートしていることも gnuplot の特徴です. gnuplotで作成したグラフは,

PostScript形式や EPS形式, tgifのオブジェクトファイル, PNGフォーマット, PBMフォーマットなど, 各種

のフォーマットで保存することができます. gnuplotには 3次元グラフを作成する機能もありますが, こちらは

2次元ほど強力ではありません.

本文書では, gnuplotの基本的な使い方について, チュートリアル形式で 2次元の場合を中心に解説してゆき

ます.

第2章 チュートリアル

以下のコマンドの例では, [Enter]という記号で Enter キーを押す箇所を標記しています.

2.1 gnuplotを起動する

gnuplotを起動するには, ktermなどで

gnuplot [Enter]

と入力します. すると, 図 2.1のように gnuplotの起動メッセージが画面に表示され, gnuplotが立ち上がり

ます.

図 2.1: gnuplotの起動画面

2.2 gnuplotを終了する

gnuplotを終了するには, gnuplotの画面内で

quit [Enter]

と入力します.

2.3 基本的な操作法

gnuplotの基本的な操作法は (t)csh や bash などとほぼ同様です. キーボードからコマンドを打ち込んで

Enter キーを押すと, 打ち込んだ内容が実行されます. コマンド行は編集可能で, 編集には以下のキーが使え

ます.

キー 1 キー 2 効果

[Ctrl]+ f → 1文字前進

[Ctrl]+ b ← 1文字後退

[Ctrl]+ p ↑ 1個前に入力した行を表示

[Ctrl]+ n ↓ 1個次に入力した行を表示

[Ctrl]+ e 行末に移動

[Ctrl]+ a 行頭に移動

[Ctrl]+ d [Del] カーソル位置の文字を消す

[Ctrl]+ h [BackSpace] カーソルの手前の文字を消す

[Ctrl]+ k カーソルの位置から行末までを消す

[Ctrl]+ l 画面を再描画する

[Ctrl]+ u 行全体を消す

[Ctrl]+ w 最後の単語を消す

2.4 1変数関数を描画する

関数のグラフを描画をするにはコマンド plotを使います. たとえば, gnuplotのウィンドウ内で

plot sin(x) [Enter]

と入力すると, 図 2.2のように正弦関数のグラフが描画されます. 横軸の範囲と縦軸は適当に調整されます.

1

-10 -5 0 5 10

図 2.2: 正弦関数のグラフ

2.5 epsファイルに保存する

描画したグラフをLATEXで処理可能なeps (encapsulated PostScript)形式で保存するには, まずset terminal

というコマンドを使って出力データの形式を postscriptに変更します.

set terminal postscript eps [Enter]

キーボードからの入力に誤りがなければ,

Options are ’eps noenhanced monochrome

dashed defaultplex "Helvetica-Ryumin" 14’

とうようなメッセージが画面に表示されます.

続いて, set outputというコマンドを使って出力ファイルの名前を指定します. ここでは, test.epsとし

ておきましょう.

set output ’test.eps’ [Enter]

ここまでの手順が終わった状態で

replot [Enter]

と入力すると, グラフが test.epsというファイルに保存されます. ここまでの作業が終わったときの gnuplot

のウィンドウの内容を図 2.3 に示します.

図 2.3: ファイルへの保存が終わったときの gnuplot のウィンドウの状態

いったん出力データの形式を postscript 形式に変更すると, それ以降は, 利用者が出力データの形式を切

り換えるコマンドを実行するまで, 出力データの形式はずっと postscript のままになります. ところで, 出

力データの形式が postscript になっていると, 描画したグラフのようすを画面で確認することはできません.

再びグラフを画面で確認できる状態に戻したいときには,

set terminal x11 [Enter]

というコマンドを実行する必要があります.

2.6 非対話的なgnuplotの実行

gnuplotコマンドをファイルにあらかじめ書いておいて, そのファイルを実行することもできます.

図 2.4のような内容のファイルがカレントディレクトリに用意されているものとしましょう. ファイル名は

plot sin(x)

pause -1

図 2.4: gnuplotのスクリプトファイルの例

test.gnuplotとします.

このようなファイルを準備した上で kterm などで

gnuplot test.gnuplot [Enter]

と入力すると, 先ほどと同じように正弦関数のグラフが画面に描画されます.

なお, 非対話的に gnuplotを実行した場合には, gnuplotはすべてのコマンドを実行し終わるとただちに終了

してしまいます. 上の例で

pause -1

とある部分は, ユーザが [Enter]キーを押すまで gnuplotが終了しないようにするためのものです.

2.7 ファイルからコマンドを実行する

gnuplotを起動し, gnuplotのウィンドウ内でコマンド load を実行することでも, あるファイルに書かれた

コマンド群を実行することができます. コマンド load を使うときには, 引数としてファイル名を単一引用符

「’」で囲って指定します.

例として, 先ほどの例と同じファイル test.gnuplotを実行する場合を考えましょう.

このためには,

load ’test.gnuplot’ [Enter]

と入力します.

2.8 コマンドのファイルへの保存

コマンド save を実行すると, 最も最近のコマンド plotに関連した一連のコマンド群をファイルに保存でき

ます.

コマンド save を使うときも, 引数としてファイル名を単一引用符「’」で囲って指定します.

例として, 今実行したコマンドを test2.gnuplotというファイルに保存してみます.

save ’test2.gnuplot’ [Enter]

コマンド save を実行した場合には, gnuplot が標準で設定するいろいろなパラメータの内容がすべてファイ

ルに書き込まれるため, ファイルの内容はかなり複雑になります.

第3章 gnuplotの基本的なコマンド

表 3.1に gnuplot の基本的なコマンドを示します. 本節ではこれらの中の代表的なものについて簡単に説明

してゆきます.

表 3.1: gnuplotの基本的なコマンド

機能 コマンド

ファイルとディレクトリの操作

終了する exit, quit

現在のディレクトリを表示する pwd

ディレクトリを変更する cd

ファイルをロードする call, load

ファイルをもう 1回読む reread

ファイルにデータを保存する save

グラフの描画

2次元グラフをプロットする plot

3次元グラフをプロットする splot

グラフが表示されるウィンドウをクリアする clear

グラフを再描画する replot

グラフの一時停止 pause

テスト画面を表示する test

数値やオプションなどの設定と表示

変数の値などを画面に表示する print

各種オプションを設定する set

各種設定の内容を表示する show

すべての設定をクリアする reset

データファイルの操作

データに直線や曲線をあてはめる fit

直線あてはめ時にパラメータをファイルに書き出す update

その他

条件分岐 if

シェルに戻る shell

オンラインマニュアルを見る help

3.1 ファイルとディレクトリの操作

3.1.1 現在のディレクトリを表示する

現在のディレクトリを表示するには, pwdというコマンドを使います.

コマンド pwd の使い方は UNIXのそれと同じで,

pwd [Enter]

です.

コマンド pwd を実行すると, たとえば図 3.1のように, 現在作業をしているディレクトリが表示されます.

図 3.1: コマンド pwdを実行したところ

3.1.2 ディレクトリを移動する

ディレクトリを移動するには cd というコマンドを使います.

コマンド cdの使い方は UNIXとは異なり, オプションとしてディレクトリ名を単一引用符で囲ったものを与え

ます.

たとえば, /home/test というディレクトリに移動したいときには,

cd ’/home/test’ [Enter]

と入力します.

ディレクトリを指定するときには, 文字「.」(ピリオド 1個)で現在のディレクトリを, 文字列「..」(ピリオ

ド 2個)で現在のディレクトリの 1個根本のディレクトリをあらわすことができます. また, 記号「/」がディ

レクトリ名の先頭についている場合は絶対パス (ルートディレクトリが起点), ディレクトリ名の先頭に記号「/」

がついていない場合は相対パス (現在作業中のディレクトリが起点)として解釈されます.

これ以外の UNIXの標準的な記法, たとえば記号「~」などは使えません.

3.1.3 ファイルをロードする

gnuplotのコマンドや変数などが書かれたファイルを読むためのコマンドには, load と call の 2種類があ

ります.

3.1.3.1 オプションなしでファイルをロードする – load –

コマンド loadは, 単純に gnuplotのコマンドなどが書かれたファイルを読み込むコマンドです. あらかじ

めファイルに一連の gnuplotのコマンドを書いておいてからそのファイルを load コマンドで読み込むと, その

ファイルに書かれたコマンドすべてがキーボードから打ち込んだ場合と同じように上から下に順に実行されてゆ

きます.

コマンド load の使い方は,

load ’ファイルの名前’ [Enter]

です. ここに, 「ファイルの名前」と書かれた部分には, gnuplotのコマンドなどが書かれたファイルが入り

ます.

たとえば, 図 3.2のような内容のファイルを用意しておいて (ファイル名を sample1.gp とします), gnuplot

のウィンドウ内で

load ’sample1.gp’ [Enter]

というコマンドを実行することは, キーボードから

set terminal x11 [Enter]

set xrange [0:10] [Enter]

set yrange [0:10] [Enter]

plot x+1 [Enter]

set terminal postscript eps [Enter]

set output ’test.eps’ [Enter]

replot [Enter]

という内容を 1行ずつ打ち込んでゆくことと同一の効果があります.

set terminal x11 set xrange [0:10] set yrange [0:10] plot x+1 set

terminal postscript eps set output ’test.eps’ replot

図 3.2: ファイル sample1.gp の内容

なお, コマンド load は, 読み込まれるファイルがカレントディレクトリにあると仮定します. カレントディ

レクトリ以外にあるファイルを読み込みたいときには, 絶対パスあるいは相対パスでディレクトリと合わせてファ

イル名を指定するか, あるいはコマンド cd を使ってそのファイルがあるディレクトリまで移動して下さい.

たとえば, 絶対パスを使って読み込むファイル名を指定するときには

load ’/home/test/sample1.gp’ [Enter]

のようにします.

3.1.3.2 オプションつきでファイルをロードする – call –

コマンド call のはたらきは load とほぼ同様なのですが, コマンド load にはパラメータがないのに対して

コマンド call は 10個までのパラメータを持つことができるという点が違います. ですから, コマンド call

を使った場合は, あらかじめファイルに書き込んでおく gnuplotのコマンドをまとめたファイルにいくつか可変な

部分を持たせておいて, そのファイルを読み込むときにいくつかパラメータを指定する, ということができます.

コマンド call を使って読み込まれるファイル中では, パラメータが$0から$9までの記号で指示されます.

例として, やや人工的ですが, 図 3.3のようなファイルを考えましょう. ただし, ファイル名を sample2.gp

とします.

plot [$0:$1] sin(x) with $2

図 3.3: ファイル sample2.gp の内容

このファイルを call を使って読み込む場合は, パラメータとしてグラフの初期値, グラフの終了値とグラフ

の書き方を指定することができます.

なお, ここで出て来たコマンド plot の使い方といろいろなオプションの指定法については第 7章で詳しく説

明されています. ここでは, $0と書かれたところに横軸の左端, $1と書かれたグラフの横軸の右端の座標を書

き込み, $2と書かれたところに「lines]という文字を書き込むとグラフが線で描画され, $2と書かれたところ

を「points]という文字に変えるとグラフが点で描画されるということだけ理解しておいて下さい.

さて, 図 3.3のようなファイルを用意した上で, gnuplotのウィンドウで

call ’sample2.gp’ 0 10 lines [Enter]

と入力すると図 3.4 のようなグラフが,

call ’sample2.gp’ 5 25 points [Enter]

と入力すると図 3.5 のようなグラフが得られます.

横軸の読みとグラフの形状から, 今指定したパラメータが確かにグラフに反映されていることが理解できるで

しょう.

コマンド call も, 読み込まれるファイルがカレントディレクトリにあると仮定します. カレントディレク

トリ以外にあるファイルを読み込みたいときには, 絶対パスあるいは相対パスでディレクトリと合わせてファイ

ル名を指定するか, あるいはコマンド cd を使ってそのファイルがあるディレクトリまで移動して下さい.

なお, gnuplot のコマンドが書かれたファイルでパラメータがまったくないものを読み込むときにも コマン

ド call を使うことができます. この場合の挙動はコマンド load と完全に同一になります.

3.1.4 ファイルをもう 1回読む

コマンド load を使って読み込まれたファイルをもう 1回読み直して実行するには reread というコマンドを

使います.

コマンド reread の使い方は,

reread [Enter]

です.

図 3.4: コマンド call の使用例 (1)

図 3.5: コマンド call の使用例 (2)

表 3.2: コマンド save の使い方

機能 コマンド

最も最近のコマンド plotとすべて
の関数, 変数および設定を保存

save ’ファイル’ [Enter]

すべての関数を保存 save functions ’ファイル’ [Enter]

すべての変数を保存 save var ’ファイル’ [Enter]

すべての設定を保存 save set ’ファイル’ [Enter]

3.1.5 関数や変数などを保存する

いろいろな関数や変数などを保存するときには, コマンド saveを使います. コマンド save には 4種類の使

い方があります. これらを表 3.2にまとめておきます.

なお, 「ファイル」と書かれた部分には適当なファイル名が入ります.

コマンド save を実行すると gnuplotを起動した標準で設定される膨大なパラメータがすべてファイルの保

存されるため, ファイルの大きさがかなり大きくなります.

3.2 グラフの描画

グラフの描画に関連したコマンドについては次章以降で詳しく説明されているので, ここでは, 以下であまり

触れる機会がないコマンドについて簡単に紹介しておきます.

3.2.1 一時停止

グラフが表示されたところで一旦作業を止めるコマンドが pause です. このコマンドは gnuplot のコマン

ドをファイルから読み込んで実行していて, グラフが表示されるごとに一旦実行を停止したい場合に有効です.

コマンド pause の使い方は 2種類あります.

ひとつめは,

pause 数 [Enter]

というものです. これは, 「数」の部分に指定された秒だけ gnuplotの実行を中断します.

ふたつめは,

pause -1 [Enter]

というものです. これは, [Enter]キーが押されるまで実行を中断します.

実行を中断しているときに何かメッセージを表示したいときには,

pause 数 ’表示するメッセージ’ [Enter]

あるいは

pause -1 ’表示するメッセージ’ [Enter]

のように, 表示したいメッセージを単一引用符で囲って指定します.

3.2.2 テスト画面を描画する

今グラフが描画されるウィンドウがどのような状態になっているかをユーザに知らせるために, テスト画面を

表示するコマンドが用意されています. このコマンドの名前が test です.

test [Enter]

のようにして実行します.

3.3 数値やオプションなどの設定と表示

3.3.1 オプションを設定する

オプションを設定するには set というコマンドを使います. set の使い方は

set キーワード 1 キーワード 2 ... [Enter]

のように, コマンド set に続いて必要なキーワードをたくさん並べてゆく, というものです. 具体的な使い方

については次章以降で説明します.

3.3.2 オプションの状態を表示する

いろいろなオプションを設定するコマンド set に対応して, オプションの状態を表示するコマンドがありま

す. それが show です.

コマンド show の基本的な使い方は,

show オプションの名前 [Enter]

です. 具体的な使い方は次章以降で説明します.

3.3.3 変数の値を表示する

変数の値などを画面に表示するコマンドが print です.

gnuplotを起動した段階では, 円周率をあらわす変数 pi だけが設定されています.

この数値を表示するには,

print pi [Enter]

とします. すると, 図 3.6に示すように, 変数 πの値 (円周率)が表示されます.

図 3.6: 変数 pi の値が表示されたところ

3.3.4 オプションをすべてクリアする

コマンド set で設定できるすべてのオプションをクリアするためのコマンドが reset です. 使い方は

reset [Enter]

です.

3.4 オンラインマニュアルを見る

オンラインマニュアルを見るためのコマンドが help です. このコマンドの基本的な使い方は

help トピック [Enter]

です.

コマンド help の使い方は,

help help [Enter]

とすることで参照できます.

マニュアル表示をやめるときには,

q

と入力します.

コマンド help を実行すると, そのトピックに関する簡単な説明に続いて, もしあればサブトピックの一覧が

表示されます.

たとえば, gnuplotのウィンドウで,

help help [Enter]

と入力すると, 画面は図 3.7のようになります.

図 3.7: コマンド help のオンラインマニュアル

コマンド help のオンラインマニュアルはサブトピックを含まず, かつマニュアル本文が短いので, マニュア

ルが 1画面におさまっていて, マニュアルの末尾に (END)という文字が白黒反転して表示されています. マニュ

アルが表示された状態からは,

q

と入力すれば抜けられます.

これに対し,

help plot [Enter]

と入力すると, 画面は図 3.8のようになります.

図 3.8: コマンド plot のオンラインマニュアル

図 3.8において, 画面の最下段の行に

\texttt{:}\prompt

というプロンプトが出ていることに注意して下さい. この状態では, ページャ(たいてい less, jless, more

のいずれか, 環境変数 PAGERによって決まる) が自動的に起動して, マニュアルを表示しています. ですから,

そのページャのコマンドはすべて利用可能です.

さて, [Space]キーを押すなどしてマニュアルの最後尾まで画面をスクロールさせると画面は図 3.9のように

なります. 図 3.9では

Subtopics available for plot:

という文字列に続いてキーワードの一覧が表示されていることに注意して下さい.

図 3.9: コマンド plot のオンラインマニュアルの末尾

図 3.9の状態で

q

と入力すると, 画面は図 3.10のようになります.

図 3.10: コマンド plot のオンラインマニュアルにおける subtopicの選択

ここで, 画面下段の

Subtopic of plot: \prompt

というプロンプトに注意して下さい. 図 3.10の状態からは, 3種類の操作が可能です. これらを表 3.3に示

します.

表 3.3: オンラインマニュアルの subtopic選択画面における操作

機能 コマンド

オンラインマニュアルを抜ける [Enter]

サブトピックの一覧を表示する ? [Enter]

特定のトピックのマニュアルを見る トピック [Enter]

ただし, 「トピック」と書かれたところには,
キーワード一覧で表示された用語のいずれか

が入る

たとえば, 図 3.10の状態で, キーワード with のマニュアルを見たいときには,

with [Enter]

と入力します. すると, 画面には, 図 3.11のようにキーワード with のマニュアルが表示されます.

図 3.11: キーワード with のマニュアルを表示しているところ

マニュアルを最後までスクロールさせて

q

と入力すると, 図 3.12のようにサブトピック選択画面に戻ります.

図 3.12: サブトピック選択画面に戻ったところ

3.5 その他のコマンド

3.5.1 条件分岐

コマンド if は条件分岐のために用いられます. 基本的な構文は

if(条件) コマンド

というものです. 条件全体をかっこで囲む必要があります.

3.5.2 シェルに戻る

gnuplotを一旦抜けてシェルでコマンドを実行するために, shell というコマンドが用意されています. 使

い方は

shell [Enter]

です.

シェルから抜けるときには

exit [Enter]

と入力します.

3.6 コマンドの省略記法

gnuplotのいろいろなコマンドにはいろいろな省略記法が用意されています. その代表的なものは, 「最初の

数文字だけを書く」という書き方です. たとえば,

plot sin(x) [Enter]

と入力しても,

p sin(x) [Enter]

と入力しても,

pl sin(x) [Enter]

と入力しても, 同一の結果が得られます. すなわち, 「p」や「pl」はコマンド plot の省略記法とみなされる

のです.

上記からわかるように, 曖昧さが生じない範囲で, 多くのコマンドにいろいろな省略記法が用意されています.

ただし, あまり省略記法を多用すると, あとで見直してみると自分でも何をやっているかわからない, というこ

とになることがあるので, 省略はほどほどにした方がよいでしょう.

第4章 gnuplotの構文と関数

4.1 構文

gnuplotは, ユーザが入力するコマンドを 1行ずつ解釈して実行してゆきます. コマンド行には, コマンド

およびオプションやキーワードなどを 1個以上の空白 (スペース)で区切って並べます. キーボードからコマン

ドを入力しているときには [Tab]キーは効果がありませんが, gnuplotのコマンドをファイルに書いておくと

きには空白のかわりにタブ文字 ( [Tab]キーを押すと入力される文字で, 4文字から 8文字分の空白に見える)

も使えます. 空白が 2個以上あるときの効果は空白 1個の場合と変わりません. なお, いわゆる全角スペース

を空白として使うことはできません.

入力するオプションなどに文字列が含まれるときには, その文字列を単一引用符 (「’」)で囲みます. たいて

いの場合は単一引用符のかわりに 2重引用符 (「"」)を使うこともできますが, 例外もあるので, つねに単一引

用符を使うほうが良いでしょう.

オプションなどで x座標と y座標を指定する必要があるときには, それらをコンマで区切って並べます. 同様

に, x座標と y座標および z座標を指定する必要があるときにも, それらをコンマで区切って並べます.

gnuplotのコマンドは改行キーを押したときに確定します.

行頭が文字「#」で始まる行は注釈とみなされます. gnuplotは注釈を無視するので, 3.1.3節に述べたよう

にあらかじめ gnuplotのコマンドが書かれたファイルを作っておいてそれを実行する場合には, 文字「#」で始ま

る行にいろいろなコメントを書いておいて, あとでそのファイルを読み直すときに便利なようにしておくことが

できます.

行の途中に注釈を入れることもできます. そのためには, 行中に文字「#」を入れます. すると, gnuplot

は文字「#」から改行文字までの部分をすべて無視します.

先に述べたように, gnuplotのコマンドは改行キーを押したときに確定するのですが, gnuplotのコマンド中

に記号「\」(バックスラッシュ)を入力してそれに続いて [Enter]キーを押すと, コマンドは実行されず, か

わりに「>」という記号が行の先頭にあらわれます.

たとえば,

plot \ [Enter]

と入力すると, gnuplotの画面は図 4.1 のような状態になります. 図 4.1の 2行目で,

図 4.1: 記号 \を使ったときの状態

gnuplot>

というプロンプトのかわりに

>

というプロンプトが出ていることに注意して下さい.

この状態で, 次に

sin(x) [Enter]

と入力すると,

plot sin(x) [Enter]

と入力した場合と同じ内容が実行されます. すなわち,

\ [Enter]

とすると, 改行文字の入力確定という意味をなくすことができるのです.

この機能は, 3.1.3節に述べたように, あらかじめ gnuplotのコマンドが書かれたファイルを作っておいてコ

マンド load を使ってそれを実行する場合に便利です. ファイルに書くコマンドが長くなってしまった場合は,

コマンド中に適宜「\」を挿入して, ファイルを見やすくすることができます.

図 4.2に, 長すぎるコマンドを「\」を使って見やすくしている例を示します.

plot [-2*pi:2*pi] [-1:1] sin(x)

title ’正弦関数’

with linespoints

linetype 3 linewidth 4

pointtype 4 pointsize 4

図 4.2: コマンドの途中における改行

なお, 記号「\」(バックスラッシュ)は端末の状態によっては「￥」記号に表示されることもあり, キーボー

ドにも「￥」記号の方が書いてあることがあります.

4.2 数式と組み込み演算子

gnuplotには, 表 4.1のような演算子があります.

演算子の優先順位が変わる部分に横線が引いてあります. 横線が引かれた部分の上側は優先順位が高く, 下側

は優先順位が低くなります.

たとえば,

x**4- 2*x + 3

と書いた場合, 演算子**の方が優先順位が高いので, これは

(x**4)- 2*x + 3,

すなわち

x4 − 2x + 3

表 4.1: gnuplotの演算子

記号 意味

** べき † (x**2は xの 2乗)

- 単項の負演算子 †

~ ビットごとの否定

! 論理否定

! 階乗

* 乗算 †

/ 除算 †

% 剰余

+ 加算 †

- 減算 †

== 等値

!= 非等値

& ビットごとの論理積 (and)

^ ビットごとの排他論理和 (xor)

| ビットごとの論理和 (and)

&& 論理積 (and)

|| 論理和 (or)

?: 条件演算子

と解釈されます.

優先順位はかっこ ()をつけることで変更できます. 演算子の優先順位はあまり記憶しやすいものではないの

で, いちいち記憶するよりは, 積極的にかっこを使って優先順位を明示した方がよいでしょう.

「意味」と書かれた欄に記号 † の付いた演算子は整数, 実数, 複素数に対して適用可能ですが, それ以外は整

数に対してのみ適用可能です.

4.3 数値の演算の結果を画面に表示する

数値の演算の結果を画面に表示するには, 3.3.3節で紹介されたコマンド print を使います.

たとえば, 2の平方根の数値を画面に表示したいときには,

print sqrt(2) [Enter]

と入力します.

結果は図 4.3のようになります.

4.4 ユーザ変数

ユーザ変数とは利用者が自分で値を定義する変数のことをいいます. 本節ではユーザ変数の操作の仕方につい

て説明します.

図 4.3: 2の平方根の値が画面に表示されたところ

4.4.1 ユーザ変数の一覧を見る

ユーザ変数の定義の方法の説明に入る前に, 現在定義されているユーザ変数の一覧を表示する方法を説明して

おきましょう.

ユーザ変数の一覧を表示するには, gnuplotのウィンドウで

show variables [Enter]

と入力します.

gnuplotを立ち上げた直後の状態では, 上記を実行した結果は図 4.4のようになります. 初期状態では pi と

図 4.4: ユーザ変数の初期状態

いうユーザ変数が定義されていて, それが円周率 πをあらわしていることがわかります.

4.4.2 ユーザ変数を新しく定義する

次に, 新しくユーザ変数を定義する方法を説明しましょう.

新しいユーザ変数を定義するには,

変数名 = 数値 [Enter]

という構文を用います.

変数名の中で使える文字は, 英文字の大文字と小文字および文字「_」(アンダースコア)です. 大文字と小文

字は区別されます. 空白は使えません. 変数名は, これらの利用可能な文字を有限個並べたものになります.

たとえば, ネイピアの数 e (約 2.71828)を定義したいときには,

e=2.71828 [Enter]

とします.

e=2.71828 [Enter]

show variables [Enter]

を実行すると, 結果は図 4.5のようになります. 図 4.5からユーザ変数 eが追加されていることがわかります.

図 4.5: ユーザ変数 e が追加された状態

グラフを描くときによく使うユーザ変数があるときには, キーボードから毎回読み込むのは無駄ですし, 誤入

力の危険もあるので, あらかじめユーザ変数をまとめたファイルを用意しておいて, 3.1.3節で述べられている

手順にしたがってファイルを読み込むとよいでしょう.

4.4.3 ユーザ変数をファイルに保存する

現在定義されているユーザ変数をすべてファイルに保存することもできます. このためには, save var とい

うコマンドを使います.

コマンド save var の使い方は,

save var ’ファイル’ [Enter]

です. ここに, 「ファイル」と書かれた部分には適当なファイルの名前が入ります. ファイルの名前に日本語

や空白などを含めると問題が生じることがあります. ファイル名には英文字と記号「_」, 「-], 「.」以外は使

わないようすれば安全です.

たとえば,

e=2.71828 [Enter]

save var ’test.gp’ [Enter]

quit [Enter]

とすると, 現在作業中のディレクトリに test.gp というファイルができます. このファイルの内容は図 4.6の

ようになります.

図 4.6において, 最初の部分はこのファイルが gnuplot のコマンドであることを示す部分, その下の行頭が

文字「#」で始まる部分はすべて注釈で, 最後から 2行目の部分にユーザ変数の情報が書かれています. 最後の

行には「# EOF」というコメントが書かれています.

このようにして保存されたファイルは第 3.1.3節で解説されているコマンド load を使って読み込むことがで

きます. ファイルを読み込むと, そこで定義されているユーザ変数が使えるようになります.

図 4.6: コマンド save var によって生成されるファイルの内容

4.5 複素数の表現

gnuplotは複素数を扱うことができます. 複素数を表現するときには, 中括弧 {}の中に実部と虚部をコンマ
で区切って指定します.

例えば, 4+3i は gnuplotでは {4,3}と表記されます.

複素数を用いてグラフを描画するときには虚数単位がよく使われるので, 第 4.4節の手順にしたがって虚数単

位に対応するユーザ変数を用意しておくとよいでしょう.

たとえば, I(大文字の i)で虚数単位をあらわすことにする場合には,

I = {0,1}

とします.

4.6 数学関数

gnuplotにはいろいろな数学関数が用意されています. この節では, これらを簡単に紹介します.

4.6.1 指数関数, 対数関数, 三角関数

gnuplotに用意されている指数関数および対数関数の一覧を表 4.2にまとめておきます.

表 4.2の関数の意味は名前から明らかだとは思われますが, 代表的なものについて簡単に説明します.

指数関数は exp という名前です. 引数としては実数あるいは複素数を取ることができます.

対数関数には, 自然対数 log と常用対数 log10 の 2種類があります.

三角関数の名前はそれぞれ sin(正弦関数), cos(余弦関数), tan(正接関数)です. gnuplotを起動したと

きには単位としてラジアンが指定されていますが, 単位を度に変更することができます.

逆三角関数としては, asin, acos, atan の 3種類が用意されています. 逆三角関数は返却値の単位をラジ

アンあるいは度に変更できます.

単位を度とラジアンで変更するには, set angles というコマンドを使います.

表 4.2: 指数関数, 対数関数, 三角関数

関数 引数の型 意味

exp(x) 任意 指数関数

log(x) 任意 自然対数

log10(x) 任意 常用対数

sin(x) 任意 正弦関数

cos(x) 任意 余弦関数

tan(x) 任意 正接関数

asin(x) 任意 逆正弦関数

acos(x) 任意 逆余弦関数

atan(x) 任意 逆正接関数

atan2(y,x) 整数,実数 tan−1 y の実部
x の実部 を返す

set angles degrees [Enter]

と入力すると単位が度に変更され,

set angles radians [Enter]

と入力すると単位がラジアンに変更されます.

1回単位をラジアンと度のあいだで変更すると, その後は (自分でもう 1回単位を変更するまで) ずっとその単

位が有効になります.

なお, 正割関数, 余割関数, 余接関数とその逆関数は用意されていません.

4.6.2 算術関数

gnuplotには 実数や複素数の演算や整数への変換のためのいろいろな関数が用意されています. これらを表

4.3にまとめておきます.

4.6.3 その他の関数

gnuplotには, 上に述べたもの以外にもいろいろな数学関数が用意されています. これらを表 4.4にまとめ

ておきます.

以下に書いたもののうち, 双曲線関数とベッセル関数の単位はラジアンに固定されているので注意して下さい.

4.7 ユーザ関数の定義

ユーザ関数とは利用者が自分で値を定義する関数のことをいいます. 本節ではユーザ関数の操作の仕方につい

て説明します.

表 4.3: 実数, 複素数の演算と整数への変換

関数 引数の型 意味

abs(x) 任意 絶対値関数. 引数の絶対値を返す

sgn(x) 任意 符号関数. 引数の実部が正なら 1, 負なら 0を返す. 引数の実部が 0のときは 0
を返す

arg(x) 複素数 複素数の偏角を返す. 返却値の単位はラジアンか度のいずれか. set anglesに
よって返却値の単位をラジアンにするか度にするかを選択する

sqrt(x) 任意 引数の平方根を返す

ceil(x) 任意 引数以上の最小の整数を返す. 引数が複素数のときにはその実部に対して作用
する

floor(x) 任意 引数以下の最大の整数を返す

int(x) 実数 引数が正のときは引数以下の最大の整数を, 引数が負のときは引数以上の最小の
整数を返す

real(x) 任意 複素数の引数の実部を返す

imag(x) 複素数 複素数の引数の虚部を返す

表 4.4: その他の関数

関数 引数の型 意味

sinh(x) 任意 双曲正弦関数.
cosh(x) 任意 双曲余接関数

tanh(x) 任意 双曲正接関数

asinh(x) 任意 逆双曲正弦関数

acosh(x) 任意 逆双曲余弦関数

atanh(x) 任意 逆双曲正接関数

gamma(x) 任意 引数の実部にガンマ関数を作用させたものを返す

igamma(x) 任意 引数の実部に不完全ガンマ関数を作用させたものを返す

lgamma(x) 任意 引数の実部にガンマ関数を作用させてから自然対数を取ったものを返す

ibeta(x) 任意 引数の実部に不完全ベータ関数を作用させたものを返す

besj0(x) 整数, 実数 0次のベッセル J関数
besj1(x) 整数, 実数 1次のベッセル J関数
besy0(x) 整数, 実数 0次のベッセル Y関数
besy1(x) 整数, 実数 1次のベッセル Y関数

rand(x) 任意 引数の実部を種と, 0から 1のあいだに分布する擬似乱数を返す
erf(x) 任意 引数の実部にオイラー関数を作用させたものを返す

erfc(x) 任意 1-erfc(x)を返す

inverf(x) 任意 引数の実部にオイラー関数の逆関数を作用させたものを返す

norm(x) 任意 引数の実部に正規分布関数を作用させたものを返す

invnorm 任意 引数の実部に逆正規分布関数を作用させたものを返す

4.7.1 ユーザ関数の一覧を見る

ユーザ関数の定義の方法の説明に入る前に, 現在定義されているユーザ関数の一覧を表示する方法を説明して

おきましょう.

29



ユーザ関数の一覧を表示するには, gnuplotのウィンドウで

show functions [Enter]

と入力します.

gnuplotを立ち上げた直後の状態では, 上記を実行した結果は図 4.7のようになります. 初期状態ではユー

図 4.7: ユーザ関数の初期状態

ザ関数は何も定義されていないことがわかります.

4.7.2 ユーザ関数を新しく定義する

次に, 新しくユーザ関数を定義する方法を説明しましょう.

ユーザ関数を定義するときには,

関数名 (x)=定義式 [Enter]

関数名 (x,y)=定義式 [Enter]

というような内容を入力します.

関数名の中で使える文字は, 英文字の大文字と小文字および文字「_」(アンダースコア)です. 大文字と小文

字は区別されます. 空白は使えません. 変数名は, これらの利用可能な文字を有限個並べたものになります.

変数の部分には x, y, z以外の文字を使ってもよいのですが, 特別な理由がある場合を除き x, y, z と指定

した方がよいでしょう.

たとえば, 余接関数 cot xを定義するときには,

cot(x)=1/tan(x) [Enter]

とします.

cot(x)=1/tan(x) [Enter]

show functions [Enter]

というコマンドを実行した結果を図 4.8に示します.

4.7.3 ユーザ関数を使う

一旦定義されたユーザ関数は, gnuplotに標準で用意されている関数と同じように使えます. ただし, 関数

を使うときには, その関数を定義するときにどのような変数を使ったかにかかわりなく, 変数としてはその文脈

に適した文字を使う必要があります.

たとえば,

図 4.8: 新しい関数が定義された状態

cot(w)=1/tan(w) [Enter]

と定義した上で, ここで定義された関数 cotのグラフをプロットするときには,

plot cot(x) [Enter]

とします.

plot cot(w) [Enter]

とするとエラーになるので注意して下さい. 関数を使うときには, その関数の変数名はつねにその文脈 (2次元か

3次元か, パラメータ付きか否か, これらについては後述) に応じて適合したものに置き換えられてゆくのです.

4.7.4 関数を使うときに変数名を変更する

関数を使うときにその関数の変数名を変更したい場合には, set dummy というコマンドを使います.

コマンド set dummy の使い方は,

set dummy 変数名 [Enter]

です. 2変数関数を使うときには,

set dummy 変数 1, 変数 2 [Enter]

とします.

たとえば

set dummy time [Enter]

plot sin(time) [Enter]

とすると,

plot sin(x)

とした場合と同様に正弦波のグラフが描画できます.

同様に,

set dummy h,v [Enter]

splot sin(h+v) [Enter]

とすると,

splot sin(x+y) [Enter]

とした場合と同様のグラフが描画されます.

4.7.5 ユーザ関数をファイルに保存する

現在定義されているユーザ関数をすべてファイルに保存することもできます. このためには, save functions

というコマンドを使います.

コマンド save functions の使い方は,

save functions ’ファイル’ [Enter]

です. ここに, 「ファイル」と書かれた部分には適当なファイルの名前が入ります. ファイルの名前に日本語

や空白などを含めると問題が生じることがあります. ファイル名には英文字と記号「_」, 「-], 「.」以外は使

わないようすれば安全です.

このようにして保存されたファイルはコマンド load を使って読み込むことができます. ファイルを読み込

むと, そこで定義されているユーザ関数が使えるようになります.

第5章 グラフの印刷とターミナルの切り換え

5.1 グラフの確認から印刷までの一連の流れ

gnuplotでグラフを確認してから印刷するまでの作業の流れは,

1. 関数をプロットする

2. ターミナルを印刷用データのものに切り変える

3. 出力ファイルを指定する

4. 画像をファイルに保存する

5. ターミナルを標準的な画面に戻す

というものです.

参考のために,

1. 正弦関数をプロットする

2. ターミナルを PostScriptの epsモードに切り換える

3. 出力を test.eps というファイルに変える

4. 画像をファイルに保存する

5. ターミナルを X Window System に戻す

という一連の作業を実行している例を図 5.1に示しておきます.

図 5.1: gnuplotの標準的な印刷の流れ

関数やコマンド, ファイル名やターミナルなどが変わることがあっても, 図 5.1に示された作業の流れはだい

たい共通です. ですから, この作業の流れは記憶しておいて下さい.

5.2 ターミナルと出力ファイルの指定

gnuplotが起動したときには, ターミナル (画像を表示する装置)が X Window System になっています.

グラフを印刷する作業は,

• ターミナルを X Window System から印刷用のものに切り換える

• 出力ファイルなどを指定する

という 2工程から成ります.

ターミナルの切り換えには set terminal というコマンドを使い, 出力ファイルなどの指定には set output

というコマンドを使います.

5.2.1 ターミナルの切り換え

コマンド set terminal は,

set terminal ターミナルの名前 [Enter]

のようにして使います. 「ターミナルの名前」の部分にて指定できるものの中で代表的なものを表 5.1にまとめ

ておきます.

表 5.1: ターミナルの種類

ターミナルの名前 説明

x11 X Window System
postscript PostScript形式, 代表的な印刷用フォーマットのひとつ
tgif tgifのオブジェクトファイル, 描画ツール tgif で使うための形式
latex 文書作成用ツール LATEXに取り込むための形式
table 関数の値を表にして作成

png png形式, 代表的な印刷用フォーマットのひとつ
pbm pbm形式, 代表的な印刷用フォーマットのひとつ

これらの使い方については次節以降で説明することにして, ここではターミナルを PostScriptの eps モード

に切り換える例を示すにとどめます.

ターミナルを PostScriptの eps モードに切り換えるには,

set terminal postscript eps [Enter]

とすると, gnuplotのウィンドウは図5.2 のような状態になり, 表示されたメッセージからターミナルがPostScript

の epsモードになっていることが確認できます.

この状態のとき, グラフは画像として確認できるフォーマットではなく, PostScript形式 (印刷用のフォー

マットで)作成されます.

印刷用のフォーマットのデータをファイルに保存するためには, これに続いて次節で述べるコマンド set output

で出力ファイルを指定する必要があります. コマンド set output を使うのを忘れると, 作成されたデータは

標準出力 (gnuplotのウィンドウ)にそのままテストとして表示されてしまいます.

上に述べたように, ターミナルが postscript になっていると, いくらグラフを描画しても作成された図は画

面上では確認できません.

ファイルの保存が終わってもう 1回画面で画像が確認できる状態に戻りたいときには, gnuplotのウィンドウ

内で

34



図 5.2: ターミナルを PostScriptの epsモードに切り換えた状態

set terminal x11 [Enter]

と入力します. すると, gnuplotのウィンドウは図 5.2のような状態になり, ターミナルが x11 になったこと

が確認できます.

図 5.3: ターミナルを X11の epsモードに切り換えた状態

ターミナルが x11になっているときには, 出力ファイルを指定する必要はありません. また, コマンド set

output で出力ファイルの名前が指定されていても無視されます.

5.2.2 出力ファイルの指定

出力ファイルを指定するときには, コマンド set output を使います.

コマンド set output の使い方は,

set output ’ファイル’ [Enter]

とします. ここに, 「ファイル」と書かれた部分には適当なファイル名が入ります.

現在の出力の設定を見るには, コマンド show output を使います.

コマンド show output の使い方は,

show output ’ファイル’ [Enter]

です.

ターミナルと出力ファイルが正しく設定された状態でコマンド plot あるいは コマンド splot を実行すると,

グラフが指定されたファイルに保存されます.

また, 1回コマンド plot あるいは splot を実行してグラフを画面で確認してからファイルに保存する場合

には, ターミナルと出力ファイルを指定したあとでコマンド replot を実行する, すなわち

replot [Enter]

と入力すると, 先ほど確認したものと同じグラフがファイルに保存されます.

ターミナル が x11以外になっているときには, コマンド set output で出力ファイルを設定することが必須

です. 忘れないようにして下さい.

5.3 いろいろなターミナル

本節では, いろいろなターミナルとそのオプションについて説明してゆきます.

5.3.1 x11

X Window Systemでグラフを画面上で確認するためのターミナルです.

set terminal x11 [Enter]

とするとこのウィンドウを選択できます.

グラフを表示するウィンドウを複数作りたいときには,

set terminal x11 番号 [Enter]

とします. ここに, 「番号」と書かれた部分には 0以上の数を指定します. ただし, 数 0を指定することと番

号を指定しないことの効果は同じです.

たとえば,

set terminal x11

plot sin(x)

set terminal x11 1

plot cos(x)

set terminal x11 2

plot exp(x)

とすると, 画面には「Gnuplot」「Gnuplot 1」「Gnuplot 2」と書かれた 3枚のウィンドウが開きます (図 5.4).

それぞれのウィンドウにマウスカーソルを合わせて

q

と入力すると, そのウィンドウを閉じることができます. また,

set terminal x11 reset

と入力すると, 開いているすべての画像ウィンドウを閉じることができます.

gnuplotを終了すると, gnuplotが作成した画像のウィンドウもすべて閉じます. このウィンドウを閉じない

ようにする (gnuplotが終了してからも画面に残っている) ように変更したいときには, gnuplotを起動すると

きに

gnuplot -persist [Enter]

のように, gnuplotに persist というオプションを付けます.

ターミナル x11 では日本語を問題なく取り扱うことができます.

図 5.4: gnuplotで複数の画像ウィンドウを開いたところ

5.3.2 postscript

PostScriptは文字や画像などのデータを保存する形式の代表的なもののひとつです. gnuplotでは, PostScript

形式は, postscript という名前のターミナルで利用できます.

ターミナルを postscritpに切り換えるには,

set terminal postscript モード オプション [Enter]

と入力します. ここに, 「モード」と「オプション」には後で説明するモードとオプションに対応する文字が入

ります. モードおよびオプションはすべて省略できますが, ほかの文書に取り込む画像を作成しているときには

モードを eps にしておいた方がよいでしょう.

ターミナルが postscript になっているときは, コマンド set output で出力ファイルを設定することが必

須です. 忘れないようにして下さい. postscript形式のファイルの拡張子は通常 eps あるいは ps です.

5.3.2.1 モード

postscriptには epsモード(他の文書に取り込む図の作成用), portraitモード(用紙縦置き), landscape(用

紙横置き), default(デフォルト)の 4種類のモードがあります. これらを表 5.2にまとめておきます.

表 5.2: ターミナル postscript のモード

モードの名称 意味

eps encapsulated PostScriptモード (他の文書に取り込む図の作成用)
portrait 用紙縦置きモード

landscape 用紙横置きモード

default 標準

5.3.2.2 オプション

postscriptモードにはいくつかオプションがあるのですが, これらの中で利用頻度が高いと思われるものを

以下にまとめておきます. 表の中で, 横線を引いてまとめてあるオプションを同時に指定することはできません

が, 他のグループにあるオプションについては複数組み合わせられます. なお, 表 5.3の左端に「標準」と書い

てある項目がターミナルを postscriptに変更したときに自動的に選択されるオプションです. これらの値を変

更する必要がないときには, オプションを指定しなくてもよいです.

なお, 表 5.3にある plus モードは図の縦軸や横軸に自由に数式などを入れることができる強力なオプション

です. この plusモードは gnuplot に対するパッチ (差分ファイル)として gnuplot の開発元とは別の開発

者が提供しているもので, 標準の gnuplot には含まれません. ですから, gnuplot がインストールされた状

態によっては, plusモードの機能が使えないこともあります. plusモードに関する詳しい説明をすると本節の

主題から逸脱してしまうので, これについては第 10章で改めて取り上げることにします.

表 5.3: ターミナル postscript の代表的なオプション

オプション 意味

noplus 数式などを取り扱うための拡張モード (10章参照)を無効にする 標準

plus 数式などを取り扱うための拡張モードを有効にする

nocolor 白黒画像を生成 標準

color カラー画像を生成

dashed 点線をそのまま保存する 標準

solid 点線を実線で置き換える (カラー画像で使われることがある)

数 フォントの大きさを指定

ターミナル postscript では日本語を問題なく取り扱うことができます.

5.3.3 tgif

描画ツール tgif の形式のターミナルが tgif です. ターミナルを tgif に変更するには,

set terminal tgif [Enter]

と入力します.

ただし, ターミナル を tgif にする場合で, 図に日本語の文字が含まれているときには, 必ず

set terminal tgif ’Ryumin-Light-EUC-H’ [Enter]

のように入力して下さい. この理由は以下で述べます.

また, ターミナルが tgif になっているときは, コマンド set output で出力ファイルを設定することが必

須です. 忘れないようにして下さい. tgifのファイルの拡張子は obj です.

ターミナル tgif にもいろいろなオプションがあります. これらを表 5.4にまとめておきます.

表 5.4: ターミナル tgif のオプション

オプション 意味

portrait 用紙縦置き 標準

landscape 用紙横置き

dashed 点線をそのまま保存する 標準

solid 点線を実線で置き換える (カラー画像で使われることがある)

’フォント’ フォントの名前を指定

数 フォントの大きさを指定

ここで1点注意すべきことがあります. ターミナル tgif では, フォントを正しく設定しないと, 日本語が表示

できず, 文字化けします. 日本語を表示するためのフォントは, やや名前が長いのですが, Ryumin-Light-EUC-H

です. ですから, ターミナル を tgif にする場合で, 図に日本語の文字が含まれているときには, 日本語フォ

ントの指定は必須です. このような場合には, 必ず

set terminal tgif ’Ryumin-Light-EUC-H,Helvetica’ [Enter]

のように明示的にフォントを指定して下さい. 他のオプションを指定するか否かは任意です.

なお, これはおそらくバグと思われるのですが, terminalを tgifにした状態では, 日本語の文字と英数字

が混在したグラフを正しく保存することができません. また, tgif形式で保存したグラフを tgifで編集した

場合,

• 日本語の部分の編集が正しくできない

• PostScript形式で保存すると文字化けする

という問題が生じることがあります. グラフを tgif形式で保存する場合には, gnuplotで描画するときにはグ

ラフには日本語を入れずにおき, 日本語入力および編集作業はすべて tgifでおこなうのが良いでしょう.

5.3.4 latex

文書作成ツール LATEX のソースコードを生成するターミナル が latex です. ターミナルを latex にする

と, グラフが LATEX の picture 環境という環境の形式で生成されます. このようにして生成されたファイルを

LATEX で作成された文書にそのまま挿入することで, 図入りの文書が作成できます.

ターミナルが latex になっているときは, コマンド set output で出力ファイルを設定することが必須です.

忘れないようにして下さい. latexのファイルの拡張子は tex です.

LATEX のピクチャー環境については LATEX のマニュアルを参照して下さい.

5.3.5 table

ターミナルを table にすると, グラフを描くかわりにグラフを描くために計算されたいろいろな点の座標と,

それに対応する関数の値が, テキスト形式の表の形で書き出されます.

ターミナルが table になっているときは, コマンド set output で出力ファイルを設定することが必須です.

忘れないようにして下さい.

5.3.6 png

ターミナルを png にすると, 代表的な画像フォーマットのひとつである Portable Network Graphics(png)

フォーマットで画像を保存することができます.

ターミナルが png になっているときは, コマンド set output で出力ファイルを設定することが必須です.

忘れないようにして下さい. ファイルの拡張子は通常 png です.

ターミナル pngには表 5.5のようなオプションがあります. ただし, 大きいフォントを選択すると日本語が

表 5.5: ターミナル png のオプション

オプション 意味

small 小さいフォント 標準

medium 中サイズのフォント

large 大きいフォント

monochrome 白黒 標準

gray グレイスケール

color カラー

表示されなくなるので注意して下さい.

5.3.7 pbm

ターミナルを pbm にすると, 代表的な画像フォーマットのひとつである Portable BitMap(pbm)フォーマッ

トで画像を保存することができます.

ターミナルが pbm になっているときは, コマンド set output で出力ファイルを設定することが必須です.

忘れないようにして下さい. ファイルの拡張子は通常 pbm です.

ターミナル pbmのオプションには pngファイルと同様のオプションがあります. これについては表 5.5を参

照して下さい. なお, 大きいフォントを選択すると日本語が表示されなくなるので注意して下さい.

第6章 gnuplotのグラフの構造

6.1 グラフの各部の名称

6.1.1 2次元グラフの各部の名称

図 6.1に gnuplotが描画した 2次元グラフの主要部の名称を示します.

図 6.1: 2次元グラフの各部の名称

図6.1の各項目の簡単な説明などを表6.1にまとめておきます. なお, 図中には記載がありませんが, zeroaxis

というキーワードは x軸と y軸の双方を指します.

表 6.1: 2次元グラフの各部の名称の説明

項目 説明 項目 説明 項目 説明

title グラフの標題 xlabel x軸の説明 ylabel y軸の説明
key 点や曲線の説明 xtics x軸の主目盛り ytics y軸の主目盛り
border グラフ全体を囲む箱 x2tics x軸の第 2目盛り y2tics y軸の第 2目盛り
zeroaxis 座標軸 (x軸と y軸) xzeroaxis 座標軸 (x軸) yzeroaxis 座標軸 (y軸)

6.1.2 3次元グラフの各部の名称

図 6.2に gnuplotが描画した 3次元グラフの主要部の名称を示します.

図 6.2: 3次元グラフの各部の名称

図 6.2の各項目の簡単な説明などを表 6.2にまとめておきます. なお, 見にくくなるために図中には示され

ていませんが, 2次元グラフと同様に, グラフを囲む箱 (3次元の場合はグラフの上部は囲われない)を border

といいます.

表 6.2: 3次元グラフの各部の名称の説明

項目 説明 項目 説明 項目 説明

title グラフの標題 xlabel x軸の説明 zlabel z軸の説明
key 点や曲線の説明 xtics x軸の目盛り ztics z軸の目盛り
border グラフを囲む箱 ylabel y軸の説明
isosamples グラフの網かけ ytics y軸の目盛り

2次元グラフは x軸と y軸にそれぞれ 2種類の目盛り (xticsと x2tics, yticsと y2tics)を持つのに対し,

3次元グラフでは各座標軸にそれぞれ 1種類の目盛りしか存在しないことを注意しておきます.

6.2 グラフを調整する – setとwith –

グラフはいろいろな目的で作成されますし, 利用者が作成したいグラフの形状は目的によって大きく異なりま

す. このようないろいろな利用者の要求に答えるために, gnuplotにはグラフを調整する様々なオプションが用

意されています. これらのオプションを設定するには,

• キーワード with に続いてオプションを指定する

• コマンド set を用いて設定する

という 2種類の方法のいずれかを用います.

コマンド set を用いて設定された事項を確認するときには, コマンド show を使います.

6.2.1 キーワード with を使う

キーワード with を使うと,

• グラフのスタイル

• 線の種類

• 線の太さ

• 点の種類

• 点の大きさ

を指定することができます. グラフのスタイルに関する説明は後回しにして, まず, 線の種類, 線の太さ, 点

の種類, 点の大きさについて説明しておきます.

線の種類, 線の太さ, 点の種類, 点の大きさを指定するときには, 表 6.3に示したようなキーワードを使いま

す. これらを指定するときには, たとえば

表 6.3: 線と点を指定するキーワード

指定する事項 キーワード 省略記法

線の種類 linetype lt

線の太さ linewidth lw

点の種類 pointtype pt

点の大きさ pointsize ps

plot sin(x) with lines linetype 3 linewidth 10 [Enter]

のようにキーワード with に続いて必要なキーワードおよびそのキーワードに対応した数値を入力してゆくので

すが, ここでひとつ注意すべきことがあります. それは何かというと,

指定するキーワードの順番は決まっている

ということです. 具体的に言うと, 表 6.3のキーワードを指定するときには, 表 6.3と同じ順番を守らなけれ

ばなりません. 指定する必要のないキーワードを飛ばすことはできるのですが, 順番を入れ換えることはできま

せん. ですから, たとえば

plot sin(x) with lines linewidth 10 linetype 3 [Enter]

と入力するとエラーメッセージが出ます. 次に, 先ほど予告したグラフのスタイルについて説明しましょう.

gnuplotでは, 表 6.4に示されているようなグラフのスタイルが使えます. ただし, これらの中には, 2次元

グラフを描画するときのみ有効なものがあります.

6.2.2 コマンド set と show を使う

コマンド set を使うと, 作図にかかわるいろいろな事項を設定することができます. コマンド set を用い

て設定された内容を確認するには, コマンド show を用います.

コマンド set を使うときには,

set 事項の名前 設定の内容 [Enter]

表 6.4: グラフのスタイル

スタイル名 機能

lines データ点のあいだを線で結ぶ (データ点には小さい点を打つ).

points データ点に適切な図記号を配置する. データ点のあいだには線を引かない.

linespoints データ点に適切な図記号を配置し, データ点のあいだを線で結ぶ.

impulses 各データ点について, x軸からデータ点まで垂直に伸びる直線を引く.

dots データ点に小さい点を打つ.

steps グラフを階段関数で表示する.

errorbars y軸にエラーバーを表示する.

xerrorbar x軸にエラーバーを表示する.

xyerrorbars x軸と y軸にエラーバーを表示する.

boxes 各データ点について, x軸からデータ点まで垂直に伸びる長方形を描く.

boxerrorbars 各データ点について, x軸からデータ点まで垂直に伸びる長方形を描き, y軸に
エラーバーを付ける.

boxxyerrorbars 各測定点について, データ点の値とその誤差 (x, y, xδ, yδ) が与えられている
ときには, 4 個の頂点 (x − xδ, y − yδ), (x − xδ, y + yδ), (x + xδ, y − yδ),
(x + xδ, y + yδ) を持つ長方形を描画する. また, データ点の値とその最小値,
最大値 (x, y, xmin, xmax, ymin, ymax) が与えられているときには, 4 個の頂点
(xmin, ymin), (xmin, ymax), (xmax, ymin), (xmax, ymax) を持つ長方形を描画す
る.

のように, コマンド set に続いていろいろな文字列を入力します. 以下に, コマンド set のあとに指定する文

字列を作業内容に応じて分類したものを表にして示します. 表 6.5には作図全般に関連したコマンドが, 表 6.6

には 3次元グラフに関連したコマンドが, 表 6.7には座標系 (曲線座標など)や目盛りの表示に関連したコマンド

が, 表 6.8には座標軸の表示に関連したコマンドがまとめられています.

ここで, 表の見方について説明しておきましょう.

まず, これらのコマンドはすべてコマンド set に続いて指定されます.

次に, 表中で

[no]border

などのように各括弧付きで表示されている部分は, 各括弧で囲われた部分を書くときと書かないときで機能が切

り換わることを意味します. たとえば, 上の例は set border というコマンドと set noborder というコマン

ドを同時にあらわしていて, 前者が枠を表示するコマンド, 後者が座標軸を表示しないコマンドになります.

さいごに, 表中で<>で囲まれた部分は, そのパラメータが省略可能であることを意味しています. また,

A OR B

という記法を「キーワード Aか Bかのいずれか」という意味で使っている部分があります. たとえば, 表 6.5の

mxgrid OR mygridという記述は, 「キーワード mxgrid と xygrid のいずれか一方を指定する」という意味に

なります.

これらのコマンドの具体的な使い方については次節以降を参照して下さい.

表 6.5: 作図全般にかかわるコマンド一覧

機能 コマンド

関数のプロットのスタイルを設定 function style <style-choice>

図を囲む枠の表示切り換え [no]border

図の端の点や直線などを切り抜き表示の制御 [no]clip <clip-type>

データファイルのプロットのスタイルを設定 data style <style-choice>

媒介変数を指定する dummy <dummy1>,<dummy2>...

角度の単位を設定する angles [degrees|radians]

始点から終点まで矢印を引く (すでにグラフが
描画されているときに有効)

arrow [<tag>][from <sx>,<sy>,<sz>]

[to <ex>,<ey>,<ez>][nohead]

グラフ中央からのオフセットの設定 offsets<left>,<right>,<top>,<bottom>

関数の標本点 (プロットに使う点)の数を指定 samples <expression>

図の縮尺を指定 size <xsize>,<ysize>

角度の範囲を設定 rrange [<rmin>:<rmax>]

端末切り換え terminal <device>

図に作成日時を入れるかどうかを指定 [no]time

図の標題を指定 title "title-text" <xoff>,<yoff>

表 6.6: 3次元グラフィックスに関連したコマンド一覧

機能 コマンド

等高線の制御 cntrparam [spline][points][order][levels]

等高線の描画 [no]contour [base|surface|both]

隠れ面の表示切り換え [no]hidden3d

グラフにかかった網目のきめを設定する isosamples <expression>

水平面の位置を調整する ticslevel <level>

視点を指定 view <rot_x>,<rot_z>,<scale>,<scale_z>

グラフの網かけの表示の切り換え [no]surface

曲線座標との切り換え mapping[cartesian|spherical|cylindrical]

パラメータ付き曲面の範囲指定 urange OR vrange

表 6.7: 座標系や目盛りの制御などに関連したコマンド一覧

機能 コマンド

パラメータ付き曲線モードのオン, オフ [no]parametric

グラフ中に方眼目盛りを描く [no]grid [mxgrid OR mygrid]

目盛りの数値のフォーマットを指定 format [<axes>]["format-string"]

パラメータ付き曲線のパラメータ範囲指定 trange [<tmin>:<tmax>]

座標軸の範囲を自動設定する autoscale [<axes>]

キーの表示切り換え key <x>,<y>,<z>

座標軸を対数目盛りにする logscale <axes> [<base>]

2次元極座標との切り換え [no]polar

目盛りの切り方の指定 tics <direction>

目盛り記号の大きさを調整する ticscale [<size>]

表 6.8: 座標軸の調整に関連したコマンド一覧

機能 コマンド

x軸のラベルを設定 xlabel "<label>" <xoff>,<yoff>

x軸の描画範囲を指定 xrange [<xmin>:<xmax>]

x軸の目盛りを変更 xtics <start>,<incr>,<end>,"<label>" <pos>

x軸の目盛りの鏡像 (通常上)の表示切り換え xtics [no]mirror

x軸の副目盛りの調整 [no]mxtics OR [no]mytics [<freq>]

x軸の表示切り換え [no]xzeroaxis

y軸のラベルを設定 ylabel "<label>" <xoff>,<yoff>

y軸の描画範囲を指定 yrange [<ymin>:<ymax>]

y軸の目盛りを変更 ytics <start>,<incr>,<end>,"<label>" <pos>

y軸の目盛りの鏡像 (通常右)の表示切り換え ytics [no]mirror

y軸の表示切り換え [no]yzeroaxis

零付近のしきい値の設定 zero <expression>

座標軸の描画切り換え [no]zeroaxis

z軸のラベルを設定 zlabel "<label>" <xoff>,<yoff>

z軸の描画範囲を指定 zrange [<zmin>:<zmax>]

z軸の目盛りを変更 ztics <start>,<incr>,<end>,"<label>" <pos>

z軸の表示切り換え [no]zzeroaxis

46



第7章 2次元グラフを描画する

7.1 関数のグラフのプロット

7.1.1 簡単な 2次元グラフを描画する

簡単な 2次元グラフを描画するには,

plot 関数名 [Enter]

と入力します. 関数のデフォルトの変数 (引数)は x です.

たとえば, 正弦関数 sin(x) を描画するときには,

plot sin(x) [Enter]

と入力します. すると, 図の x軸と y軸が自動的に調整され, 正弦関数のグラフが描画されます.

プロットが終わったグラフを eps ファイルの形式で保存するには,

set terminal postscript eps [Enter]

set output ’ファイル’ [Enter]

replot [Enter]

とします. ここに, 「ファイル」と書かれた部分には適当なファイル名が入ります.

7.1.2 y軸や x軸の範囲を変える

gnuplotではグラフの x軸や y軸の範囲の初期値は自動的に設定されます. しかし, 標準的な設定で適切な

グラフが得られないときには, これらを自分で設定し直す必要があります.

このような場合には, コマンド plot に続けて x軸と y軸の範囲を指定します. y軸, x軸のそれぞれについ

てそれぞれ

• 指定なし

• 下限のみ指定

• 上限のみ指定

• 上限と下限を指定

の 4種類の指定ができます. 指定されていない部分は自動的に調整されます. 以下にいくつか例を示します.

これとは別に, グラフを描く前に描画範囲を指定したり描画範囲の自動調整の有無を切り換えたりするコマン

ドもあります. これらについて表 7.2にまとめておきます.

表 7.1: 描画範囲の設定例

機能 コマンド

x軸, y軸とも指定しない plot exp(x) [Enter]

x軸の上限のみ指定, y軸は指定なし plot [:9] exp(x) [Enter]

x軸の下限のみ指定, y軸は指定なし plot [1:] exp(x) [Enter]

x軸の上限と下限を指定, y軸は指定なし plot [1:9] exp(x) [Enter]

x軸は指定なし, y軸の上限と下限を指定 plot [] [1:4] exp(x) [Enter]

x軸と y軸の上限と下限を指定 plot [1:9] [1:4] exp(x) [Enter]

表 7.2: 描画範囲の調整コマンド

機能 コマンド

x軸の範囲を指定 set xrange [x0:x1] [Enter]

x0と x1の部分には初期値と終了値 (数値)を指定する; 片方を省略
することもできる

y軸の範囲を指定 set yrange [y0:y1] [Enter]

y0と y1の部分には初期値と終了値 (数値)を指定する; 片方を省略
することもできる

全座標軸を自動調整する set autoscale [Enter]

全座標軸を自動調整しない set noautoscale [Enter]

x軸を自動調整する set autoscale x [Enter]

x軸を自動調整しない set noautoscale x [Enter]

y軸を自動調整する set autoscale y [Enter]

y軸を自動調整しない set nonoautoscale x [Enter]

x軸と y軸を自動調整する set autoscale xy [Enter]

x軸と y軸を自動調整しない set noautoscale xy [Enter]

7.1.3 y軸と x軸にラベルをつける

x軸や y軸にラベルをつけるときには, set xlabelおよび set ylabel というコマンドを使います. x軸の

ラベルを設定するには,

set xlabel ’ラベルの名前’ [Enter]

とします. 同様に, y軸のラベルを設定するには,

set ylabel ’ラベルの名前’ [Enter]

のようにします.

以下に, x軸に「時間」, y軸に「人口」というラベルを付けて指数関数を描画する例を示します.

set xlabel ’時間’ [Enter]

set ylabel ’人口’ [Enter]

plot exp(x) [Enter]

結果は図 7.1のようになります.

ラベルを付けたくないときには, set nolabel, set noxlabel, set noylabel というコマンドが使えます.

これらについて表 7.3にまとめておきます.

図 7.1: y軸と x軸のラベルを指定したグラフ

表 7.3: ラベル関連のコマンド

機能 コマンド

x軸のラベルを設定 set xlabel ’ラベル’ [Enter]

y軸のラベルを設定 set ylabel ’ラベル’ [Enter]

ラベルをすべて表示する set label [Enter]

ラベルをすべて表示しない set nolabel [Enter]

x軸のラベルを表示する set xlabel [Enter]

x軸のラベルを表示しない set noxlabel [Enter]

y軸のラベルを表示する set ylabel [Enter]

y軸のラベルを表示しない set noylabel [Enter]

7.1.4 関数のグラフの線のスタイルを変える

関数のグラフを表示するときには, 線以外にもいろいろなものが使えます. ここでは, これらについて順に説

明してゆきます.

7.1.4.1 グラフを点で表示

グラフを線でむすぶのをやめて, かわりに点で表示するには,

plot 関数 with points [Enter]

のようにします.

以下に三角関数を点を使って描画している例を示します.

plot sin(x) with points [Enter]

結果は図 7.2のようになります.

7.1.4.2 線と点を重ねる

線と点を重ねることもできます. このような場合には,

plot 関数 with linespoints [Enter]

のようにします.

以下に例を示します.

plot sin(x) with linespoints [Enter]

結果は図 7.3のようになります.

-1

-0.8

-0.6

-0.4

-0.2

0

0.2

0.4

0.6

0.8

1

-10 -5 0 5 10

sin(x)

図 7.2: 点を使ってプロット

図 7.3: 点と線でプロット

7.1.4.3 インパルスで表示

グラフを線や点で表示するかわりに, x軸から伸びるインパルスで関数を表示することもできます. このよう

な場合には,

plot 関数 with impulses [Enter]

のようにします.

以下に例を示します.

plot sin(x) with impulses [Enter]

結果は図 7.4のようになります.

図 7.4: インパルスでプロット

7.1.4.4 階段グラフ

階段グラフを作ることもできます. このような場合には,

plot 関数 with steps [Enter]

のようにします.

以下に例を示します.

plot sin(x) with steps [Enter]

結果は図 7.5のようになります.

x軸から伸びる長方形のボックスで関数を表示するには,

plot 関数 with boxes [Enter]

のようにします.

以下に例を示します.

plot sin(x) with boxes [Enter]

結果は図 7.6のようになります.

階段グラフには, steps, fsteps, histepsの 3種類があり, それぞれ挙動が違います. 2個の点 (x1, y1)と
(x2, y2)が与えられているとき, steps, fsteps, histepsはそれぞれ

steps (x1, y1)から (x2, y1)への水平な線と (x2, y1)から (x2, y2)への垂直な線を引く

fsteps (x1, y1)から (x1, y2)への垂直な線と (x1, y2)から (x2, y2)への水平な線を引く

histeps ((x0 + x1)/2, y1)から点 ((x1 + x2)/2, y1)への水平な線と ((x1 + x2)/2, y1)から点 ((x1 + x2)/2, y2)へ
の垂直な線を引く

という挙動を示します. 図 7.7のようなデータファイル (ファイル名を steps.dat とします)が与えられたと

きの steps, fsteps, histepsの挙動をそれぞれ図 7.8, 図 7.9および図 7.1.4.4 に示します. 挙動の違い

を見やすくするために, データ点を大きく表示してあります.

図 7.5: 階段グラフでプロット

図 7.6: x軸から伸びる長方形の箱でプロット

7.1.4.5 線の太さを変える

グラフの線の太さを変えるには

plot 関数 with lines linewidth 数値

のように, キーワード linewidth のあとに適当な値 (整数)に指定します.

以下に線の太さを通常の 10倍にしてる例を示します.

plot sin(x) with lines linewidth 10 [Enter]

結果は図 7.11のようになります.

なお, linewidthの省略形である lw というキーワードも使えます. ですから,

plot sin(x) with lines lw 10 [Enter]

図 7.7: データファイル steps.dat の内容

図 7.9: fstepsの挙動

図 7.10: histepsの挙動

図 7.11: 線の太さを 10倍にしたもの

としても同じ結果が得られます.

7.1.4.6 線の種類を変える

線の種類を変えるには,

plot 関数 with lines linetype 数値 [Enter]

linetype のあとに適当な数値を指定します.

以下に例を示します.

plot sin(x) with lines linetype 4 [Enter]

結果は図 7.12のようになります. なお, linetypeの省略形である lt というキーワードも使えます. で

すから,

plot sin(x) with lines lt 4 [Enter]

としても同じ結果が得られます.

上記の操作をおこなうと, 画面にはマゼンタ色の線が表示されるのですが, 白黒の PostScriptファイルに保

存するとこの線は点線で置き換えられます.

一般に, 線の種類を変更した場合には, 白黒のデータを作成すると, 線を色で区別することができなくなるの

で, かわりにいろいろな点線が使われます. 結果として, 画面に見えている線の見掛けとファイルに保存した線

の見掛けは変わります.

図 7.13に利用可能な線の種類の一覧を示します. 線の種類は 1から 8までの 8種類が指定でき, それぞれ X

Window System の画面では赤, 緑, 青, マゼンタ (赤紫), シアン, シエナ (赤茶色), 橙, コラル (珊瑚色)で

表示されますが, 白黒の PostScriptで保存すると図 7.13に示したような実線や点線に変わります. なお, 他

の色が使える形式 (たとえば tgif)で保存した場合, 表示される色が X Window System のそれと一致しないこ

とがあります.

図 7.12: 線の種類を変更したもの

図 7.13: いろいろな線の種類

7.1.4.7 点の種類を変える

グラフの点の種類を変えるには

plot 関数 with points pointtype 数値

のように, キーワード pointtype のあとに適当な値 (整数)に指定します.

以下に例を示します.

plot sin(x) with points pointtype 2 [Enter]

結果は図 7.14のようになります.

なお, pointtypeの省略形である pt というキーワードも使えます. ですから,

plot sin(x) with points pt 2 [Enter]

としても同じ結果が得られます.

点の種類を変えた場合も, 画面に表示される点の種類とグラフを保存したときにそのファイルで使われる点の

種類は一致しないことがあるので, 注意が必要です. 指定可能な 75種類の点の X Window System 上での見え

方図 7.15に, 対応する PostScriptファイルにおける点の見え方を図 7.16に示します. いずれも, 点の番号

は, 図の右端で 0から始まり, 反時計回りに 75まで増加してゆきます.

図 7.14: 点の種類を変更したもの

図 7.15: いろいろな点の見え方 (X Window System上の表示)

図 7.16: いろいろな点の見え方 (PostScriptファイルの表示)

7.1.4.8 点の大きさを変える

グラフの点の大きさを変えるには

plot 関数 with points pointsize 数値

のように, キーワード pointsize のあとに適当な値 (整数)に指定します. 以下に例を示します.

plot sin(x) with points pointsize 10 [Enter]

なお, pointtypeの省略形である ps というキーワードも使えます. ですから,

plot sin(x) with points ps 2 [Enter]

としても同じ結果が得られます. 出力例は省略します.

点を使ってプロットしている場合は, 点の種類と点の大きさの指定を併用できます. このような場合には,

plot 関数 with points pt 数値 ps 数値 [Enter]

のようにします.

7.1.4.9 点と線の種類を同時に変える

線と点を両方使ってプロットしている場合には,

plot 関数 with linespoints lw 数値 pt 数値 [Enter]

とすれば, 線の太さと点の種類を両方変更できます.

7.1.5 描画するときのサンプル点の間隔を調整する

さて, ここで

plot [0:5] sin(exp(x)) [Enter]

と入力してみます. 表示されたグラフは図 7.17のようになります. 本来はこの関数の値は 0と 1のあいだで

振動するはずなのですが, 変なふうに値が欠落してしまっています. 何が悪いのでしょうか?

gnuplot は, x軸のいくつかの点について対応する y軸の値を計算して, そのあいだをなめらかな線で結ぶ,

ということを実行しています.

plot 関数 with points [Enter]

としたときに表示される点の数が, 標準で計算される点の数です. この点の数が少な過ぎると, グラフが欠けて

しまいます.

x軸に取る点の数を指定するには,

set samples 数値 [Enter]

と入力します.

x軸に取る点の数を 10000個に変更してからグラフを書き直している例を以下に示します.

set samples 10000 [Enter]

plot [0:7] sin(exp(x)) [Enter]

こんどは, 表示されたグラフは図 7.18のようになり, データの欠落はなくなります.

図 7.17: サンプルが少なすぎるグラフ

図 7.18: サンプルが十分にあるグラフ

7.1.6 キー (関数など説明)の操作

gnuplotで図を作成すると, 標準では図の右上にグラフと同じスタイルの水平な線が表示され, その左側にグ

ラフの説明として関数の名前が表示されます. この部分をキー (key)といいます. 本項ではこの部分を変更す

る方法を説明します.

7.1.6.1 グラフを説明するテキストを変更する

線の横の関数名が表示されている部分を変更するには, title というキーワードを使います.

plot 関数 title ’グラフの説明’ [Enter]

とすると, 関数名が表示されていた部分が「’’」で囲まれた文字列で置き換えられます.

たとえば,

plot sin(x) title ’正弦関数’ [Enter]

のようにすると, 図 7.19のように図の右上に「正弦関数」という文字が入ります.

この部分に数式などを入れることもできます. 数式などを入れる方法については, 10章を参照して下さい.

7.1.6.2 キーのスタイルと位置の制御

キーを制御するには, set key というコマンドに続いていろいろなオプションを指定してゆきます.

図 7.19: 線の説明付きのグラフ

まず, 基本的なものとして, 表 7.4に 3個のコマンドを挙げておきます.

表 7.4: グラフの説明の制御 (基本)

機能 コマンド

キーを表示する set key

キーを表示しない set nokey

キーの状態を見る show key

次に, キーの表示場所や表示のされかたを変更するためのコマンド set key のさまざまなオプションを表 7.5

にまとめておきます. 表 7.5のオプションはすべてコマンド set key のあとに続けて指定されます. たとえ

ば, 表 7.5の最初の「キーの場所を左上にする」という機能を使うためには,

set key left top [Enter]

と入力します.

表 7.5において, 文字が斜体になっている部分には, 適当な数値が入ります. 特に断ってある場合を除き, 正

の数と負の数のいずれも指定可能です.

また, 表 7.5のオプションの中で種類が違うものについてはいくつでも組み合わせて指定することができます.

表 7.5では, オプションの種類が変わるごとに横線を引いて区別してあります.

線の説明を箱で囲むときには, さらに囲む線のスタイルなどを変更することもできます. 指定の仕方は 7.1.4

と同じです.

7.1.7 図の標題の制御

図に標題をつけるには,

set title ’図の標題’

とします. 上で「図の標題」と書いてある部分に適当なグラフの題目を指定すると, その題目が図の上側に表示

されます.

ただし, 図の標題を図の上に入れることは, 技術文書の書き方としては, 必ずしも標準的ではありません. た

とえば, レポートを作成するときに, 図の標題を必ず図の下に入れるように指示されることもあります.

ですから, gnuplotを使って図に標題をつけるよりは, gnuplotで作図するときには標題をつけずにおいて,

他の文書作成ソフトウエアを使って図入りの文書を作成するときに改めて標題をつけ直す方がよいでしょう.

表 7.5: コマンド set key のさまざまなオプション

機能 オプション

キーの場所を左上にする left top

キーの場所を左下にする left bottom

キーの場所を右上にする right top

キーの場所を右下にする right bottom

キーの場所を画面外上段にする top outside

キーの場所を画面外下段にする bottom outside

キーの場所をグラフの下側にする below

キーが表示される場所の座標 xと yを指定する x, y

線の見本の長さにm文字分の長さを追加する samplen m

説明を線の右側に付ける noreverse

説明を線の左側に付ける reverse

線の説明を箱で囲む box

線の説明を箱で囲まない nobox

文字が表示される部分の前側にm文字分の空白を追加する width w

グラフが複数あるとき (7.2節参照), 説明の文字を左にそろえる Left

グラフが複数あるとき説明の文字を右にそろえる Right

グラフが複数あるとき説明のあいだの空白の幅を標準のm倍にする (m > 0) spacing m

キー全体に標題をつける title ’キーの標題’

7.1.8 詳細な目盛りの制御

7.1.8.1 目盛りの表示

x軸の目盛りを切り直すには,

set xtics 初期値, 増分, 終了値 [Enter]

とします. 同様に, y軸の目盛りを等間隔で切り直すには,

set ytics 初期値, 増分, 終了値 [Enter]

とします.

gnuplotは, 標準では, x軸についてはグラフの上下に, y軸についてはグラフの左右に目盛りを打ちます.

この表示切り換え (両方に目盛りを打つか片方だけにするか) には, [no]mirrorというキーワードを使います.

たとえば, x軸に関し, 上下双方に目盛りを打つのをやめたいときには,

set xtics nomirror [Enter]

と入力し, 逆に上下双方に目盛りを打つようにしたいときには

set xtics mirror [Enter]

とします. y軸に関しても同様で, 左右双方に目盛りを打つのをやめたいときには,

set ytics nomirror [Enter]

と入力し, 逆に左右双方に目盛りを打つようにしたいときには

set ytics mirror [Enter]

とします.

目盛りを不等間隔で切りたいときには, ( )内に目盛りを打つ場所の座標を指定してゆきます. たとえば, グ

ラフ下側の x軸の座標 1, 4, 9.16という点に目盛りを打ちたいときには,

set xtics (1,4,9,16) [Enter]

とします. 同様に, グラフ左側の y軸の座標 1, 4, 9.16という点に目盛りを打ちたいときには,

set ytics (1,4,9,16) [Enter]

とします.

目盛りを表示するかしないかを切り換えることもできます. たとえば, 下の目盛りを表示するときには

set xticsk [Enter]

とし, たとえば, 下の目盛りを表示しないときには,

set noxtics [Enter]

とします.

ひとつの目盛りの中により細かい副目盛りを刻むこともできます. このためには,

set mxtics 数 [Enter]

set mytics 数 [Enter]

のように指定します. 「数」と書かれた部分には, ひと目盛りの中にいくつ小さい目盛りを打つかを整数で指定

します.

目盛りの表示/非表示の切り換えに関連したコマンドを表 7.6にまとめておきます.

表 7.6: 目盛りの表示, 非表示関連のコマンド

機能 コマンド

x軸の目盛りを表示 set xtics [Enter]

x軸の目盛りを非表示 set noxtics [Enter]

x軸の目盛りを上下の片方だけに set xtics nomirror [Enter]

x軸の目盛りを上下の両方に set xtics mirror [Enter]

y軸の目盛りを表示 set ytics [Enter]

y軸の目盛りを非表示 set noytics [Enter]

y軸の目盛りを左右の片方だけに set ytics nomirror [Enter]

y軸の目盛りを左右の両方に set ytics mirror [Enter]

目盛りの全状態を報告 show tics [Enter]

7.1.8.2 目盛りの数値の書式を変更する

座標軸の目盛りに表示される数値の書式を変更するには,

set format 座標軸 "書式" [Enter]

という構文を使います. 「座標軸」と書かれた部分にはなにも書かれないか, x,y,xy,x2,y2 のいずれかを入れ

ます. 座標軸を指定しないと, 指定した書式が x軸と y軸に適用されます. x2,y2の意味についてはひとつ上

の項を参照して下さい.

「書式」の部分には C言語の浮動小数点数の書式指定と同じ様式で書式を指定します. 具体的には,

%数.数 e

%数.数 f

%g

などのように, 文字%のあとに e, f, gの 3文字のいずれかを書き, そのあとに必要に応じて 2個の数を小数点

で区切ってならべます. 2個の数を書いた場合, 小数点の左側が全桁数, 小数点の右側が小数点以下の桁数をあ

らわします. これらは片方あるいは両方とも省略することができます. 文字 e, f,gの意味については表 7.7

にまとめておきます. gnuplotの標準の書式は g です.

表 7.7: 書式指定で使える文字

意味 文字

指数表示 e
ふつうの小数表示 f
自動 g

なお, 現在の目盛りの数値の書式を確認したいときには,

show format

とします.

いくつか例を示しましょう.

表 7.8: 書式切り換えの例

機能 コマンド

全座標軸に書式 gを設定 set format "%g" [Enter]

x軸に書式 fを設定, 全 2桁, 小数点以下 1桁 set format x "%2.1f" [Enter]

y軸に書式 eを設定, 全 2桁, 小数点以下 1桁 set format y "%2.1e" [Enter]

x軸と y軸に書式 fを設定, 全 4桁, 小数点以下 2桁 set format xy "%4.2f" [Enter]

現在の目盛りの数値の書式を確認 show format [Enter]

7.1.8.3 副目盛りの表示と目盛りの調整

gnuplotは, グラフの x軸や y軸に, 定規についているような主目盛りよりも細かい副目盛りを打つことがで

きます.

たとえば, x軸に副目盛りを打ちたいときには,

set mxtics [Enter]

とします. また, x軸の副目盛りを消したいときには,

set nomxtics [Enter]

と入力します.

副目盛りはデフォルトでは 1個の主目盛を 10等分するように打たれます. これを変更するには,

set mxtics 数 [Enter]

と入力します. ただし, 「数」と書かれた部分では, 1個の主目盛りを何等分するかを指定します. x2tics,

yticsなどについても同様です.

gnuplotはデフォルトでは目盛りをグラフ全体を囲むボックスの内側に表示しますが, この目盛りの表示位置

をボックスの外側に変更することもできます. 目盛りの表示位置をグラフの外側にするには

set tics out [Enter]

と入力し, 目盛りの表示位置をグラフの内側にするには

set tics in [Enter]

と入力します.

目盛りの制御に関連したコマンドを表 7.9にまとめておきます. なお, 表中で「[数]」と書かれている部分

では, 「数」という部分には適当な数値が入ります. また, 記号「[]」はその部分の記述が省略可能であること

を意味しています. たとえば, 表 7.9の 1段目の記述は, set mxticsというコマンドが

set mxtics [Enter]

あるいは

set mxtics 10 [Enter]

などのようにして使われることを意味しています.

表 7.9: グラフの大きさの指定

機能 コマンド

x軸の副目盛りを表示, 設定 set mxtics [数] [Enter]

x軸の副目盛りを非表示 set nomxtics [Enter]

x軸の第 2副目盛りを表示, 設定 set mx2tics [数] [Enter]

x軸の第 2副目盛りを非表示 set nomx2tics [Enter]

y軸の目盛りを表示,設定 set mytics [数] [Enter]

y軸の目盛りを非表示 set nomytics [Enter]

y軸の第 2副目盛りを表示, 設定 set my2tics [数] [Enter]

y軸の第 2副目盛りを非表示 set nomy2tics [Enter]

z軸の目盛りを表示, 設定 set mztics [数] [Enter]

z軸の目盛りを非表示 set nomztics [数] [Enter]

目盛りをボックスの内側にする set tics in [Enter]

目盛りをボックスの外側にする set tics out [Enter]

目盛りの状態を表示 show tics [Enter]

7.1.8.4 x軸や y軸の目盛りを曜日で表示する

gnuplotには, x軸や y軸の目盛りの表示を曜日に切り換える機能があります. x軸の目盛りを曜日に切り換

えるには

set xdtics [Enter]

とします.

目盛り表示が曜日になっているときには, x 軸や y 軸の読みが 0 のときは日曜日 (Sun), 1 のときは月曜日

(Mon),..., 6のときは土曜日 (Sat)と解釈されます. 7になると日曜日に戻ります.

例として,

set xdtics [Enter]

set ydtics [Enter]

plot [0:13] sin(x) [Enter]

とした場合の出力を図 7.20に示します.

図 7.20: 横軸を曜日表示にしたグラフ

表 7.10: 座標軸の読みを曜日に切り換えるコマンド一覧

機能 コマンド

x軸の読みを曜日表示にする set xdtics

x軸の読みの曜日表示をやめる set notdtics

x軸の読みの状態を表示する show xdtics

y軸の読みを曜日表示にする set ydtics

y軸の読みの曜日表示をやめる set notdtics

y軸の読みの状態を表示する show ydtics

7.1.9 座標軸の対数目盛りと通常の目盛りの切り換え

x軸と y軸のそれぞれについて, 座標軸を対数目盛りに変えたり, ふつうの目盛りに戻したりできます. 表

7.11 にこれらのコマンドをまとめておきます.

表 7.11: 対数目盛りの切り換えコマンド

機能 コマンド

座標軸を対数目盛りにする set logscale [Enter]

x軸と y軸の両方の対数目盛りを解除する set nologscale [Enter]

x軸のみ対数目盛りに変える set logscale x [Enter]

x軸のみ対数目盛りをやめる set nologscale x [Enter]

y軸のみ対数目盛りに変える set logscale y [Enter]

y軸のみ対数目盛りをやめる set nologscale y [Enter]

7.1.10 枠や座標軸, グリッドの表示と非表示

枠 (グラフを囲う長方形)や座標軸, グリッド (格子)の表示と非表示を切り換えるコマンドを以下にまとめま

す. 座標軸を表示する設定に変更すると, 点線で座標軸が表示されるようになります. x軸と y軸の一方だけ

を表示することもできます.

表 7.12: 枠や座標軸の表示切り換えコマンド

機能 コマンド

枠の表示 set border [Enter]

枠の非表示 set noborder [Enter]

座標軸の表示 set zeroaxis [Enter]

座標軸の非表示 set nozeroaxis [Enter]

グリッドの表示 set grid [Enter]

グリッドの非表示 set nogrid [Enter]

x軸の表示 set xzeroaxis [Enter]

x軸の非表示 set noxzeroaxis [Enter]

y軸の表示 set yzeroaxis [Enter]

y軸の非表示 set noyzeroaxis [Enter]

7.1.11 グラフの大きさの指定

グラフの大きさを指定するときには,

set size

というコマンドを使います. 指定する数値は, グラフ表示用のウィンドウから見た相対的な数値になります. 指

定できる内容は,

• グラフを正方形にする

• グラフの縦横の比率を指定する

• グラフの横の単位長さと縦の単位長さの比率を指定する

• グラフの表示比率を標準に戻す

表 7.13: グラフの大きさの指定

機能 コマンド

描画範囲を指定する set size rx, ry [Enter]

rx, ry は 0以上 1未満の数で, それぞれウィンドウの横幅とグ
ラフの横幅, ウィンドウの高さとグラフの高さの比率をあらわ
す

グラフを正方形にする set size square [Enter]

グラフの正方形表示をやめる set size nosquare [Enter]

グラフの横縦の比率を rにする set size ratio r [Enter]

y軸の単位長さを x軸の r倍にする set size ratio −r [Enter]

比率の指定をやめる set size noratio [Enter]

現在の大きさ指定を表示する show size [Enter]

の 4種類です. これらの使い方を表 7.13に示します. なお, rx, ry と rの部分には適当な数値が入ります.

「縦横の比率を指定する」ことと「横の単位長さと縦の単位長さの比率を指定する」の違いに注意して下さい.

前者ではグラフの見た目が指定されているのに対し, 後者では y軸の単位長さを x軸の単位長さの何倍に取るか

が指定されているだけで, グラフの見た目は指定されていません.

比較のために,

set size ratio 1 [Enter]

plot [0:2] sin(x) [Enter]

というコマンドを実行した結果を図 7.21に,

set size ratio -1 [Enter]

plot [0:2] sin(x) [Enter]

というコマンドを実行した結果を図 7.22に示します.

図 7.21: 縦横の比率を指定

図 7.22: 横の単位長さと縦の単位長さの比率
を指定

7.1.12 グラフの起点の指定

画面に描画されるグラフの起点 (左端の点)を指定するには, set origin というコマンドを使います.

具体的な指定の仕方は,

set origin x,y [Enter]

です. ここに, x,yと書かれた部分には適当な数値が入ります. コマンド set origin は グラフの起点 (左端

の点)をグラフが表示されるウィンドウから見た相対的な位置で指定するので, xと yは 0以上 1以下の値でなけ

ればなりません.

たとえば,

set origin 0.5,0.5 [Enter]

とすると, グラフを描き始める点がグラフが表示されるウィンドウの中央になります.

set origin 0.5,0.5 [Enter]

plot sin(x) [Enter]

とした場合に X Window Systemの画面に表示されるグラフの例を図 7.23 に示します.

図 7.23: 原点を指定したグラフ

なお, たとえば図 7.23のようなグラフを描いてからそのグラフを PostScript形式で保存した場合には, 原点

の情報は失われます. 図 7.23のような不完全なグラフがそのまま保存されることはありません.

7.1.13 グラフ上下左右の余白の調整

グラフの上下左右の余白を調整するためには, たとえば

set lmargin

というコマンドを使います. また, 現在の空白の設定を見るためには,

show margin

というコマンドを使います. これらを表 7.14にまとめておきます. なお, 表 7.14で「数」と書かれた部分に

は適当な数値が入ります. 値が大きいほど余白が大きくなります. グラフを PostScript形式で保存した場合

にも, 余白の情報はそのまま保存されます.

表 7.14: グラフの大きさの指定に関連したコマンド

機能 コマンド

左の欄外の幅を設定 set lmargin 数

右の欄外の幅を設定 set rmargin 数

上の欄外の幅を設定 set tmargin 数

下の欄外の幅を設定 set bmargin 数

欄外の設定を表示 show margin

7.1.14 グラフとボックスの間隔の調整

グラフをボックスで囲むときには, グラフとボックスのあいだの間隔を調整することができます. この間隔は

デフォルトでは 0になっていて, グラフの描画範囲に一致するボックスが生成されるようになっています.

グラフとボックスのあいだの間隔を調整するには, set offsets というコマンドに続いてグラフの左, 右, 上,

下の空白の大きさをコンマで区切って指定します.

set offsets 0,1,2,4 [Enter]

plot sin(x) [Enter]

と入力したときに描画されるグラフを図 7.24に示します.

図 7.24: グラフとボックスのあいだの間隔の調整

表 7.15: グラフとボックスのあいだの間隔の調整に関連したコマンド

機能 コマンド

グラフとボックスのあいだの間隔を指定 set offsets 左, 右, 上, 下 [Enter]

グラフとボックスのあいだの間隔をなくす set nooffsets [Enter]

現在のオフセットの情報を表示 show offsets [Enter]

7.1.15 グラフに作成日時を入れる

gnuplotには作成したグラフに作成日時を入れる機能があります. これらのコマンドを表 7.16にまとめてお

きます.

set timestamp [Enter]

と入力してからグラフを描くと, グラフの左下に作成日時が入るようになります.

表 7.16: 作成日時に関連したコマンド一覧

機能 コマンド

グラフに作成日時を入れる set timestamp [Enter]

グラフに作成日時を入れるのをやめる set notimestamp [Enter]

作成日時の状態を表示する show timestamp [Enter]

7.1.16 グラフィックスウィンドウに描画されたグラフを消す

グラフィックスウィンドウに描画されたグラフを消す (クリアする)には,

clear [Enter]

と入力します. グラフィックスウィンドウそのものは消えません. このコマンドは第 7.3節で述べる 1個のウィ

ンドウ複数のグラフを重ね書きしてゆく場合に有用です. 1個のウィンドウにグラフを 1個しか描いていないと

きにはあまり使い道がありません.

7.2 2個以上の関数を同時にプロットする

2個以上の関数を同時にプロットするときには,

plot 関数 1, 関数 2 [Enter]

のようにします. 線のスタイルはこれらの関数が区別できるように自動的に調整されます.

各関数について線のスタイルなどのパラメータを指定することもできます. このような場合は,

plot 関数 1 with オプション,

関数 2 with オプション [Enter]

などのように指定します. 上で「オプション」と書かれた部分にはこれまで説明してきたいろいろなコマンドが

入ります. なお, 紙面の都合でコマンドの途中で改行されていますが, これは上記の通りに入力するという意味

ではありません.

以下に例を示します.

plot sin(x),cos(x) [Enter]

結果は図 7.25のようになります.

線のスタイルなどのパラメータを指定する場合は, たとえば次のようになります.

plot sin(x) with lines lw 4, cos(x) with points pt 5 ps 3 [Enter]

結果は図 7.26のようになります.

図 7.25: 2個の関数の重ね書き

-1

-0.8

-0.6

-0.4

-0.2

0

0.2

0.4

0.6

0.8

1

-10 -5 0 5 10

sin(x)
cos(x)

図 7.26: 2個の関数の重ね書き (パラメータを指定)

7.3 グラフをどんどん重ね書きしてゆく

gnuplotでグラフを描くときには, plotが実行されるたびに以前描いてあったグラフが消え, 新しいグラフが

作成されました. では, 1枚の図にグラフをどんどん追加してゆくにはどうしたらよいでしょうか?

このためには, multiplot というコマンドを使います.

multiplotの使い方を説明する前に, multiplotモードでは画面に描画したグラフをあらためてファイルに保

存することはできないということを注意しておきます. gnuplotでは描画したグラフを保存するときには

terminal と output の切り換え

という操作をおこなうわけですが, 一旦 multiplotモードに入ってしまうと, この切り換えができません. で

すから, multiplotモードで描画したグラフをファイルに保存するためには,

• multiplotモードに入る前に terminal を postscript などに指定し, さらに output に適当なファイ

ルを指定しておく

• xwd, xwpick, gimpなどのツールを使って描画したグラフをキャプチャする

のいずれかの操作をおこなう必要があります. 前者の操作をおこなう場合は, ふつうは事前に試し書きをしてグ

ラフを調整してから改めて描画をすべてやり直すことになり, 若干面倒です. 後者の操作では, 得られるグラフ

の品質はかなり低くなります.

では, multiplot の使い方を説明しましょう.

set multiplot [Enter]

と入力すると, gnuplotのウィンドウは図 7.27のような状態になり, さらに何も描かれていないグラフィックス

ウィンドウが画面にあらわれます. プロンプトが

図 7.27: multiplotモード

gnuplot>

から

multiplot>

に変わったことで, グラフの重ね書きモードに入っていることが確認できます. これ以降コマンド plot を実

行するたびに, グラフがこのウィンドウにどんどん追加されてゆきます. ただし, 描画範囲の自動調整などはう

まくいかないことがあるので, set origin, set size といったコマンドを実行してグラフの原点や大きさを調

整する必要があります.

では, 例として

set multiplot [Enter]

set size 0.5 [Enter]

set origin 0,0 [Enter]

plot sin(x) [Enter]

set origin 0.5,0 [Enter]

plot cos(x) [Enter]

set origin 0,0.5 [Enter]

plot x**2 [Enter]

set origin 0.5,0.5 [Enter]

plot x**3 [Enter]

図 7.28: multiplotによって作成されたグラフの例

というコマンドを実行した場合に作成されるグラフを図 7.28に示します. なお, 先に述べたように, このよう

なグラフをそのまま PostScript形式で保存するためのコマンドは gnuplotには用意されていません. ですか

ら, グラフを保存したいときには, X Window System に用意されている xwdや xwpickなどのコマンドを使う必

要があります. これらのコマンドの使い方については UNIXシステムのオンラインマニュアルを参照して下さい.

multiplotモードを抜けて通常のモードに戻りたいときには,

set nomultiplot [Enter]

と入力します.

ここまでの結果をまとめておきましょう.

機能 コマンド

multiplotモードに入る set multiplot [Enter]

multiplotモードを抜ける set nomultiplot [Enter]

表 7.17: multiplotモードへの切り換え

7.4 直交座標を用いたパラメータ付き曲線のプロット

パラメータ付き曲線とは, 変数 tを連続的に変化させたとき,

x = f(t), y = g(t)

を満たす点が描く軌跡をいいます.

本節では 2次元のパラメータ付き曲線を描画する方法について説明します.

7.4.1 パラメータ表示モードの起動と終了

パラメータ付きの曲線をプロットしたいときには,

set parametric [Enter]

と入力します. すると, gnuplotのウィンドウは図 7.29のような状態になります. 画面に表示される

n

図 7.29: parametricモードへの切り換え

dummy variable is t for curves, u/v for surfaces

というメッセージから, 描画モードがパラメータ付き曲線モードに切り換わったことが確認できます. パラメー

タ付き曲線モードでは, 変数は xから tに変わります.

パラメータ付き曲線モードを抜けるには

set noparametric [Enter]

と入力します. すると, gnuplotのウィンドウは図 7.30のような状態になります. 画面に表示される

dummy variable is x for curves, x/y for surfaces

というメッセージから, 描画モードが通常のモードに戻ったことが確認できます. 変数も tから xに戻ります.

7.4.2 基本的なな描画コマンド

直交座標を用いてパラメータ付き曲線を描画するときには, コマンド plot に続いて, x軸の座標と y軸の座

標をあらわす関数を t の関数として指定します.

例として, 円をパラメータ付き曲線としてプロットしてみましょう. この場合, x座標は cos t, y座標は sin t

です.

まず, 直交座標によるパラメータ付き曲線描画モードに移行します.

図 7.30: parametricモードを抜けたところ

set parametric [Enter]

続いて描画コマンドを入力します.

plot cos(t),sin(t) [Enter]

すると, 図 7.31のように楕円が描かれます. これは本来は円なのですが, gnuplotは標準ではグラフを横長で

作成するためにこのようなグラフになってしまっています.

図 7.31: 円が楕円に見えているグラフ

上記の例では, 真円を描画したいのに, 描画結果は楕円になってしまいました. これは, 画面の縦横の比率

を指定しなかったからです. 一般に, ふつうの関数のグラフは画面が横長の方が見やすいですが, パラメータ付

き曲線の場合は, どのような画面を使うとグラフが見やすくなるかは曲線の形に大きく依存します. ですから,

パラメータ付き曲線を描画するときには, 試行錯誤によって見やすい画面の縦横の比率を自分で見付ける必要が

あります. 画面の縦横の比率の指定の方法は第 7.1.11節に記載されているので, ここでは例 7.31のグラフを

画面の縦横比率を 1に変えて再描画する例を挙げるにとどめます.

上記のようにするためのコマンドは, 以下の通りです.

set size ratio -1 [Enter]

plot cos(t),sin(t) [Enter]

とします. このようにすると, 図 7.32のように真円が描画されます.

図 7.32: 真円の描画

7.4.3 パラメータ付き曲線の例

パラメータ付き曲線の例として, リサージュ図形と, アストロイドと呼ばれる図形を示しておきます.

リサージュ図形とは,

x = Ax cos(ωxt + ψx), y = Ay sin(ωyt + ψy)

なるパラメータ表示であらわされる曲線です. 電気系の学生にとっては, オシロスコープの取り扱いの際にお馴

染みの図形だと思います. まず,

x = cos 3t, y = sin t

の場合についてプロットしてみます.

描画コマンドは

plot cos(3*t),sin(t) [Enter]

です. 描画結果は図 7.33 のようになります. wx と wy の比率が複雑になる程, 図形の形も複雑になります.

図 7.33: リサージュ図形

例として,

plot cos(11*t),sin(13*t) [Enter]

によって描画した結果 (x = cos(11t), y = sin(13t))を図 7.34に示します. wxとwxの比率が有理数でない場合,

図 7.34: 複雑なリサージュ図形

リサージュ図形は閉曲線になりません.

次に, アストロイドと呼ばれる図形をプロットします. これは,

x = cos3 t, y = sin3 t

なるパラメータ表示を持つ曲線です.

描画コマンドは

plot cos(t)**3,sin(t)**3 [Enter]

です. 描画結果を図 7.35 に示します.

7.4.4 パラメータの範囲や描画範囲の指定

パラメトリック付き曲線の描画モードでは, コマンド plot に続けて, パラメータの範囲, 横軸と縦軸の 3種

類が範囲を指定できます.

パラメータ, 縦軸, 横軸のそれぞれについてそれぞれ

• 指定なし

• 下限のみ指定

• 上限のみ指定

• 上限と下限を指定

の 4種類の指定ができます. 指定されていない部分は自動的に調整されます. 以下にいくつか例を示します.

これとは別に, あらかじめ変数や半径の範囲を指定しておくこともできます.

変数の動く範囲を指定するには,

図 7.35: アストロイド

表 7.18: 描画範囲の設定例

機能 コマンド

パラメータ, 横軸, 縦軸とも指定しない plot cos(t),sin(t) [Enter]

パラメータの上限のみ指定 plot [:2*pi] cos(t),sin(t) [Enter]

パラメータの下限のみ指定 plot [0:] cos(t),sin(t) [Enter]

パラメータの上限と下限を指定 plot [0:2*pi] cos(t),sin(t) [Enter]

パラメータは指定なし, 横軸の上限と下限を指定 plot [] [-1:1] cos(t),sin(t) [Enter]

横軸と縦軸の上限と下限を指定 plot [] [-1:1] [-1:1] cos(t),sin(t) [Enter]

パラメータ, 横軸, 縦軸の上限と下限を指定 plot [0:2*pi] [-1:1] [-1:1] cos(t),sin(t) [Enter]

set trange [下限:上限] [Enter]

などとします. 横軸と縦軸の範囲を指定するには, それぞれ

set xrange [下限:上限] [Enter]

set yrange [下限:上限] [Enter]

などとします. trangeの初期値は [-10:10]に, xrangeの初期値は [-10:10]になっています.

7.5 極座標を用いたパラメータ付き曲線のプロット

極座標を用いたパラメータ付き曲線のプロットとは, 図 7.36に示すように, 曲線の軌跡を, 原点と曲線上の

点を結ぶ半直線の角度 tとその半直線の長さ r(t)の関係によって表現したものをいいます.

t

r(t)

x

y

o

図 7.36: パラメータ付き曲線の極座標表示

本節では, 極座標を用いてパラメータ付き曲線をプロットする方法について説明してゆきます.

7.5.1 極座標モードの起動と終了

パラメータ付きの曲線をプロットしたいときには, まず

set polar [Enter]

と入力します. すると, 描画モードが極座標を用いたパラメータ付き曲線モードに切り換わります. 7.4節と

同様に, 独立変数は xから tに変わります.

パラメータ付き曲線モードを抜けるには

set nopolar [Enter]

と入力します. すると, 独立変数が t から x に戻ります.

set polar [Enter]

として極座標を用いたパラメータ付き曲線描画モードにした状態で,

r = f(t)

なる曲線 (たとえば f(t) = sin t, f(t) = t2 など)をプロットするには,

plot f(t) [Enter]

とします. すると, 変数 tを連続的に変化させたときに原点からの角度が tで長さが f(t)の弦の先頭が描く軌跡
が画面に描画されます.

7.6 簡単な例

いくつか例を見てみましょう.

まず,

r = sin2t

なる図形をプロットしてみます. これは, 4つ葉のクローバーの形の図形です. ただし, 描画にあたって, 画

面の縦横比率を 1に固定しておくことにします.

まず

set size ratio [Enter]

set polar [Enter]

と入力して画面の縦横比率を 1にしてから極座標を用いたパラメータ付き曲線描画モードに移行し, 続いて

plot sin(2*t) [Enter]

と入力すると, 図 7.37のようなグラフが得られます.

図 7.37: 4つ葉のクローバーの形の図形

上のグラフにおいて正弦関数の周期を変えると葉の数が変わります. たとえば, 葉の数を 8枚にするには,

r = sin4t

なる極座標表示を用います. 実際,

plot sin(4*t) [Enter]

というコマンドを実行すると, 図 7.38 のような図形が得られます.

7.6.1 変数と描画範囲の指定

変数や描画範囲を指定する方法は直交座標を用いた場合と同じです. いくつか例を挙げます.

これとは別に, あらかじめ変数や半径の範囲を指定しておくこともできます.

変数の動く範囲を指定するには,

set trange [下限:上限] [Enter]

などとします. また, 描画される半径の範囲を指定するには,

図 7.38: 4つ葉のクローバーの形の図形

表 7.19: 変数や描画範囲の指定

機能 コマンド

変数の動く範囲のみを指定 plot [0:pi] sin(2*t) [Enter]

半径の範囲のみを指定 plot [0:pi] sin(2*t) [Enter]

set rrange [下限:上限] [Enter]

などとします. 極座標モードに入ったときには, trangeの初期値は [0:2*pi]に, rrangeの初期値は [0:10]

になっています.

なお, 描画にあたってサンプル点の数を十分多く取っておかないと, サンプル点が足りないために予想に反す

る奇妙な図形が得られることがあります. 極座標を用いたパラメータ付き曲線描画モードを使うときは, あらか

じめ

set samples サンプル点の数 [Enter]

と自分で指定しておくと良いでしょう (「サンプル点の数」のところには適当な自然数が入ります). サンプル

点の数の初期値は 100ですが, 100では足りないことも多いです.

第8章 いろいろな3次元グラフを描画する

8.1 直交座標系を用いた3次元グラフの描画

3次元グラフをプロットするときにはコマンド plotのかわりにコマンド splot を使います.

8.1.1 単純な 3次元グラフの表示

まず単純なグラフを描画してみましょう.

splot exp(-(x**2+y**2)/20) [Enter]

と入力すると, 図 8.1のような結果が得られます.

図 8.1: 関数 exp[−(x2 + y2)/20のプロット

プロットが終わったグラフを eps ファイルの形式で保存するには,

set terminal postscript eps [Enter]

set output ’ファイル’ [Enter]

replot [Enter]

とします. ここに, 「ファイル」と書かれた部分には適当なファイル名が入ります.

8.1.2 描画範囲の調整

3次元グラフを描画するときには, x軸, y軸および z軸の範囲をそれぞれ指定できます. 基本的な構文は

splot [xの下限:xの上限] [yの下限:yの上限] [zの下限:zの上限] 関数 [Enter]

というものです. ただし, 後半の部分を省略して, 最初の 1個だけ, あるいは最初の 2個だけのパラメータを

指定することもできます. 途中の部分を省略するときには省略したい部分に []と書きます.

いくつか例を示します.

表 8.1: 各軸の範囲の指定の例

機能 コマンド

xの範囲のみ指定 splot [0:1] sin(x+y) [Enter]

xと yの範囲を指定 splot [0:4] [0:1] sin(x+y) [Enter]

x,y,zの範囲を指定 splot [0:4] [0:1] [0:4] sin(x+y) [Enter]

x,zの範囲を指定 splot [0:4] [] [0:1] sin(x+y) [Enter]

ここで注意してほしいのは, splotにおける各軸の範囲指定は, 先に実行されたコマンドの影響を受けるとい

うことです. gnuplotを起動したときにはどの軸の範囲もデフォルトで verb?[-10:10]?になっているのです

が, たとえば

splot [0:1] sin(x+y) [Enter]

というコマンドを実行すると, x軸のデフォルトの範囲は [0:1]に変わってしまいます. なお, コマンド plot

にはこのようなことはありません.

あらかじめ x軸, y軸, z軸の範囲を指定しておくこともできます. このためには,

set xrange [下限:上限] [Enter]

set yrange [下限:上限] [Enter]

set zrange [下限:上限] [Enter]

のように入力します. ただし, 「下限」と書かれた部分と「上限」と書かれた部分には適当な数が入ります. こ

のようにして指定した場合も, 先の例のように

splot [0:1] sin(x+y) [Enter]

と明示的に範囲指定して splot を使うと, 先ほど指定した範囲は上書きされてしまいます.

8.1.3 グラフにかかった網目の大きさを変える

splotは標準では 3次元グラフに横軸に 10個, 縦軸に 10個の目がある網をかけて表示しています. しかし,

この状態では, 網の目が粗すぎて関数の様子をうまく表示できないことがあります.

このような場合には, set isosamples というコマンドを使うことで網目の数を調整することができます.

コマンド set isosamples の使い方は,

set isosamples 横軸の網目, 縦軸の網目 [Enter]

です. ここに, 「横軸の網目」, 「縦軸の網目」と書かれた部分には適当な整数が入ります.

例題として, 図 8.1のグラフにおいて, 網目を横軸 50, 縦軸 50に切り直してみます.

set isosamples 50,50 [Enter]

splot exp(-(x**2+y**2)/20) [Enter]

結果は図 8.2のようになります.

exp(-(x**2+y**2)/20)

図 8.2: 網目の切り直しの例

8.1.4 隠れ面の表示の切り換え

splotで 3次元グラフを網目を使って表示しているとき, 画面に描画されるのは, 網目をある方向から見た図,

すなわち, ある方向に射影した像です. さて, 現実の世界で 3次元図形を見るときには, 目に見えるのは一番近

い図形のみで, その影に隠れた図形 (隠れ面)は見えません. ところが, gnuplotは, 標準状態では他の網目に

隠された面も表示します. この隠れ面を表示しないようにするには, set hidden3dというコマンドを使います.

隠れ面の表示と非表示の切り換えのコマンドを表 8.2にまとめておきます.

表 8.2: 隠れ面の表示切り換え

機能 コマンド

隠れ面を表示しない set hidden3d [Enter]

隠れ面を表示する set nohidden3d [Enter]

例として,

set hidden3d [Enter]

splot exp(-(x**2+y**2)/20) [Enter]

とした場合の図を 8.3に示します. 図 8.1と見比べてみて下さい.

8.1.5 視点の変更

3次元グラフを画面に表示するときには, 視点の位置によってグラフが見やすくなったり見にくくなったりし

ます.

そこで, グラフの見やすさを調整するために, set view というコマンドが用意されています.

図 8.3: 隠れ面を表示しない場合のグラフ

われわれの視点 (モニタの正面)から見たグラフが描かれる座標系は, 初期状態ではモニタの左右方向が x軸,

モニタの上下方向が y軸, モニタに立てた垂線の方向が z軸になっています.

この座標系を最初に x軸のまわりで回転させ, さらに新しい z軸のまわりで回転さた座標系がグラフが実際に

描画される座標系になります (図 8.4).

図 8.4: 座標系の回転

我々の視点はつねにモニタの正面に固定されていて, グラフが描画される座標系の方が回転しているというこ

とに注意して下さい. コマンド名 set view からは視点の方を移動することが連想されますが, これは実際の

挙動とは異なります.

さらに, グラフ全体の縮尺と z 軸の縮尺を指定することで, 実際にグラフがどのように描かれるかが決まり

ます.

以上の説明をふまえた上で, コマンド set view の使い方を紹介しましょう.

コマンド set view は

set view 回転 1, 回転 2, グラフの拡大率, z軸の拡大率 [Enter]

のようにして使います. 回転 1は x軸のまわりの回転の角度で, 0度から 180度のあいだの値を取ります. 回

転 2は少しわかりにくいのですが, 回転 1で回転したあとの座標系に関する z 軸まわりの回転で, 0度から 360

度の値を取ります. 回転 1と回転 2の意味を図 8.4に示します.

拡大率には任意の数が指定できます. また, 回転 2以降は省略可能です.

gnuplotを起動したときには, 座標系は

set view 60, 30, 1, 1 [Return]

を実行した場合と同じ位置にあります.

以下に, 座標軸を回転させたときの例を示します.

まず最初に, 回転角 1, 回転角 2をともに 0, 0に指定してみましょう.

set view 0,0,1,1 [Return]

splot exp(-(x**2+y**2)/20) [Return]

結果は図 8.5(a)のようになります. z軸がモニタの画面に垂直で, グラフを真上から見ていることになるので,

関数の形はまったくわかりません続いて, 回転角 1を 30度ずつ増やし, 回転角 2は零のまま固定しててみます.

結果を図 8.5(b), (c), (d)に示します. 結果を図 8.5(d)では, グラフを真横から見た図が表示されています.

続いて, 回転角 1を 10度に固定して, 回転角 2を 0度, 30度, 60度, 90度とした場合のグラフをプロット

してみましょう.

8.1.6 水平面 (座標が書かれる面)の位置の変更

gnuplotの 3次元グラフは, 適当な高さに座標平面 (xy平面)が描かれ, その上あるいは下に関数の形が網の

かかった面の形で表示される, という構造になっています. そして, 表示される座標平面の高さが適切でないこ

ともあります.

このような場合に座標平面の表示される高さを調整するためのコマンドが set ticslevel です.

コマンド ticslevelの使い方は,

set ticslevel p [Enter]

です. ここに, pと書かれた部分には数値が入ります.

数値 pがどのようにして決まるかというと,

p =
座標平面の z座標− z軸の最小値

z軸の最小値− z軸の最大値

という式にしたがって決まります.

この式の分母において z軸の最小値から z軸の最大値が引かれていることに注意して下さい. このため, pが
正の値のときには, 座標平面は必ず z軸の最小値より下側に書かれます. 座標平面を z軸の最小値より上に描画

したいときには, pを負の値に設定する必要があります.

gnuplotを起動したときには ticslevel は 0.5に設定されています.

現在の ticslevel の設定を知りたいときには,

show tics [Enter]

と入力します.

座標平面の表示に関連したコマンドを表 8.3にまとめておきます.

(a)回転角 1=0, 回転角 2=0

(b)回転角 1=30, 回転角 2=0

(c)回転角 1=60, 回転角 2=0

(d)回転角 1=90, 回転角 2=0

図 8.5: 回転角 1の指定

(a)回転角 1=10, 回転角 2=0

(b)回転角 1=10, 回転角 2=30

(c)回転角 1=10, 回転角 2=60

(d)回転角 1=10, 回転角 2=90

図 8.6: 回転角 2の指定

表 8.3: 座標平面の表示に関連したコマンド

機能 コマンド

座標平面の場所を設定する set ticslevel 位置 [Enter]

座標平面の設定を調べる show tics [Enter]

8.1.7 網のかかった面の表示をやめる

網のかかった面の表示をやめる方法もあります. これは, 次の 8.1.8節で述べる等高線の表示と組み合わせ

て使われます.

網のかかった面を表示するには set surface というコマンドを, 表示をやめるには set nosurface という

コマンドを使います. これらを表 8.4にまとめておきます.

表 8.4: 網のかかった面の表示に関連したコマンド

機能 コマンド

等高線を表示する set surface [Enter]

等高線を表示しない set nosurface [Enter]

8.1.8 等高線の表示

8.1.8.1 等高線をプロットする

グラフを等高線を表示することもできます. このためには, set contour というコマンドを使います.

set countourを実行してから関数をプロットすると, その関数の等高線が表示されます. しかし, これには

いくつか問題があります.

set contour [Enter]

splot exp(-(x**2+y**2)/20) [Enter]

とすると, 結果は図 8.7 のようになります.

図 8.7には,

• 等高線を斜めの視点から見ると見にくい

• 網かけされたグラフが邪魔

という問題があります.

より見やすい等高線グラフを作成するには, 8.1.5節の視点変更と 8.1.7節の網かけ面の非表示を組み合わせ

る必要があります. さらに, このままでは図が粗いので, 8.1.3節の網目の切り直しを併用した方がよいでしょ

う. さらに, 見やすいようサイズも適当に調整すべきです.

以上を踏まえた上で,

set nosurface [Enter]

set contour [Enter]

set size 0.66,1 [Enter]

set view 0,0,1,1 [Enter]

図 8.7: 等高線のプロット (望ましくない例)

set isosamples 50,50 [Enter]

splot exp(-(x**2+y**2)/20) [Enter]

という一連のコマンドを実行すると, 図 8.8のような等高線グラフが得られます.

図 8.8: 等高線のプロット

ここまでに出てきた等高線の表示に関連したコマンドを表 8.5にまとめておきます.

表 8.5: 等高線の表示に関連したコマンド

機能 コマンド

等高線を表示する set contour [Enter]

等高線を表示しない set nocontour [Enter]

等高線の高さのを指定した書式で表示する set clabel ’書式’ [Enter]

等高線の高さを表示しない set noclabel [Enter]

現在の等高線の高さの表示形式を見る show clabel [Enter]

88



8.1.8.2 等高線の高さ表示のフォーマットを変える

等高線グラフを描いたときには, 図の右側に各等高線の数値が表示されます. この数値を変更するときには,

set clabel というコマンドを使います.

set clabelの使い方は,

set clabel ’フォーマット’ [Enter]

というものです. ただし, ’フォーマット’の部分には, C言語と同じ構文の浮動小数点数の書式指定が入りま

す. 書式指定のしかたについては 7.1.8.2節に説明されているものと同じなので, そちらを参照して下さい.

標準の書式は%8.3gになっています.

等高線の高さの数値を表示したくないときには,

set noclabel [Enter]

とします.

また, 現在の現在の等高線の高さの表示形式を見たいときには

show clabel [Enter]

とします.

8.1.8.3 等高線のよりきめ細かい制御

等高線をよりきめ細かく制御したいときには, set cntrparam というコマンドを使います. このコマンドを

使う前には, 等高線を描画するときには gnuplotで標準的に設定されている値が使われます.

等高線のグラフを描くときには, 等高線を描くべき高さにある標本点をいくつか取ってからこれらのあいだを

線で結んでゆくのですが, この結び方や標本点の高さの取り方がこのオプションによって決まります.

表 8.6に, set cntrparam で指定できる事項の一覧を示します.

表 8.6: 等高線の制御のためのコマンド

機能 コマンド

標本点のあいだを直

線でむすぶ

set cntrparam linear [Enter]

標本点のあいだをな

めらかな直線でむす

ぶ

set cntrparam cubicspline [Enter]

標本点を適当に避け

るなめらかな線を引

く

set cntrparam bspline [Enter]

cubicspline と

bspline で線のなめ
らかさを指定する

(注 1)

set cntrparam points数 (注 2) [Enter]

bspline において線
のなめらかさを指定

する (注 3)

set cntrparam order 数 [Enter](注 4)

等高線の数を自動調

整する

set cntrparam levels auto [Enter]

自動調整で指定され

た数の等高線を引く

set cntrparam levels auto 数 [Enter](注 5)

指定された本数の等

高線を引く

set cntrparam levels 数 [Enter](注 6)

指定された高さの点

に等高線を引く

set cntrparam levels discrete z1, z2, ..., zn [Enter](注 7)

指定された間隔で等

高線を引く

set cntrparam levels incremental 初期値, 刻み, 終了値 [Enter](注 8)

等高線の設定を見る show contour [Enter]

注 1 gnuplotは, 曲線を描画するときには, 曲線を小さな直線の集まりで近似している. このパラメータは, 近似直線をどのく
らい細かく取るかを決定する.

注 2 「数」の部分には適当な整数が入る. 値が大きいほど曲線はなめらかになる.

注 3 set cntrparam pointsの方は曲線が標本点の近くを通るときの通り方には影響を与えないが, set cntrparam order

の方はこの通り方に影響を与える.

注 4 「数」の部分には 2から 10までの整数が入る; 値が大きいほど等高線がなめらかになるが, 標本点からは大きく外れる

注 5 「数」の部分には適当な整数が入る. ただし, 正確に指定された数の等高線が引かれるわけではない.

注 6 「数」の部分には適当な整数が入る

注 7 z1, z2, ... zn の部分には等高線が引かれる高さ (z座標)をコンマで区切ってならべて指定する;等高線の数はいくつでも
よい

注 8 「初期値」, 「刻み」, 「終了値」には適当な数値が入る; 初期値から終了値まで刻みで指定された間隔で等高線が引かれ
る; 終了値は省略できる

8.1.9 グラフの表示スタイルを網かけや等高線以外にする

7.1.4節と同様に, グラフの表示スタイルを網かけ以外のものに変更することもできます.

方法は 7.1.4節と同様なので, ここではいくつか簡単な例を紹介します.

まず, グラフをインパルスで表示してみましょう.

splot exp(-(x**2+y**2)/20) with impulses [Enter]

とすると, 図 8.9のような結果が得られます.

図 8.9: 3次元グラフのインパルス表示

次に, グラフを点のみで表示してみましょう.

splot exp(-(x**2+y**2)/20) with dots [Enter]

とすると, 図 8.10のようなグラフが得られます.

図 8.10: 3次元グラフの点を使った表示

8.1.10 PostScriptファイルのBoundingBoxについて

gnuplotで 3次元グラフを描画し, グラフを PostScriptファイルに保存して文書に取り込むと, グラフのま

わりに大きい空白ができてしまうことがあります. このような場合, PostScriptファイル中の BoundingBoxと

いうパラメータを手動で編集すれば, 空白を小さくすることができます.

BoundingBoxは, PostScriptファイルで, 図を描画するために確保する領域の左下の点と右上の点の座標を

指定するパラメータです. 通常, PostScriptファイルの上から数行のところで指定されています. パラメー

タの指定の順番は,

%%BoundingBox: 左下隅の x座標 左下隅の y座標 右上隅の x座標 右上隅の y座標

のようになります. 座標軸の原点は画面左下隅で, 座標軸はx軸が画面横方向, y軸が画面縦方向です. BoundingBox

を試行錯誤によって図が実際に描画されている領域に合わせれば, 余分な空白を小さくすることができます. た

だし, BoundingBoxに変な値を指定してしまうと, 文書に図を取り込んだときに適切に表示されなくなります.

BoundingBoxを手動で編集するのにはそれなりのリスクがあることを認識して下さい.

図 8.11に, 実際の PostScriptファイルで Bounding Box が指定されている部分を示します.

%!PS-Adobe-2.0 EPSF-2.0

%%Title: gnuplot-3d-contour-2.eps

%%Creator: gnuplot 3.7 patchlevel 1 (+1.2.0 2001/01/11)

%%CreationDate: Tue May 25 16:07:57 2004

%%DocumentFonts: (atend)

%%BoundingBox: 100 130 237 232 << ここが BoundingBox

%%Orientation: Portrait

%%EndComments

図 8.11: BoundingBoxの指定箇所

8.2 パラメータ表示された3次元グラフ

パラメータ表示された 3次元グラフとは, 適当な関数 f(u, v), g(u, v), h(u, v)に対して

x = f(u, v), y = g(u, v), z = h(u, v)

を満たす点の全体のことを言います. この図形は多くの場合は 3次元中の曲面になります.

本節ではパラメータ表示された 3次元グラフを描画する方法について説明してゆきます.

8.2.1 パラメータ表示モードの起動と終了

パラメータ表示された 3次元グラフを表示したいときには, パラメータ表示された 23次元グラフを表示すると

きと同様に,

set parametric [Enter]

と入力します. すると, 画面は図 8.12のような状態になります. 図 8.12からもわかるように, パラメータ付

き曲線および曲面モードは, 2次元と 3次元で共通です. そして, コマンド plot を使うと 2次元のパラメータ

付き曲線が, コマンド splot を使うと 3次元のパラメータ付き曲線や曲面が描画されます. plot でパラメー

タ付き曲線を描画するときには独立変数として t を使用しましたが, splot でパラメータ付き曲線や曲面を描

画するときには独立変数として u と v を使用します.

パラメータ表示モードを抜けるには,

set noparametric [Enter]

と入力します.

図 8.12: パラメータ付き曲面描画モード

8.2.2 簡単な例題

この節では, 例題として, いくつかの簡単なパラメータ表示された 3次元グラフを描画してみます.

まず, トーラス (ドーナツ型)をプロットしてみます. トーラスのパラメータ表示は, たとえば

x = cos u(3 + cos v), y = sin u(3 + cos v), x = sin v

によって与えられます. これをプロットしてみましょう.

あらかじめ

set parametric [Enter]

とした上で,

splot cos(u)*(3+cos(v)),sin(u)*(3+cos(v)),sin(v) [Enter]

と入力すると, 図 8.13のように, トーラスが描画されます. 残念ながら, 図 8.13の描画は見やすいとは言い

図 8.13: トーラス

難いです. この理由は, パラメータの範囲が適切でないこと, 隠れ面の処理がなされていないこと, 網目の切

り方が粗すぎることにあります. これらを調整する方法を以下で順に見てゆきます.

8.2.3 描画範囲の調整

パラメータ付きの 3次元グラフを描画するときには, 変数 u,vおよび x軸, y軸および z軸の範囲をそれぞれ

指定できます. 基本的な構文は

splot [uの下限:uの上限] [vの下限:vの上限]

[xの下限:xの上限] [yの下限:yの上限] [zの下限:zの上限] 関数 [Enter]

というものです. ただし, 後半の部分を省略することもできます. 途中の部分を省略するときには省略したい

部分に []と書きます. なお, 紙面の都合で途中に余分な改行が入っていますが, この改行は実際に入力するわ

けではありません.

具体的な指定の仕方については, 8.1.2節と同様ですので説明を省略します.

なお, あらかじめ変数 u, vおよび x軸, y軸, z軸の範囲を指定しておくこともできます. このためには,

set urange [下限:上限] [Enter]

set vrange [下限:上限] [Enter]

set xrange [下限:上限] [Enter]

set yrange [下限:上限] [Enter]

set zrange [下限:上限] [Enter]

のように入力します. ただし, 「下限」と書かれた部分と「上限」と書かれた部分には適当な数が入ります.

トーラスを変数 uと vの範囲をそれぞれ 0から 2πまでに設定してプロットしている例を以下に示します.

splot [0:2*pi] [0:2*pi] cos(u)*(3+cos(v)),sin(u)*(3+cos(v)),sin(v) [Enter]

結果は図 8.14のようになります. これだけでも, 図 8.13より幾分かは見やすくなっています.

図 8.14: トーラス:変数の範囲を適切に設定

8.2.4 隠れ面や網の切り方の処理

図 8.14でも, 隠れ面が処理されていないのと, 網の切り方が粗すぎるために, まだあまり見やくありません.

これを修正するにはどうしたらよいでしょうか?

隠れ面を処理する方法や網の切り方をより細かくする方法は 8.1.3節, 8.1.4節で述べられている通りです.

すなわち, パラメータ isosamples の値を大きくすれば網目は細かくなりますし, set hidden3dとすれば隠れ

面は表示されなくなります.

では, これらを併用してみましょう.

set hidden3d [Enter]

set isosamples 40,40 [Enter]

splot [0:2*pi] [0:2*pi] cos(u)*(3+cos(v)),sin(u)*(3+cos(v)),sin(v) [Enter]

としてみます. すると, 結果は図 8.15のようになります. こんどは非常に見やすいドーナツ型の図が得られ

図 8.15: トーラス:隠れ面の処理とより細かい描画

ています.

8.2.5 その他の各種のオプション

図の標題の付け方などは通常の 2次元グラフや 3次元グラフと同じです. 詳細については第 6章, 第 7章お

よび第 8章などを参照して下さい.

8.2.6 3次元のパラメータ付き曲線を表示する

3次元のパラメータ付き曲線 (曲面ではなく)を表示したいときには, 変数 u,vのどちらか一方だけを使えばよ

いです.

例として螺旋をプロットしてみましょう.

set parametric [Enter]

set isosamples 40,40 [Enter]

splot [0:10*pi] u,sin(u),cos(u) [Enter]

結果は図 8.16のようになります.

図 8.16: 螺旋の描画

8.2.7 注意事項

gnuplot-3.7.1におけるコマンド splotはパラメータ表示モードでは誤動作あるいは暴走することがあります.

第9章 データファイルのプロット

データファイルとは, 実験の測定値や統計データのような数値が規則正しく並べられたファイルのことをいい

ます. 本節では, データファイルに記載されたデータ点からグラフを作成する方法について説明します.

9.1 2次元データのプロット

9.1.1 単純なデータファイルのプロット

データファイルをプロットするときには,

plot ’ファイル’ [Enter]

とします. ただし, 「ファイル」と書かれた部分にはデータファイルの名前が入ります.

例として, 図 9.1のような単純なデータファイルをあった場合を考えます. ファイル名を’sample1.dat’と

します.

20
60
80
60
100

図 9.1: データファイル sample.dat の内容

このデータファイルをプロットするには,

plot ’sample.dat’ [Enter]

とします. すると, 図 9.2 のようなグラフが描画されます. ただし, データ点の表示が小さすぎるのと, デー

タ点が記号「+」で標記されていて, 座標軸と重なると見えなくなってしまうため, あまり見やすくありません.

第 6章および第 7でも説明したように, データ点の表示を切り換えるには, pointtype (略称 pt ) を指定し

ます. どのような指定をするとどのような点が表示されるかについては, 第 7章の図 7.15と図 7.16 (56ペー

ジ)を参照して下さい. また, pointsize (略称 ps )を指定すると, 点の大きさを変えることができます. こ

こでは, ○印を使ってデータ点を表示し直してみることにします. ○印に対応するのは, pointtype 6 (略称

pt 6 )です. また, 点の大きさを 2に指定 pointsize 2 (略称 ps 2 )とすることにします.

上記のような設定で描画するには,

plot ’sample.dat’ with points pt 6 ps 2[Enter]

とします. すると, 図 9.3のようなグラフが得られます.

次に, データ点のあいだを線で結んでみます. このためには,

plot ’sample.dat’ with lines [Enter]

図 9.2: データファイル sample.datのプロット (1)

図 9.3: データファイル sample.datのプロット (2)

図 9.4: データファイル sample.datのプロット (3)

とします. ここに, with lines というのは, データ間を線で結ぶことを意味するキーワードです. 描画結果

は図 9.2のようになります.

データ点のあいだを線で結び, かつデータ点自身も明示したいときには, linespoints というキーワードを利

用します. キーワード linespoints を利用するときには, pointtype , pointtypeも指定できます. たと

えば,

plot ’sample.dat’ with linespoints pt 6 ps 2 [Enter]

とすると, 描画結果は図 9.5のようになります.

図 9.5: データファイル sample.datのプロット (4)

データ点に点を打ち, あいだに線を引かない場合は,

plot ’sample.dat’ with points [Enter]

です. これがデフォルトの挙動です. 描画におけるオプション指定については, 第 6章および第 7でも説明し

ましたが, データファイル特有のオプションもあるので, 第 9.1.7.1節でも改めて議論します.

プロットが終わったグラフを eps ファイルの形式で保存するには,

set terminal postscript eps [Enter]

set output ’ファイル’ [Enter]

replot [Enter]

とします. ここに, 「ファイル」と書かれた部分には適当なファイル名が入ります.

これ以降の例では, 描画結果を見やすくするため, 座標軸の目盛り (xtics, ytics)や線の説明 (key)などを

必要に応じて省略します. また, データ点のスタイルや大きさも必要があれば調整します. ですから, みなさ

んが例と同じコマンドを打ち込んでも, 描画結果がここに記載された例と若干異なることがあります. これらの

調整の仕方については, 第 6章および第 7を参照して下さい.

9.1.2 データファイルの書式

データファイルは以下に述べるような書式および規則にしたがって作成され, gnuplotによって解釈されます.

<

• 1行中に複数の数値が空白で区切って並べられる.

• 空白は英語のスペースあるいはタブ文字を 1個以上 (個数はいくつでもよい)並べたもので構成される.

• 空白に日本語のスペースを使うことはできない.

• 数値のフォーマットとしては, ふつうの整数および小数の表記と指数形式が使える.

• 指数形式でデータを表記するときには, たとえば 2.998E+8 あるいは 2.24e-2 などのように,

仮数部, 文字 Eあるいは e, 10のべきの桁数

という順番で数値および文字を並べる. 文字 Eあるいは eは指数部の始まりを示すために用いられる. 文

字 Eと文字 eの意味は同一である. なお, 先に示した第 1の例は 2.998 × 108 という意味になり, 第 2の

例は 2.24 × 10−2 という意味になる. 指数部については, たとえば 2.998E+8と書くかわりに 2.998E+08

や 2.998E+008 などのように冒頭に数字 0を追加した書き方をしても, 意味は変わらない.

• 標準的な書式は 1行にグラフの x座標と y座標を空白で区切って並べるというものである.

• x座標を省略して 1行に数値を 1個ずつ書いていった場合, gnuplotはこのデータファイルの x座標を 0か

ら始まる整数であると解釈して実行する.

• 1行中にデータ点を 3個以上ならべることもできる.

• 1行中にたくさんデータ点がある場合, gnuplotを実行中にデータのどの列を使ってグラフを 描画するか

を指示することができる.

• gnuplotが作成するグラフの中には 1行中に含まれるデータ点の数を 3個あるいは 4個要求するものもある.

• 文字#以降はコメントとみなされ, 無視される.

• データが記載された行のあいだに空白行 (改行文字のみからなる行)が 1個ある場合, gnuplotはこのデー

タファイルの該当部分をデータの不連続点と解釈する. データ点のあいだを線で結ぶ場合にも, 不連続点

の部分に線が引かれることはない.

• データが記載された行のあいだに空白行が 2個以上あった場合 (2個以上ならいくつでもよい), gnuplot

は, データファイルには複数のグラフをプロットするためのデータが含まれていて, 2個以上の空白行の部

分から新しいグラフのデータが始まっているものとみなす.

• 同一のデータファイルに複数のグラフのデータを含ませる場合, 1個のファイルに含めることができるグラ

フの数には制限はない.

9.1.3 データファイルの例

いくつかデータファイルの例を示しておきます.

図 9.6に, 1行あたりに含まれるデータ点が 1個のデータファイルと, それをプロットした結果を示します.

ただし, データ点を記号「○」であらわし, かつデータ点のあいだを直線で結んでいます.

1
2
1
2
1

1

1.2

1.4

1.6

1.8

2

0 0.5 1 1.5 2 2.5 3 3.5 4

(a) データファイル (b) グラフ

図 9.6: 1行あたりのデータ点が 1個のデータファイルとそのグラフ

図 9.7に, 1行あたりに含まれるデータ点が 2個含まれる場合は, 1番目の数値が x座標, 2番目の数値が y座

標と解釈されます. 図 9.7に, そのようなデータファイルとそれをプロットした結果を示します. 先の例と同

様に, データ点を記号「○」であらわし, かつデータ点のあいだを直線で結んでいます.

図 9.7: 1行あたりのデータ点が 2個のデータファイルとそのグラフ

図 9.8: 指数形式で表現されたデータファイルとそのグラフ

また, 図 9.8に, データ点が指数形式で標記されたデータファイルとそれをプロットしたグラフを示します.

先と同様に, データ点を記号「○」であらわし, かつデータ点のあいだを直線で結んでいます.

なお, 先に述べたように, べきの部分をあらわすときには, 英文字 eの大文字, 小文字の双方が使えます. ま

た, E-03 のような表記において, 符号の直後の 0は省略することができます.

9.1.3.1 不連続点を含むデータファイル

先に述べたように, データが記載された行のあいだに単一の空白行があると, その部分はグラフにおける不連

続点であると解釈されます. 第 9.1.3.2節で述べるように, データファイルに空白行が 2個あると, gnuplot

はこれを不連続点ではなく異なるデータだと解釈します. 単一の空白行と 2個の空白行で意味がまったく異なる

ので注意して下さい.

図 9.9に, 2個の不連続点を含むデータファイルと, そのグラフを線付きでプロットした結果を示します. 先

と同様に, データ点を記号「○」であらわし, かつデータ点のあいだを直線で結んでいます. 図 9.9から, デー

タに空白行がある部分で補間直線が途切れていることがわかります.

9.1.3.2 複数のグラフのデータを含むデータファイル

先に述べたように, データが記載された行のあいだ 2個以上の連続した空白行があると, データファイルには

複数のグラフをプロットするためのデータが含まれていて, 2個以上の空白行の部分から新しいグラフのデータが

始まっているものと解釈されます.

図 9.10に, 2個の不連続点を含むデータファイルと, そのグラフを線付きでプロットした結果を示します. 先

と同様に, データ点を記号「○」であらわし, かつデータ点のあいだを直線で結んでいます. 図 9.10において,

グラフが 3本表示されていることに注意して下さい. 図 9.10(a)では, 4行目および 5行目と 9行目および 10

図 9.9: 不連続点を含むデータファイルとそのグラフ

行目それぞれに 2行分の空白行があります. ですから, gnuplotは, このデータファイルには 3個のグラフを

描くためのデータが含まれていると解釈します.

図 9.10: 3個のグラフに対応したデータが含まれたデータファイルとそのグラフ

9.1.4 複数のグラフが含まれるデータファイルからグラフを抜き出す

複数のグラフが含まれるデータファイルからグラフを抜き出すには, コマンド plot に index というキーワー

ドを付けます. キーワード index の使い方は,

plot ’ファイル’ index グラフの番号 [Enter]

です. ここに, 「グラフの番号」と書かれた部分には 0以上の整数を指定します.

データファイルに複数のグラフのためのデータが含まれている場合, 最初のグラフのためのデータには 0, 2番

目のグラフのためにのデータには 1などといった番号が付きます (番号が 0から始まることに注意). このグラ

フの番号を指定することで, 特定のグラフだけを抜き出すことができます.

先に図 9.10(a)で挙げたファイル (ファイル名を multi-graphs.dat とします)から, 最初のグラフだけ抜き

出してプロットしてみましょう. ただし, グラフを見やすくするために, データ点を記号「○」であらわし, か

つデータ点のあいだを直線で結ぶことにします.

plot ’multi-graphs.dat’ index 0 with lines [Enter]

と入力すると, 図 9.11のように, 図 9.10のグラフの中から最初の 1個を抜き出したグラフを描画することがで

きます.

図 9.11: 複数のグラフを含むデータファイルから 1個のグラフを抜き出してプロットしたもの

データファイルから 2個以上のグラフを抜き出して, それぞれのスタイルを変えるということもできます. た

とえば,

plot ’multi-graphs.dat’ index 0 with linespoints pt 6 ps 2\ [Enter]

’multi-graphs.dat’ index 2 with linespoints pt 4 ps 2[Enter]

とすると, 図 9.12のようなグラフが得られます. ただし, 上のコマンドにおいて, ptは pointtype の省略

形です. pointtype は点の種類を指定するキーワードでした (第 7.1.4.7節の図 7.16を参照). なお, 記号

\はコマンド中に改行を入れなければならないときに使われます.

図 9.12: 複数のグラフを含むデータファイルから 2個のグラフを抜き出してプロットしたもの

実は, キーワード indexそのものにも, 複数のグラフを選択するための, 2通りの使い方があります. その

うちのひとつめは,

index m: n

というものです. この場合, 第 m番目のグラフから第 n番目のデータまでが描画されます. もちろん, 実際

には, mおよび nと書かれた部分に適当な自然数を指定します. ふたつめは, そのうちのひとつめは,

index m: n: p

というものです. この場合, 第m+kp番目のグラフ (ただし k ≥ 0, m+kp ≤ n) がすべて重ね書きされます.

たとえば,

plot ’multi-graphs.dat’ index 1:2 with linespoints pt 6 [Enter]

と入力すると, 図 9.13のようなグラフが得られます. 範囲指定が 1:2となっているので, 1番目と 2番目のグ

5

7

図 9.13: 複数のグラフを含むデータファイルから 2番目と 3番目のグラフを抜き出している例

ラフが選択されています. また,

plot ’multi-graphs.dat’ index 0:2:2 with linespoints pt 6 [Enter]

と入力すると, 図 9.14のようなグラフが得られます. 範囲指定が 0:2:2となっているので, 抜き出されるグラ

図 9.14: 複数のグラフを含むデータファイルから, 0番目以降のグラフを 1個飛ばして抜き出している例

フの番号は, 第 0番, 第 2番, ... となるのですが, 第 2番が選択されたときにデータの番号が 2になって終

了条件が満たされるので, ここでプロットは終わります.

9.1.5 データファイルのどの列を使ってプロットするかを変更する

この節ではデータファイルのどの列を使ってプロットするかを変更する方法について説明します.

説明のために, 図 9.15のようなファイルを考えます. このファイルには,

multi-data.dat

という名前がついているものとします. このデータの第 1列, 第 2列, 第 3列の値はそれぞれ x, sinx, cos x

になっています.

さて, 図 9.15に示されたファイル multi-data.dat を使い,

  0.0000   0.0000   1.0000
  0.3140   0.3089   0.9511
  0.6280   0.5875   0.8092
  0.9420   0.8087   0.5882
  1.2560   0.9509   0.3096
  1.5700   1.0000   0.0008
  1.8840   0.9514  -0.3081
  2.1980   0.8097  -0.5869
  2.5120   0.5888  -0.8083
  2.8260   0.3104  -0.9506
  3.1400   0.0016  -1.0000
  3.4540  -0.3074  -0.9516
  3.7680  -0.5862  -0.8101
  4.0820  -0.8078  -0.5895
  4.3960  -0.9504  -0.3111
  4.7100  -1.0000  -0.0024
  5.0240  -0.9518   0.3066
  5.3380  -0.8106   0.5856
  5.6520  -0.5901   0.8073
  5.9660  -0.3119   0.9501

図 9.15: ファイル multi-data.datの内容

plot ’multi-data.dat’ with lines [Enter]

というコマンドによりグラフをプロットすると, 図9.16のようなグラフが得られます. ファイル multi-data.dat

0 1 2 3 4 5 6

図 9.16: ファイル multi-data.datを単純にプロットした結果

の第 1列を横軸, 第 2列を縦軸に取ったグラフがプロットされているわけです.

では, 第 1列と第 3列を使って, cos xのグラフがプロットできるようにするにはどうしたらよいでしょうか?

このためには, コマンド plot に using というキーワードを付けて実行します.

キーワード using がついたコマンド plot の基本的な構文は,

plot ’ファイル’ using 横軸に使う列の番号:縦軸に使う列の番号 [Enter]

です. ここに, 「横軸に使う列の番号」と書かれた部分と「縦軸に使う列の番号」と書かれた部分には, それぞ

れデータファイル中で描画に使いたい列の番号を書きます. ただし, データの最初の列が第 1列になります.

例題として, ファイル multi-data.dat の第 1列と第 3列を使ってグラフを描いてみましょう. ただし, グ

ラフを見やすくするために, コマンド plot に with lines というキーワードを付け, データ点のあいだを線で

結んでおくことにします.

plot ’multi-data.dat’ using 1:3 with lines [Enter]

図 9.17: ファイル multi-data.datの第 1列と第 3列を使ってプロットした結果

というコマンドを実行すると, 図 9.17に示すように, 希望通り cos xのグラフがプロットされます.

もうひとつの例題として, 横軸にファイル multi-data.dat の第 2列, 縦軸に第 3列を取ったグラフを描い

てみましょう. ただし, この場合は, 横軸が sinxで縦軸が cos xなので, 描画結果が円になることがわかって

ます. ですから, あらかじめグラフの縦と横を一定にするコマンド set size ratio -1 を実行しておくこと

にします.

set size ratio -1 [Enter]

plot ’multi-data.dat’ using 2:3 with lines [Enter]

というコマンドを実行すると, 図 9.18に示すように, 円が描画されます. ただし, サンプルの数が少ないの

と, データファイルをプロットするときには標準では標本点のあいだは直線で結ばれるので, 描画結果はやや角

張ってしまっています.

図 9.18: ファイル multi-data.datの第 2列と第 3列を使ってプロットした結果

実は, キーワード using にはもうすこし高級な使い方があります. これについて説明するためには, gnuplot

が読み込んだデータをどのようにして解釈するかについてもう少し詳しく見ておく必要があります.

キーワード

using m: n

を使うときには, mおよび nにはデータの列の番号が入る (最左端が 1列目と数える)のですが, この部分に数 0

を指定することもできます. 列の番号として 0が指定された場合, gnuplotは, 第 0列目として 0,1,2,3,...と

いうふうに続く整数を自動的に補います.

さらに, 第mの実際に読み込まれたデータを, 「m」という記号で参照することができます.そして,「m」とい
う記号を関数の引数として使うことができます. ですから,

plot ’ファイル’ using m:(関数 ($m)) [Enter]

とすると, 横軸を指定されたデータファイルの第m列に, 縦軸を指定されたデータファイルの第 n列に関数を作
用させたものに取ったグラフが描画されます. ここに, 「ファイル」と書かれた部分には適当なファイルの名前

が, 「関数」と書かれた部分には gnuplot に用意されている数学関数かユーザが自分で定義した関数が入りま

す. 「(関数 ($m))」のように全体を括弧「( )」で囲う必要があることに注意して下さい.

いくつか例を見てみましょう.

まず, キーワード using で列 0を指定した場合の例を見ます.

plot ’multi-data.dat’ using 0:2 with lines [Enter]

というコマンドを実行すると, 図 9.19に示すようなグラフが得られます. このグラフの形は図 9.16と同じです

図 9.19: ファイル multi-data.datの第 0列と第 2列を使ってプロットした結果

が, 横軸の目盛りが違います. この違いがどこから来ているかというと, 図 9.16では横軸の数値としてファイル

multi-data.dat (図 9.15)の第 1列目が使われているのに対し, 図 9.19では横軸の数値として 0,1,2,...が

使われていることから来ています.

次に, 「$n」という記号を使った例を見ます. 横軸をファイル multi-data.dat の第 1列目に, 横軸をファ

イル multi-data.dat の第 1列目に関数 sqrt (平方根を取る関数)を作用させた結果に取ったグラフを表示し

てみます. この場合, 「第 1列目に関数 sqrt を作用させた結果」は

(sqrt($1))

のように記述されます. 「sqrt($1)」と書かれた部分を括弧で囲む必要があることをに注意して下さい.

では例を見てみましょう.

plot ’multi-data.dat’ using 1:(sqrt($1)) with lines [Enter]

というコマンドを実行すると, 図 9.20をのようなグラフが得られます. この例の場合, グラフの横軸がファイ

ル multi-data.datの第 1列に, 縦軸がファイル multi-data.datの第 1列に関数 sqrt を作用させたものに

なっています.

9.1.6 データを間引く

データファイルからグラフを作るときに, 適切なグラフを作るのに必要なデータ点の数と比較して実際のデータ

点の数が極端に多いと, グラフを作成して保存したときにできるファイルが無駄に大きくなってしまいます. で

図 9.20: ファイル multi-data.datの第 1列を横軸に, それに関数 sqrt を作用させたもを縦軸に取ったグラフ

すから, このような場合には, データを適当に間引く, すなわち何個かに 1個データ点を残して残りは捨ててし

まってからグラフを作ったほうが好都合です. これとは別に, グラフが不連続点を含んでいるときに, グラフの

連続な部分だけを取り出してプロットしたいという状況もあります.

このようにデータを適当に間引いてからプロットしたいときには, コマンド plotに every というキーワード

を付けて実行します.

キーワード every の使い方はややこしいので, まず一番簡単な使い方から見ることにします.

再び図 9.15のデータファイル multi-data.dat を考えます. このグラフで, データ点を 2個ずつ飛ばして

プロットしてゆくにはどうしたらよいでしょうか?

plot ’multi-data.dat’ every 3 with linespoints [Enter]

とします. なお, データが間引かれていることを明示するために, with linespointsというキーワードを使用

しています.

上のようなコマンドを打ち込んだときに得られるグラフは図 9.21のようになります.

図 9.21: ファイル multi-data.datのデータ点を 2個ずつ飛ばしてプロットしたグラフ

図 9.16と図 9.21を比較すると, 図 9.21ではデータ点が間引かれている分だけグラフがぎざぎざになってい

ることがわかるでしょう.

データ点を間引くときの一般的な記法は,

plot ’multi-data.dat’ every n [Enter]

108



です. ここに, 「n」の部分には適当な正の整数が入ります. 上記のように指定されると, gnuplotはデータ

ファイルのデータ点を 1個使うたびに, それに続いた n-1個のデータ点を読み飛ばしてゆきます.

キーワード every のより詳しい使い方を説明する前に, gnuplotがデータファイルの各行を解釈する仕方に

ついてもう少し詳しく理解しておく必要があります.

例として, 図9.22(a)のような, いくつかの空白行を含むデータファイルを考えます. ファイル名を block-data.dat

としておきましょう.

このファイルを, gnuplotは次のように解釈します.

• 単一の空白行を境に, データファイル全体は「ブロック」と呼ばれる単位に分割される

• 各ブロックには上から順番に 0から始まる整数の番号が付く

• ブロック中のデータ行には, そのブロックの最初の行から順番に 0から始まる整数の番号が付く

なお, データファイルに空白行がない場合には, データは 1個のブロックから成るものとみなされ, 最初の行

から順番に 0,1,2,...という番号が振られてゆきます.

ファイル block-data.dat が実際にどのように解釈されるかについては, 図 9.22(b)を参照して下さい.

(a)ファイル block-data.dat (b)gnuplotによる解釈

図 9.22: ファイル block-data.doc とその gnuplotによる解釈

さて, 図 9.22をふまえた上で, キーワード every のより詳しい説明に移りましょう.

キーワード every の完全な使い方は,

every 行刻み:ブロック刻み:初期行:初期ブロック:終了行:終了ブロック

のようになります. 上記において, 「行刻み」, 「ブロック刻み」, 「初期点」, 「初期ブロック」, 「終点」

および「終了ブロック」と書かれた部分には, それぞれ 0以上の整数が入ります. これらの意味について表 9.1

にまとめておきます.

これらのオプションはすべて省略可能です. ただし, 省略の仕方は少し変わっていて,

• 指定する最後のオプション以降には記号「:」も含めて何も書かない

表 9.1: キーワード every におけるオプション指定

意味

行刻み データ行を何行ごとに読み出すか指定, 何も指定しなければ 1になる (すべての
データ行を読む)

ブロック刻み データのブロックを何ブロックごとに読み出すか指定, 何も指定しなければ 1に
なる (すべてのブロックを読む)

初期行 データ行をどの行から読み始めるかの指定, 何も指定しなければ 0になる (最初
の行から読む)

初期ブロック データブロックをどのブロックから読み始めるかの指定, 何も指定しなければ 0
になる (最初のブロックから読む)

終了行 データ行をどの行で読み終えるかの指定, 何も指定しなければデータファイルの
最後の行まで読む

終了ブロック データブロックをどのブロックで読み終えるかの指定, 何も指定しなければデー
タファイルの最後の行まで読む

• 最後のオプション以前に省略するものがある場合には, オプションそのものは指定しないが, オプションに

対応する記号「:」だけは残しておく

という規則にしたがいます.

これだけではわかりにくいと思われるので, 表 9.2にいくつか例を示しておきます. オプションの最初あるい

表 9.2: オプションの省略のしかた

省略する項目 記法

完全なオプションの指定 行刻み:ブロック刻み:初期行:初期ブロック:終了行:終了ブロック

終了ブロックを省略 行刻み:ブロック刻み:初期行:初期ブロック:終了行

終了行と終了ブロックを省略 行刻み:ブロック刻み:初期行:初期ブロック

終了行のみを省略 行刻み:ブロック刻み:初期行:初期ブロック::終了ブロック

初期ブロックと終了行を省略 行刻み:ブロック刻み:初期行:::終了ブロック

終了行以降を省略 行刻み:ブロック刻み:初期行:初期ブロック

終了行以降とブロック刻みをを省略 行刻み::初期行:初期ブロック

終了行以降とブロック刻み, 初期行を省略 行刻み:::初期ブロック

行刻み以外をすべて省略 行刻み

終了ブロック以外をすべて省略 :::::終了ブロック

は途中に省略部分がある場合には記号「:」がいくつも続いていることと, オプションを指定している部分の最後

に記号「:」が付くことはないということを確認して下さい.

では, 実際に図 9.22(a)に示されたデータファイル block-data.dat を使って, いくつかのグラフをプロット

してみましょう.

まず参考のために, どのデータ点も間引かれていないグラフをプロットしておきます.

plot ’block-data.dat’ pt 6 ps 2 [Enter]

とすると, 図 9.23のようなグラフが得られます.

続いて,

図 9.23: データファイル block-data.dat の省略のないプロット

plot ’block-data.dat’ every :::1 pt 6 ps 2 [Enter]

としてみます. これは, 「グラフを第 1ブロックまでプロットする」という意味になります. 得られるグラフ

は図 9.24のようなものになります.

図 9.24: 終了ブロックを指定したデータファイル block-data.dat のプロット

さらに,

plot ’block-data.dat’ every :::2:3 pt 6 ps 2 [Enter]

としてみます. これは, 「第 3ブロックまで, 各ブロックの番号 2の行までをプロットする」という意味になり

ます. 結果として, 図 9.25のようなグラフが得られます.

続いて,

plot ’block-data.dat’ every 2:2 pt 6 ps 2 [Enter]

としてみます. これは, 「2個のブロックごとに 1個のブロックを使う, 各ブロック内でも 2個のデータ行ごと

に 1個のデータ行を使う」, すなわち 1ブロック読むごとに 1ブロック読み飛ばし, 各ブロック内部でも 1行読

むごとに 1行読み飛ばすという意味に解釈されます. 上記のコマンドを実行すると, 図 9.26のようなグラフが

得られます.

途中のデータブロックから描画を始める例も見ておきましょう.

plot ’block-data.dat’ every ::1:1:3:2 pt 6 ps 2 [Enter]

図 9.25: 終了行と終了ブロックを指定したデータファイル block-data.dat のプロット

図 9.26: ブロックの読み飛ばしとブロック内の読み飛ばしを指定したデータファイル block-data.dat のプロット

とすると, 「第 1ブロック (2番目のブロック: 番号は 0から始まっていることに注意) から第 2番目のデータ

ブロックまで, 番号 1のデータ行から番号 3のデータ行までをプロットする」という意味になります. これをプ

ロットすると, 図 9.27のようなグラフが得られます.

図 9.27: 途中から描画を始めたデータファイル block-data.dat のプロット

9.1.7 グラフのスタイルを変更する

グラフのスタイルを変える方法は, 第 7.1.4節とほとんど同じなのですが, 第 7.1.4節に述べられたもの以外

にデータファイルに固有のスタイルがいくつかあります.

9.1.7.1 点や線のスタイルを変更する

第 まず, 第 7.1.4節と共通のものについて表 9.3にまとめておきます. 例題については第 7.1.4節を参照

して下さい. なお, 標準では 「データ点を点で表示する」グラフの作成法が選択されています. 標準のまま

で良い場合にはオプションを指定する必要はありません.

表 9.3: グラフのスタイルを変更するためのオプション

機能 オプションおよび説明

データ点を点で表示する with points 標準

データ点のあいだを線で結ぶ with lines

データ点を点で表示し, あいだを線で結ぶ with linespoints

線の種類を変える linetype 数

省略形 lt, with lines か with linespoints の後に指定,
linewidth の後には指定できない

線の太さを変える linewidth 数

省略形 lw, with lines か with linespoints の後に指定

点の種類を変える pointtype 数

省略形 pt, with pointsか with linespointsの後に指定,
pointsize の後には指定できない

点の大きさを変える pointsize 数

省略形 ps, with points か with linespoints の後に指定

データをインパルスで表示する with impulses

データを箱で表示する with boxes

データを階段グラフで表示する with stepsあるいは with fstepsあるいは histeps

9.1.7.2 エラーバーを表示する

誤差が含まれるデータをプロットする場合, グラフにエラーバーを付けて表示することがあります. 本節では

グラフにエラーバーを付ける方法について説明します.

9.1.7.2.1 縦軸のみに誤差を含むグラフ ここでは, まず横軸の数値には誤差がなく縦軸の数値にのみ誤差が含

まれる場合を考えます.

縦軸のみに誤差を含むデータファイルの書式としては, 図 9.28の 2種類のうちのいずれかが想定されています.

そして, データのエラーバーを付ける場合には, コマンド plot に with errorbars というキーワードを使

います. なお, with yerrorbars というキーワードを使っても同じことができます. ですから, 典型的な縦

軸のみにエラーバーを付ける方法は,

plot ’ファイル’ with errorbars [Enter]

図 9.28: 縦軸にエラーバーを付けるグラフが前提としているデータファイルの書式

となります. ここに, 「ファイル」と書かれた部分には適当なファイル名が入ります.

コマンド plot が with errorbars というオプション付きで起動された場合, gnuplot は, データファイル

が図 9.28の形式 1の場合は点 (xi, yi)に点を打ってから, (xi, yi − δy,i)から (xi, yi + δy,i)に伸びる直線を引き,

さらに点 (xi, yi − δy,i)および点 (xi, yi + δy,i)を中心とする短い横方向の線分を引きます.

一方, コマンド plot が with errorbars というオプション付きで起動され, データファイルが図 9.28の

形式 2だった場合には, gnuplotは点 (xi, yi)に点を打ってから, (xi, y
min
i )から (xi, y

max
i )に伸びる直線を引き,

さらに点 (xi, y
min
i )および点 (xi, y

max
i )を中心とする短い横方向の線分を引きます.

図 9.29(a)に示すような形式 1 のデータファイル (ファイル名を eb1.dat とします) を使い,

plot ’eb1.dat’ with errorbars pt 6 ps 2[Enter]

とした結果を図 9.29(b)に, 図 9.30(a)に示すような形式 2 のデータファイル (ファイル名を eb2.dat とし

ます)

plot ’eb2.dat’ with errorbars pt 6 ps 2[Enter]

を使い, とした結果を図 9.30(b)に示します. 形式 1と形式 2の違いを明確にするために, 図 9.30(a)のデー

タでは, エラーバーのデータ点より上の部分がデータ点より下の部分よりもかなり長くなるように取ってあります.

9.1.7.2.2 横軸のみに誤差を含むグラフ 次に, 縦軸の数値には誤差がなく横軸の数値にのみ誤差が含まれる場

合を考えます.

横軸のみに誤差を含むデータファイルの書式としては, 図 9.31の 2種類のうちのいずれかが想定されています.

そして, データのエラーバーを付ける場合には, コマンド plot に with xerrorbars というキーワードを使

います. ですから, 典型的な横軸のみにエラーバーを付ける方法は,

plot ’ファイル’ with xerrorbars [Enter]

となります. ここに, 「ファイル」と書かれた部分には適当なファイル名が入ります.

コマンド plot が with xerrorbars というオプション付きで起動された場合, gnuplot は, データファイ

ルが図 9.28の形式 1の場合は点 (xi, yi)に点を打ってから, (xi − δx,i, yi)から (xi + δx,i, yi)に伸びる直線を引
き, さらに点 (xi − δx,i, yi)および点 (xi + δx,i, yi)を中心とする短い縦方向の線分を引きます.

一方, コマンド plot が with errorbars というオプション付きで起動され, データファイルが図 9.28の

形式 2だった場合には, gnuplotは点 (xi, yi)に点を打ってから, (xmin
i , yi)から (xmax

i , yi)に伸びる直線を引き,

さらに点 (xmin
i , yi)および点 (xmax

i , yi)を中心とする短い縦方向線分を引きます.

9.1.7.2.3 横軸と縦軸誤差を含むグラフ 横軸と縦軸に縦軸のみに誤差を含むデータファイルの書式としては,

図 9.31の 2種類のうちのいずれかが想定されています.

横軸は形式 1, 縦軸は形式 2, というふうにいくつかの形式を混ぜて使うことはできません.

グラフの横および縦方向のエラーバーを付ける場合には, コマンド plot に with xyerrorbars というキー

ワードを付けます. ですから, 典型的な横軸のみにエラーバーを付ける方法は,

plot ’ファイル’ with xyerrorbars [Enter]

図 9.29: 誤差を含む形式 1のデータファイルとそのグラフ

図 9.30: 誤差を含む形式 2のデータファイルとそのグラフ

図 9.31: 横軸にエラーバーを付けるグラフが前提としているデータファイルの書式

図 9.32: 横軸と縦軸にエラーバーを付けるグラフが前提としているデータファイルの書式

となります. ここに, 「ファイル」と書かれた部分には適当なファイル名が入ります.

エラーバーがどのように描かれるかは, 横軸のみ, あるいは縦軸のみにエラーバーを付ける場合と同じです.

9.1.7.2.4 誤差の表記が混在したデータファイルからグラフを作りたいとき いろいろな誤差表記が混在した

データファイルからエラーバー付きのグラフを付けるには, 9.1.5で述べたキーワード using を使います. 実

は, キーワード using の後に指定できる列の種類は 2種類に限られるわけではなく,

using m:n:p:q

などのように何種類でも列を指定できます. そして, 上の例の場合は, 「第 m列と第 n列と第 p列と第 q列」
というふうに解釈されます. ですから, いろいろな誤差表記が混在したデータファイルからエラーバー付きのグ

ラフを作成するときには, キーワード using を使ってグラフから必要な部分を抜き出す作業が必要になります.

9.1.8 データ点のあいだを結ぶ線のなめらかさを変える

データファイルをプロットするときには, gnuplotは標準ではデータ点のあいだを直線で結びます.

この部分を変更して, データ点のあいだを曲線で結ぶようにするには, キーワード smooth に続いてオプショ

ンを指定します. 指定できるオプションは unique, csplines, acsplines, bezier, sbezierの 5種類です.

このそれぞれの意味について表 9.4にまとめておきます. 複数のデータ点が同一の x座標を持つことがない場合に

表 9.4: キーワード smooth のオプション

オプション 意味

unique まずデータファイルを横軸の座標に関してソーティングする. x座標が同一で y
座標が異なる複数のデータ点 (xi, yi1),. . . ,(xi, yim) がある場合は, これらの y座

標を平均したデータ点
(

xi,
yi1 + · · · + yim

m

)
をグラフにプロットし, もともとの

データ点は捨てる. 最後に, プロットされたデータ点のあいだを折れ線で結ぶ.

csplines まず uniqueと同一の処理を実行してからデータを近似する 3次の自然スプライ
ンによる補間をおこなう.

acsplines まず uniqueと同一の処理を実行してから重み付きの自然スプライン補間をおこ
なう. 重みはデータファイルの第 3列で指定される.

bezier データを近似する, データ点の数と同じ次数のベジエ曲線を引く.

sbezier まず uniqueと同一の処理を実行してからデータを近似するベジエ曲線を引く.

は, 上記のうち unique と csplines を使った場合にはデータ点をすべて通る曲線が引かれますが, acsplines,

bezier, sbezier を使った場合には曲線は場合によってはデータ点を避けます.

これらの例を以下で見てゆくことにします. だたし, acsplinesは必要とするデータファイルの形式が他と

異なるので, これの紹介を最後に回します.

9.1.8.1 準備

図 9.33(a)のようなデータファイルを考えます. ファイル名を for-smooth.dat としておきましょう.

plot ’for-smooth.dat’ with points pt 6 ps 2 [Enter]

とした結果を図 9.33(b)に示します. このデータファイルでは, 同一の x座標データ点が 3個ずつあることが

わかります.

まず最初に,

plot ’for-smooth.dat’ with linespoints, ’for-smooth.dat’ with points pt 6 ps 2 [Enter]

として, 単純にデータ点のあいだを線で結んでみます. ただし, 上の例で,

’for-smooth.dat’ with points pt 6 ps 2

1 0
5 5
4 2
2 2
4 3
5 4
1 2
2 3
3 3
3 1
1 1
2 4
3 2
4 1
5 3

(a)ファイル (b)グラフ

図 9.33: x座標が小さい順に並んでいないデータファイルとそのグラフ

となっている部分は, 図を見やすくするためにグラフにデータ点そのものを重ね書きしている部分であり, 本来

は必要ではありません. 以下の例では, データ点と曲線の関係を見やすくするために, グラフにデータ点をつね

に重ね書きしておくことにします.

さて, 上のようなコマンドを実行した結果は結果は図 9.34のようになります. データが x座標の小さい順に

図 9.34: データファイル for-smooth.dat を単純に線で結んだもの

並べられていないことと同じ x座標を持つ点が複数あることから, 滅茶苦茶な線が引かれています.

9.1.8.2 キーワード unique を指定したグラフ

次に, キーワード unique を指定してみます.

plot ’for-smooth.dat’ smooth unique, ’for-smooth.dat’ with points pt 6 ps 2 [Enter]

と入力すると, 結果は図 9.35のようになります. 図 9.35から, グラフの縦軸の平均値を通る折れ線が引かれ

ていることがわかります.

9.1.8.3 キーワード csplines を指定したグラフ

続いて, キーワード csplines を指定してみます.

図 9.35: キーワード unique を適用した結果

plot ’for-smooth.dat’ smooth csplines, ’for-smooth.dat’ with points pt 6 ps 2 [Enter]

と入力しましょう. 結果は図 9.36のようになります. 図 9.35の折れ線が図 9.36ではなめらかな曲線で置き

図 9.36: キーワード csplines を適用した結果

換えられています.

9.1.8.4 キーワード bezier を指定したグラフ

さらに, キーワード bezier を指定してみます.

plot ’for-smooth.dat’ smooth bezier, ’for-smooth.dat’ with points pt 6 ps 2 [Enter]

と入力しましょう. 結果は図 9.37のようになります. キーワード bezier を指定した場合には, x軸の値に

関して整列されていないデータ点のあいだを単純に線で結んでいった場合と同様, ねじれた曲線が描かれること

があります. 図 9.37もそのようになっています.

9.1.8.5 キーワード sbezier を指定したグラフ

次はキーワード sbezier を指定してみます.

plot ’for-smooth.dat’ smooth sbezier, ’for-smooth.dat’ with points pt 6 ps 2\rt

図 9.37: キーワード bezier を適用した結果

図 9.38: キーワード sbezier を適用した結果

と入力しましょう. 結果は図 9.38のようになります. キーワード sbezier を指定した場合には, まず unique

と同じ処理がおこなわれるため, 曲線がねじれることはありません. 先に述べた通り, csplinesと sbezierの

違いは, csplinesは計算された点をすべて通過する曲線を引くのに対し, sbezierはそれらの点を適当に避け

た点を引くことです.

9.1.8.6 キーワード acsplines を指定したグラフ

さいごに, キーワード acsplinesを使った例を示します. acsplinesを使う場合には, データファイルには

「重み」という追加情報が必要になります. acsplinesを使った場合にも引かれる曲線はデータ点を適当に避け

るのですが, 重みの数値が大きいところほど曲線がデータ点に近付き, 数値が小さいところほど曲線はデータ点

から離れます.

キーワード acsplines の使い方は unique その他と同じで,

plot ’ファイル’ smooth acsplines [Enter]

です.

例として, 図 9.39(a)のようなデータファイル (ファイル名を ac1.dat とします)をコマンド

plot ’ac1.dat’ smooth acsplines, ’ac1.dat’ pt 6 ps 2 [Enter]

によってプロットしたもの図 9.39(b)に, 図 9.39(b)のようなデータファイル (ファイル名を ac2.dat としま

す)をコマンド

plot ’ac2.dat’ smooth acsplines, ’ac2.dat’ pt 6 ps 2 [Enter]

プロットしたもの図 9.40(b)に示します.

ファイル ac1.dat と ac2.dat の違いは. ac2.dat の方は重みを指定している第 3列の数値がデータの後

半で大きくなっていることです. 結果として, ac2.dat をプロットした図 9.40(b)では, 図の右側の方で曲線

がデータ点に近付いています.

100  0.4127 0.01
110  0.3610 0.01
120  0.6479 0.01
130  0.5184 0.01
140  0.8210 0.01
150  0.8784 0.01
160  0.3303 0.01
170  0.7377 0.01
180  0.3865 0.01
190  0.3862 0.01
200  0.6141 0.01

(a) ファイル ac1.dat (b) グラフ

図 9.39: キーワード acsplines を使った例 (1)

100  0.4127 0.01
110  0.3610 0.01
120  0.6479 0.01
130  0.5184 0.01
140  0.8210 0.01
150  0.8784 0.01
160  0.3303 1000
170  0.7377 1000
180  0.3865 1000
190  0.3862 1000
200  0.6141 1000

(a) ファイル ac2.dat (b) グラフ

図 9.40: キーワード acsplines を使った例 (2)

なお, 重みをデータファイルで指定するかわりに gnuplotのコマンド行で直接指定することもできます. た

とえば, 図 9.39(a)のデータファイルで, 重みに第 3列目のデータを使うのをやめ, かわりにすべての行につい

て数値 10.0を使うことにする場合には,

plot ’ac1.dat’ using 1:2:(10.0) smooth acsplines [Enter]

とします. この場合, キーワード using のあとの部分の表記は

横軸に使う列番号:縦軸に使う列番号:(重み)

のようになっています. 3番目の部分を括弧 ()で囲む必要があることに注意して下さい.

9.1.9 デフォルトのデータファイル描画のスタイルの確認と変更

デフォルトのデータファイルの描画のスタイルを確認するには,

show data style [Enter]

と入力します. また, デフォルトのデータファイルの描画のスタイルを変更したいときには, たとえば

set data style lines [Enter]

などのように,

set data style

というキーワードに続けて希望するスタイルを指定します.

デフォルトのスタイルを変更すると, その変更は gnuplot を終了するまで有効になります.

9.2 3次元データファイルのプロット

9.2.1 単純な 3次元データファイルのプロット

3次元データファイルをプロットするときには, コマンド plot のかわりに コマンド splot を使います.

データ点を単純にプロットしてゆくだけの場合, 3次元データをプロットするやり方は 2次元データファイルを

プロットする場合と変わりません.

また, データファイルの記法も, データ点の座標として x座標, y座標, z座標の 3種類を指定する必要があ

ること以外は 2次元のものと同じです.

例として, 図 9.41(a) に簡単な 3次元データのデータファイルを示します. ファイルの名前は 3d0.dat と

しておきます. このデータファイルを使い,

splot ’3d0.dat’ pt 6 ps 2 [Enter]

というコマンドを実行すると, 図 9.41(b)のようなグラフが得られます. だたし, データ点を見やすくするた

めにデータ点の表示スタイルを変更した上でデータ点の大きさを大きくしています. この例からもわかるよう

1 1 250
1 2 300
1 3 400
2 1 300
2 2 400
2 3 600

(a) データファイル 3d0.dat (b) グラフ

図 9.41: 簡単な 3次元データとそのグラフ

に, 点の種類や大きさの変更法は 2次元グラフを描画するときと同一です.

9.2.2 網かけグラフのプロット

ところで, 3次元の関数をプロットするときには関数のグラフに網かけをすることで関数の形状が表現されたわ

けですが, 同じことをデータファイルに対して実行するときには, いくつか注意が必要です.

そもそも網かけグラフとはどのようなものかというと, xy平面に置かれた格子を曲面 z = f(x, y)に投影した
影, すなわち, 曲面 z = f(x, y)が光を半分透過するスクリーンであり, xy平面に格子が置かれていて, xy平面

の下の方から拡散しない光をあてたときに, 格子が曲面 z = f(x, y) 上に作る像を斜めから見ているものです (図

9.42). 関数のグラフを作成するときにはこの格子が自動的に作成されたのですが, 3次元データファイルから

網かけグラフを作るときには, この格子を自分で作成しておく必要があります.

具体的には, たとえば図 9.43(a)に示すように,

• x軸に平行な第 1番目の格子の格子点の y座標とその格子点における関数値から成るブロック

• x軸に平行な第 2番目の格子の格子点の y座標とその格子点における関数値から成るブロック

• ............

122



図 9.42: 網かけグラフの作り方

のようなデータを作成しておきます (ブロックとは, データファイル中の空白行で区切られたかたまりのことでし

た). なお, どのブロックも同じ数のデータ点を含むようにしないといけません.

図 9.43(a)のファイルの名前を 3d1.dat としましょう. このファイルを

splot ’3d1.dat’ with linespoints pt 6 ps 2 [Enter]

によってプロットすると, 図 9.43(b)のような網かけグラフが作成できます.

1 1 20
1 2 30
1 3 40

2 1 50
2 2 60
2 3 70

3 1 40
3 2 30
3 3 20

4 1 10
4 2 10
4 3 10

1 1.5 2 2.5 3 3.5 4 1
1.5

2
2.5

3

(a) データファイル 3d1.dat (b) グラフ

図 9.43: 3次元データの網かけグラフそのグラフ

実は, 図 9.43の例は実際には網かけグラフを作成しているわけではなく, 同じブロックに含まれるデータ点の

あいだを線で結んでから各々のブロックを結ぶ線を補っているだけです.

9.2.3 x座標と y座標の自動割り付け

2次元データファイルでは, 縦軸の値だけを書いたファイルを作っておくと, gnuplotが横軸の値として 0,

1,2,...という整数を自動的に割り振ってくれました. 3次元データファイルについても同様のことができます.

z座標だけが書かれたファイルがあると, gnuplotは以下のような規則にしたがって x座標と y座標を補って

ゆきます.

• データの最初の行では, gnuplotはは x座標および y座標の両方に 0を割り振る.

• データが 1行下にゆくにしたがって x座標を 1ずつ増やす. y座標は変えない.

• 空白行が見付かると, 空白行の次のデータ行では y座標を 1増やす. x座標は 0に戻す.

ですから, たとえば, 図 9.44(a)のデータファイルと図 9.44(b)のデータファイルを gnuplotによってプロッ

トした場合, 同じグラフが得られます.

1
1
1

2
2
2

3
3
3

0 0 1
1 0 1
2 0 1

0 1 2
1 1 2
2 1 2

0 2 3
1 2 3
2 2 3

(a)x座標と y座標を省略 (b) 全座標を記載

図 9.44: x座標と y座標が省略されたデータファイルと省略されていないデータファイル

9.2.4 グラフのスタイルの変更やデータファイルからのデータの抜き出し

2次元データファイルのプロットで使えたキーワード index, every, using は 3次元グラフでも使えます.

使い方については 2次元データファイルの対応する箇所を参照して下さい.

9.2.5 球座標と円柱座標

データファイルから 3次元グラフを描くときには, データの座標系として球座標 (3次元極座標)や円柱座標を

使うことができます.

9.2.5.1 座標系の切り換えのためのコマンド

直交座標と球座標, 円柱座標の切り換えには set mapping というコマンドを使います. 座標変換に関連した

のコマンドの一覧を表 9.5にまとめておきます.

表 9.5: 座標系の変換

機能 コマンド

球座標を選択 set mapping spherical [Enter]

円柱座標を選択 set mapping cylindrical [Enter]

直交座標を選択 set mapping cartesian [Enter]

現在選択されている座標系を表示 show mapping

球座標や円柱座標を用いたグラフのプロットはデータファイルのプロットのときのみ有効です. 関数をプロッ

トするときには使えません.

9.2.5.2 球座標によるデータファイル

球座標用のデータファイルは 1行あたり 2個あるいは 3個の数値を含みます.

1行あたりの数値が 3個のデータファイルでは, i番目のデータ点の極座標が (ri, θi, φi)であるとき, すなわち

直交座標に直すと

xi = ri cos θi cos φi, yi = ri sin θi cos φi, zi = ri sinφi

という座標にデータ点があるとき, 対応するデータファイルの行には, 数値が

θi, φi, ri

という順で並びます. ri が第 3番目に来ることに注意して下さい.

1行あたりの数値が 2個のデータファイルでは, データ点の原点からの距離はつねに 1であると解釈されます.

すなわち, データファイルの第 i行目に

θi, φi

という数値が記載されているとき, gnuplotはこのデータ点を極座標 (1, θi, φi)を持つ点であると解釈します.

例として, 半球面上にある点の極座標がいくつか保存されたファイル spherical.dat のプロットを示します.

ファイル spherical.dat の内容は図 9.45(a) のようなものであるとします. ファイル spherical.dat には

250個のデータ点が含まれるのですが, 図 9.45(a)にはそのうち最初の 5個だけが記載されています.

図 9.45(a)のようなデータファイルを準備した上で

set mapping spherical [Enter]

splot ’spherical.dat’ [Enter]

というコマンドを実行すると, 図 9.45(b)のようなグラフが描画されます.

0.000  0.000
0.000  0.063
0.000  0.126
0.000  0.188
0.000  0.251

 

(以下略)

(a)データファイル (b) グラフ

図 9.45: 半球面上のデータ点の描画例

9.2.5.3 円柱座標によるデータファイル

円柱座標用のデータファイルは 1行あたり 2個あるいは 3個の数値を含みます.

1行あたりの数値が 3個のデータファイルでは, i番目のデータ点の極座標が (ri, θi, zi)であるとき, すなわち

直交座標に直すと

xi = ri cos θi, yi = ri sin θi, zi = zi

という座標にデータ点があるとき, 対応するデータファイルの行には, 数値が

θi, zi, ri

という順で並びます. 球座標の場合と同様に ri が第 3番目に来ることに注意して下さい.

1行あたりの数値が 2個のデータファイルでは, データ点の原点からの距離はつねに 1であると解釈されます.

すなわち, データファイルの第 i行目に

θi, zi

という数値が記載されているとき, gnuplotはこのデータ点を球座標 (1, θi, zi)を持つ点であると解釈します.

例として, 円筒にある点の円柱座標がいくつか保存されたファイル cylindrical.dat のプロットを示します.

ファイル cylindrical.dat の内容は図 9.46(a) のようなものであるとします. ファイル cylindrical.dat

には 250個のデータ点が含まれるのですが, 図 9.46(a)にはそのうち最初の 5個だけが記載されています.

図 9.46(a)のようなデータファイルを準備した上で

set mapping cylindrical [Enter]

splot ’cylindrical.dat’ [Enter]

というコマンドを実行すると, 図 9.46のようなグラフが描画されます.

 

(以下略)

0.000  0.000
0.000  0.020
0.000  0.040
0.000  0.060
0.000  0.080

(a) データファイル (b) グラフ

図 9.46: 円筒上のデータ点の描画

9.3 非線形最小2乗法によるデータへの曲線のあてはめ

gnuplotには, 与えられたデータファイルに記載されたデータに曲線や曲面をあてはめる機能があります. 本

節ではこの機能について簡単に紹介します.

9.3.1 簡単な例題

最小 2乗法でデータに曲線や曲面をあてはめる手順は,

• パラメータを含む関数を定義する

• コマンド fit を使って先に定義された関数をデータファイルに記載されたデータにあてはめる

というものです.

さっそく, 例題を見てみましょう.

図 9.47(a)のようなデータファイルを考えます. ファイル名は f0.dat とします.

このデータに, f(x) = ax + bという直線をあてはめてみることにします. このためには,

10 10
20 30
30 20
40 70
50 30
60 10
70 80
80 40
90 60

(a)データファイル f0.dat (b) f0.dat をプロットしたもの

図 9.47: データファイル f0.dat とそのプロット

f(x)=a*x+b [Enter]

fit f(x) ’f0.dat’ via a,b [Enter]

と入力します. すると, gnuplotのウィンドウは図 9.48のような状態になります. このウィンドウには, パラ

図 9.48: コマンド fit の実行が終わったときの gnuplot のウィンドウの状態

メータの値, 計算が終了するまでにおこなわれた繰り返し計算の回数およびいくつかの統計情報が表示されます.

では, これに続いて, 計算された直線とデータを重ね書きしてみましょう.

plot ’f0.dat’ pt 6 ps 2, f(x) [Enter]

というコマンドを実行し, データファイル f0.dat と 関数 f(x)を重ね書きしてみます. すると, 図 9.49の

ような結果が得られます.

次に, このグラフに

f(x) = ax + b + c sin(dx)

図 9.49: データ f0.dat に直線を当てはめた結果

と関数をあてはめてみます. ただし, 計算を始める前に, パラメータ aは 0.5 bは 15, cは 10, dは 0.1 で

初期化しておくことにします.

a=0.5 [Enter]

b=15 [Enter]

c=10 [Enter]

d=0.1 [Enter]

e=1 [Enter]

f(x)=a*x+b + c *sin(d*x) [Enter]

fit f(x) ’f0.dat’ via a,b,c,d [Enter]

plot ’f0.dat’ pt 5 ps 3, f(x) [Enter]

というコマンドを実行すると, すると, 図 9.50のような結果が得られます.

図 9.50: データ f0.dat に非線形関数を当てはめた結果

計算が終わったときのパラメータをファイルに保存しておきたいときには,

save variables ’パラメータファイル’ [Enter]

とします. ここに, 「パラメータファイル」と書かれた部分には適当なファイル名が入ります. 上記のコマン

ドを実行すると, たとえば図 9.51のようなファイルが作成されます.

図 9.51: パラメータが保存されたファイル

9.3.2 コマンド fit の使い方

コマンド fit の基本的な使い方は 2種類あります.

ひとつめは,

fit 関数 ’データファイル’ via 変数 1, 変数 2, ... [Enter]

というものです. この場合, 「変数 1」, 「変数 2」と書かれた部分には, 当てはめに使う関数に含まれる調整

可能なパラメータの名前が入ります.

ふたつめの使い方は,

fit 関数 ’データファイル’ via ’パラメータファイル’ [Enter]

というものです. この使い方をする場合には, あらかじめパラメータの初期値が書かれたファイルを用意してお

く必要があります.

パラメータファイルの作るときには, たとえば図 9.52に示すように, 変数の名前とその初期値を等号をはさん

で並べます. 先ほどの例で見た, 図 9.51のファイルも, 註釈行 (文字「]」で始まる行)を除くと, この形をし

ています.

a=0.5

b=10

図 9.52: パラメータファイルの記法

ふたつめの使い方をした場合, コマンド fit の実行が終了した段階で,

update ’パラメータファイル’ [Enter]

というコマンドを実行すると, パラメータファイルの内容が更新されます.

パラメータの中に値を変えてはいけないものがあるときには, たとえば図 9.53のパラメータ aの部分のよう

に, パラメータ値の右側に「#FIXED」という文字を書いておきます. そうすると, パラメータの値が更新され

ることはありません.

a=0.5 \#FIXED

b=10

図 9.53: 固定パラメータを含むパラメータファイルの記法

9.3.3 データに非線形関数を当てはめるときの注意

データに当てはめる関数が線形のときには, パラメータの初期値をどのように取っても最終的に最適な当ては

めが達成されることが保証されます. これに対し, 非線形関数を当てはめるときには, 最終的な当てはめの結果

の良し悪しがパラメータの初期値に大幅に依存します.

• 線形関数を当てはめることができる場合には, 可能な限り線形関数を使う

• 非線形関数を当てはめに使うときには初期値の選択に注意する

ということを心掛けて下さい.

130



第10章 plusモード: 数式の組版とテキストのより
細かい制御

gnuplotには, plusモードという, 簡単な数式を組んだり, いくつかの図記号を入れたり, 文字の大きさや

書体などを細かく制御したりするためのモードがあります.

本節ではこの plusモードについて説明してゆきます.

10.1 plusモードの開始と終了

先に述べた通り, plusモードでは, 簡単な数式を組んだり, いくつかの図記号を入れたり, 文字の大きさや書

体などを細かく制御したりすることができます.

ただし, この機能は, ターミナルが postscript になっている場合, すなわち PostScript形式でファイルに

保存する場合にしか有効になりません.

さらに, ただの PostScript モードでも, 数式は図記号などが取り扱えるわけではありません. ですから,

plusモードの機能を使いたいときには, ターミナルを postscriptの plusモードにしておく必要があります.

このためには,

set terminal postscript eps plus [Enter]

あるいは

set terminal postscript plus [Enter]

とします.

印刷結果をプリンタに直接送る場合には後者でもよいのですが, 作成した図をほかの文書に取り込む予定なら

ば前者のほうが良いでしょう. この節の説明の内容は前者でも後者でも変わらないのですが, 説明の便宜上本節

では前者に限定して議論を進めます.

plus モードを起動すると, 作成される PostScriptファイル には日本語のデータが入ります. ですから, 手

持ちのプリンタが日本語に対応していないなどの理由で日本語のデータを含まない PostScriptファイルを作成

しなければならないときには,

set locale "" [Enter]

と指定する必要があります. 逆に, 日本語のデータを入力できるモードに戻るには,

set locale "ja_JP.EUC" [Enter]

とします.

ターミナルが postscriptの状態で plusモードを無効にしたいときには,

set terminal postscript eps noplus [Enter]

とします.

ここまでに出て来たコマンドを表 10.1にまとめておきます.

表 10.1: plus モードの起動と終了に関連したコマンド

機能 コマンド

plusモードを有効にする set terminal postscript eps plus [Enter]

plusモードを無効にする set terminal postscript eps noplus [Enter]

日本語を有効にする set locale "ja_JP.EUC" [Enter]

日本語を無効にする set locale "" [Enter]

なお, plusモードでは, 文字列は必ず単一引用符「’ ’」で囲まなければなりません. これ以外のモードで

は「" "」と「’ ’」のどちらを使っても良かったのですが, plusモードで「" "」を使うと誤動作が起きること

があります. また, いわゆる半角カナは使えません.

• 文字列は単一引用符「’ ’」で囲まなければならない

• 半角カナは使えない

ということを記憶しておいて下さい.

10.2 plusモードのコマンド

plusモードにおけるコマンドの記法は基本的に LATEXと同じで, 最初の文字が「\」(バックスラッシュ)で始ま

ります. コンピュータの状態によっては記号「\」が画面には円記号 (「￥」のような形) として表示されるこ

とがあるので注意して下さい. また, キーボードの表面にも, 記号「\]のかわりに記号「￥」が表示されてい

ることがあります.

コマンドの有効範囲を明示する必要が生じることもあります. この典型的な例が, 後の 10.3.2節で述べる

フォントの選択です.

コマンドの有効範囲を指示するためには, 記号「{」「}」を使います. 「{」が有効範囲の始まり, 「}」が有

効範囲の終わりです. たとえば, コマンド\it(10.3.2節参照)の有効範囲を限定するには,

{\it }

などのようにします.

コマンドの中には引数を取るものもあります. コマンドの引数をあらわすのにも, 多くの場合は記号「{」「}」

を使います.

たとえば, コマンド\sqrtは平方根記号を表示するコマンドなのですが (後述, 表 10.5), 数 2に平方根の記

号を付けるには,

\sqrt{2}

のように, コマンド sqrtの引数として 2を指定します.

これ以外に, 引数を取るコマンドで,

\command=値

という書き方を要求するものもあります (後述, 10.3節).

10.3 文字の大きさや字体の指定と回転

plusモードでは, 文字の大きさや字体を指定したり, テキスト全体を指定した角度だけ回転したりすることが

できます.

10.3.1 文字の大きさの指定

文字の大きさを指定するときには, 文字列のどこかに

\size=数

と書きます. ここに, 「数」と書いてある部分には文字の大きさが入ります. 単位はポイントです.

文字の大きさをテキストの最初で指定すると, 次に\sizeコマンドがあらわれるまでは先に指定した大きさの

文字が使われます. テキストの途中でも文字の大きさを変更できます.

図中でいろいろな大きさの文字を使う例として,

set terminal postscript eps plus? [Enter]

set output ’gnuplot-plus-size.eps’? [Enter]

set label ’\size=30 30ポイント \size=32 32ポイント’ at 0.2,0.8 [Enter]

set label ’\size=26 26ポイント \size=28 28ポイント’ at 0.2,0.7 [Enter]

set label ’\size=22 22ポイント \size=24 24ポイント’ at 0.2,0.6 [Enter]

set label ’\size=18 18ポイント \size=20 20ポイント’ at 0.2,0.5 [Enter]

set label ’\size=14 14ポイント \size=16 16ポイント’ at 0.2,0.4 [Enter]

set label ’\size=10 10ポイント \size=12 12ポイント’ at 0.2,0.3 [Enter]

set label ’\size=6 6ポイント \size=8 8ポイント’ at 0.2,0.2 [Enter]

set label ’\size=2 2ポイント \size=4 4ポイント’ at 0.2,0.1 [Enter]

plot [0:2] [0:1.0] 0 [Enter]

というコマンドを実行したときの出力を図 10.1に示します.

図 10.1: いろいろな文字の大きさの指定

10.3.2 字体の指定

字体を指定するためのコマンドとして, \rm, \it, \tt, \sf, \bf があります. これらの意味を表 10.2に

示します.

ボールド体を指定するとローマン体に対応する太字の字体が選択されますが, \it,\tt,\sfと組み合わせるこ

とで, それぞれ対応する字体の太字が得られます.

表 10.2: 利用可能な字体一覧

記号 コマンド 記号 コマンド 記号 コマンド

\rm ローマン体 \it イタリック体 \tt タイプ文字

\sf サンセリフ体 \bf 太字

以下にいろいろな字体を選択する例を示します. ただし, グラフをみやすくするために, 図の大きさを縦横と

も標準の半分にしてあります (最初の行).

set terminal postscript eps plus? [Enter]

set output ’gnuplot-plus-font.eps’ [Enter]

set label ’{\sf Sanserif} {\tt\bf BoldSanserif} ’ at 0.5,0.7 [Enter]

set label ’{\tt TypeWriter} {\tt\bf BoldTypeWriter}’ at 0.5,0.5 [Enter]

set label ’{\it Italic} {\it\bf BoldItalic} ’ at 0.5,0.3 [Enter]

set label ’{\rm Roman} {\bf Bold} ’ at 0.5,0.1 [Enter]

plot [0:2] [0:1.0] 0 [Enter]

結果は図 10.2のようになります.

図 10.2: いろいろな字体の選択

10.3.3 テキスト全体の回転

テキスト全体を回転させるには, 文字列の開始をあらわす単一引用符の直後に

\rotate=角度

と書きます. ここに, 「角度」と書いてある部分には回転の角度が入ります. 単位は度です.

テキストを回転させたいときには, 必ず文字列の最初 (文字列の開始をあらわす単一引用符の直後)でコマンド

\rotateを使わなければなりません. それ以外の場所ではエラーになります.

例として, 文字列をいろいろな角度回転させたものを含む図を作成してみます. ただし, グラフをみやすくす

るために, 図の大きさを縦横とも標準の半分にしておきます (最初の行のコマンド).

set size 0.5,0.5 [Enter]

set terminal postscript eps plus [Enter]

set output ’gnuplot-plus-rotate.eps’ [Enter]

set label ’\rotate=315 315度回転’ at 1.1,0.4 [Enter]

set label ’\rotate=270 270度回転’ at 1.0,0.4 [Enter]

set label ’\rotate=225 225度回転’ at 0.9,0.4 [Enter]

set label ’\rotate=180 180度回転’ at 0.9,0.5 [Enter]

set label ’\rotate=135 135度回転’ at 0.9,0.6 [Enter]

set label ’\rotate=90 90度回転’ at 1.0,0.6 [Enter]

set label ’rotate=45 45度回転’ at 1.1,0.6 [Enter]

set label ’rotate=0 0度回転’ at 1.1,0.5 [Enter]

plot [0:2] [0:1.0] 0 [Enter]

というコマンドを実行すると, 図 10.3 のような結果が得られます.

図 10.3: いろいろな角度の回転

10.4 特殊記号および各種の記号

10.4.1 特殊記号

コマンドの一部として使われることが多いいくつかの記号はふだんは単体では表示できなくなっています. そ

れらを表示するためのコマンドを表 10.3に示します.

表 10.3: 特殊記号一覧

記号 コマンド 記号 コマンド 記号 コマンド

$ \$ ^ \^ _ \_

\ \backslash { \{ } \}

使用例を以下に示します.

set terminal postscript eps plus [Enter]

set output ’gnuplot-plus-special.eps’ [Enter]

set label ’\size=150 \$ \^ \_ \{ \}’ at 0.1,0.5 [Enter]

plot [0:2] [0:1.0] sin(x) [Enter]

結果は図 10.4のようになります.

図 10.4: 特殊記号の使用例

10.4.2 各種記号, 外国語の文字

著作権表示の記号, 英語以外の外国語の記号などのいくつかの記号が用意されています. これらを表 10.4に

示します.

表 10.4: 各種記号と外国語記号一覧

記号 コマンド 記号 コマンド 記号 コマンド 記号 コマンド

† \dag ‡ \ddag § \S ¶ \P

c© \copyright £ \pounds œ \oe Œ \OE

æ \ae Æ \AE å \aa Å \AA

ø \o Ø \O Ãl \l ÃL \L

ß \ss

使用例を以下に示します.

set output ’gnuplot-plus-misc-symbols.eps’ [Enter]

\verb?set title ’\size=24 \copyright 2001 ’ [Enter]

\verb?set label ’\size=48 \dag \ddag \S \P \pounds \AE \ae \OE \oe’ at 0.1,0.5 [Enter]

plot [0:2] [0:1] sin(x) [Enter]

結果は図 10.5のようになります.

10.5 数式の表現

plusモードにおける数式の表現は LATEXの数式組版機能を簡略化したものです. LATEXを知らない人は,

図 10.5: いろいろな記号の使用例

• 記号$ではさまれている領域が数式として認識される

• 下付き添字を使いたいときには数式環境中 (記号$ で挟まれた領域) に a_1などのように書く

• 上付き添字を使いたいときには数式環境中に a^1などのように書く

ということだけ覚えておいて下さい.

10.6 グラフの標題に簡単な数式を入れる

まず簡単な例題を見ておくことにします.

以下に, グラフの横軸に「下付き添字の例: a1」, 縦軸に「上付き添字の例: a2」というラベルを付け, グ

ラフ全体に「平方根の例:
√

2」という題目を付けている例を示します.

set terminal postscript eps plus [Enter]

set output ’gnuplot-plus-1.eps’ [Enter]

set xlabel ’下付き添字の例: $a_1$’ [Enter]

set ylabel ’上付き添字の例: $a^2$’ [Enter]

set title ’平方根の例: $\sqrt{2}$’ [Enter]

plot sin(x) [Enter]

結果は図 10.6のようになります.

この例からもわかるように, 数式を入れるときには, 文字列を

’数式でない部分$数式$’

のように書きます. 数式が 2個含まれる場合には,

’数式でない部分$数式 1$数式でない部分 2$数式 2$’

図 10.6: グラフに簡単な数式を入れた例

のようにします. 数式が 3個以上含まれる場合も同様です. 数式が 2個以上ある場合には, 数式と認識される

部分は

• 1個目の記号$と 2個目の記号$のあいだ

• 3個目の記号$と 4個目の記号$のあいだ

• ......

となります. 記号$の数が奇数個の場合はエラーになります.

10.7 数式を組むためのいくつかのコマンド

plusモードには数式を組むときに必要となる, 分数, 平方根を表示するためのコマンドや, 数式の上や下に線

を引くコマンドが用意されています. これらの一覧を表 10.5に示します.

表 10.5: 数式を組むためのコマンド一覧

機能 コマンド 使用例 結果

分数 \frac \frac{2}{3} 2
3

分数の平方根 \fracsqrt \fracsqrt{\frac{2}{3}}
√

2
3

分数の上に線を引く \fracoverline \fracoverline{\frac{2}{3}} 2
3

平方根 \sqrt \sqrt{2}
√

2

上に線を引く \overline \overline{テキスト,数式} テキスト,数式

下に線を引く \underline \underline{テキスト,数式} テキスト,数式

ボックスに入れる \mbox \mbox{テキスト,数式} テキスト, 数式

表 10.5の最後の 3個のコマンドは数式以外でも使えます.

10.8 利用可能な数学記号

plusモードで利用可能な数学記号は LATEXで用意されている数学記号に若干の変更を加えたものです. 以下に

これらを列挙してゆきます.

10.8.1 矢印

上下左右の向きの矢印と両側に矢のついた矢印が利用可能です. 表 10.6に一覧を示します. 「記号」とい

う見出しがついた部分に書かれているのが表示される記号, 「コマンド」という見出しがついた部分に書かれて

いるのが対応するコマンドです.

表 10.6: 数式モードにおける矢印記号一覧

記号 コマンド 記号 コマンド 記号 コマンド 記号 コマンド

⇓ \Downarrow ⇐ \Leftarrow ⇔ \Leftrightarrow ⇒ \Rightarrow

⇑ \Uparrow ↓ \downarrow ← \leftarrow ↔ \leftrightarrow

→ \rightarrow ↑ \uparrow

10.8.2 ギリシャ文字

ギリシャ文字については, いくつかの大文字と小文字すべて, さらに小文字の書体を変えたものが利用可能で

す. ギリシャ文字の大文字に省略されているものがあるのは, それらの字形が英語の大文字と同じだからです.

表 10.7に大文字一覧を, 表 10.8に小文字一覧を示します. 表の見方は先と同様です.

表 10.7: ギリシャ文字 (大文字)一覧

記号 コマンド 記号 コマンド 記号 コマンド 記号 コマンド

Γ \Gamma ∆ \Delta Θ \Theta Λ \Lambda

Ξ \Xi Π \Pi Σ \Sigma Υ \Upsilon

Φ \Phi Ψ \Psi Ω \Omega

表 10.8: ギリシャ文字 (小文字)一覧

記号 コマンド 記号 コマンド 記号 コマンド 記号 コマンド

α \alpha β \beta γ \gamma δ \delta

ε \epsilon ε \varepsilon ζ \zeta η \eta

θ \theta ϑ \vartheta ι \iota κ \kappa

λ \lambda µ \mu ν \nu ξ \xi

π \pi $ \varpi ρ \rho σ \sigma

ς \varsigma τ \tau υ \upsilon φ \phi

ϕ \varphi χ \chi ψ \psi ω \omega

10.8.3 数式中のアクセント記号

数式を組んでいると, 文字の上に微分をあらわすドットなどのいろいろな印をつける必要が生じることがあり

ます. このようなときのためのコマンドを表 10.9にまとめておきます. なお, 表 10.9ではアクセントが付け

られる文字の例を aにしてありますが, この部分には任意の文字を指定を指定できます.

表 10.9: 数式中のアクセント記号一覧

記号 コマンド 記号 コマンド 記号 コマンド 記号 コマンド

â \hat{a} ǎ \check{a} ă \breve{a} á \acute{a}

à \grave{a} ã \tilde{a} ā \bar{a} ~a \vec{a}

ȧ \dot{a} ä \ddot{a}

10.8.4 関数記号

いろいろな数学関数のために対応する記号が用意されています. 関数記号を表示するためのコマンドの大部分

は, その関数名の先頭に記号「\ 」をつけたものです. 関数記号の一覧を表 10.10 に示します.

表 10.10: 関数記号一覧

記号 コマンド 記号 コマンド 記号 コマンド 記号 コマンド

arccos \arccos arcsin \arcsin arctan \arctan arg \arg

cos \cos cosh \cosh cot \cot coth \coth

cse \cse deg \deg det \det dim \dim

exp \exp gcd \gcd hom \hom inf \inf

ker \ker lg \lg lim \lim lim inf \liminf

lim sup \limsup ln \ln log \log max \max

min \min Pr \Pr sec \sec sin \sin

sinh \sinh sup \sup tan \tan tanh \tanh

10.8.5 2項演算子と関係記号

数式を組むときには, 加減乗除の記号「+」, 「-」, 「*」,「/」以外にもいろいろな記号がありますし, そ

もそも数学でよく使われる乗算記号×は英文字 xとまぎらわしいためにコンピュータではキーボードから直接入

力できるようにはなっていません. また, 不等号の記号「<」「>」はキーボードにありますが, 記号「≤」「≥」
はキーボードにはありません.

plusモードには, このような記号を表示する能力があります. 利用可能な記号の一覧を表 10.11

10.8.6 その他の数学記号

これまで述べてきた以外にも, 数式を組んでいるときには, 和の記号
∑

, 積の記号
∏
, 偏微分記号 ∂ などの

いろいろな記号が必要になります.

これまで述べてきたもの以外で用意されている数学記号の一覧を表 10.12に示します.

表 10.11: 2項演算子と関係記号一覧

記号 コマンド 記号 コマンド 記号 コマンド 記号 コマンド

± \pm × \times · \cdot ÷ \div

† \dagger ≡ \equiv ∼ \sim ≈ \approx
∼= \cong ∝ \propto

≤ \leq ≥ \geq 6= \neq

≤ \le ≥ \ge 6= \ne

表 10.12: その他の数学記号一覧

記号 コマンド 記号 コマンド 記号 コマンド 記号 コマンド

ℵ \aleph ~ \hbar ı \imath ℘ \wp

< \Re = \Im ∠ \angle ∇ \nabla
√

\surd ∂ \partial ∞ \infty
∑

\sum∫
\smallint 〈 \langle 〉 \rangle

10.8.7 大きい括弧

複雑に入り組んだ数式を組んでいるときには, いろいろな大きさの括弧を使って数式をみやすくする, という

ことがよく行われます. このような目的のために, 通常より大きい括弧が何種類か用意されています.

表 10.13: 大きい括弧一覧

記号 コマンド 記号 コマンド 記号 コマンド 記号 コマンド(
\bigl(

)
\bigr)

{
\bigl\{

}
\bigr\}

10.8.8 いくつかの例

参考のために, 数式の組版の例をいくつか表 10.14に示します. これらは数式モード (記号「$で囲まれた領

域)でしか使えないことに注意して下さい. また, 実際の組版の内容は表 10.14に示されているものに比べてか

なり汚くなります.

10.8.9 機能の制限

これまで紹介してきたコマンドには, いくつか機能制限があります. どのような制限があるかというと,

• 添字に分数を使うことはできない

• 分数の中にさらに分数を指定することはできない

というものです. このうち, 最初の方の制限は若干問題になることがあります. たとえば, a
1
10 のような数式

を組むことはできません.

また, 10.8.8節でも述べたように, 組まれる数式は一般にあまり美しくありません. より美しい数式を組みた

い場合は, terminal を latex にしてグラフを保存するか, terminal を tgif にしてグラフを保存して tgif

を使って図を入れるとよいでしょう.

表 10.14: 数式組版の例

数式 コマンド

sin t ≤ t \sin t \leq t{(
(a2

1 + a2
2) + a2

3

)
+ a2

4

}
\bigl\{ \bigl( (a_1^2+a_2^2)+a_3^2 \bigr)+a_4^2 \bigr\}∑∞

i=1 ai × bi \sum_{i=1}^{\infty} a_i \times b_i∫ p2

p1
exp τdτ \smallint_{p_1}^{p_2} \exp [\tau] d \tau

∂Ψ
∂x

\frac{\partial \Psi}{\partial x}√
b0s + b1

s2 + a1s + a2
\fracsqrt{\frac{b_0 s+b_1}{s^2+a_1 s + a_2}}

limx→0
sin x

x \lim_{x\rightarrow 0}\frac{\sin x}{x}

ẍ = −λẋ − µx + cos(t) \ddot{x}=-\lambda \dot{x} -\mu x + \cos(t)

